


PRESCRIBE Commands

Command Reference



We shall have no liability or responsibility to customers or any other person or entity with respect to any liability, loss or damage caused or alleged to be caused directly or indirectly by equipment sold or furnished by us, including, but not limited to, any interruption of service, loss of business or anticipatory profits, or consequential damages resulting from the use or operation of the equipment or software.

NO LIABILITY WILL BE ASSUMED FOR ANY DAMAGE CAUSED BY IMPROPER INSTALLATION.

Notice on Software

SOFTWARE USED WITH THIS PRINTER MUST SUPPORT THE PRINTER'S NATIVE MODE OR ONE OF ITS EMULATION MODES.

Notice

This manual, the computer programs in the page printer referred to in this manual, and any other copyrightable subject matter sold or provided with or in connection with the sale of the page printer, are protected by copyright. All rights are reserved. Copying or other reproduction of all or part of this manual, the computer programs, or any other copyrightable subject matter without the prior written consent of KYOCERA Document Solutions Inc. is prohibited. Any copies made of all or part of this manual, the computer programs, or any other copyrightable subject must contain the same copyright notice as the material from which the copying is done.

The information in this manual is subject to change without notification. Additional pages may be inserted in future editions. The user is asked to excuse any omissions or errors in the present edition.

No responsibility is assumed if accidents occur while the user is following the instructions in this manual. No responsibility is assumed for defects in the printer's firmware.

Regarding Tradenames:

PRESCRIBE is a registered trademark of Kyocera Corporation. KPDL is a trademark of Kyocera Corporation.

Diablo 630 is a product of Xerox Corporation. IBM Proprinter X-24E is a product of International Business Machines Corporation. Epson LQ-850 is a product of Seiko Epson Corporation. HP LaserJet III, HP LaserJet 4, HP LaserJet 4 Plus, HP LaserJet 5Si, HP LaserJet 5P, HP LaserJet 5M, HP LaserJet 2100, HP LaserJet 4000, and HP 7550A are products of Hewlett-Packard Company. Hewlett-Packard, PCL, and HP-GL are registered trademarks of Hewlett-Packard Company. Centronics is a trade name of Centronics Data Computer Corp. Other product names and company names that appear in this manual are trademarks or registered trademarks of their respective owners.

Typeface Trademark Acknowledgement

All resident fonts in the print system are licensed from Bitstream Inc. and Agfa corporation. For font license information for each model, refer to the User's Manual.

Helvetica, Palatino and Times are registered trademarks of Linotype-Hell AG. Century Schoolbook, Stymie, and Cooper-Black are trademarks of Kingsley-ATF Type Corporation. ITC Avant Garde Gothic, ITC ZapfChancery, ITC ZapfDingbats, ITC Souvenir, ITC Benguiat, and ITC Bookman are registered trademarks of International Typeface Corporation. Revue is a trademark of Esselte Pendaflex Corporation in the U.S., Letraset Canada Ltd. in Canada, and Esselte Letraset Ltd. elsewhere.

Monotype Imaging License Agreement

1. "Software" shall mean the digitally encoded, machine readable, scalable outline data as encoded in a special format as well as the UFST Software.
2. You agree to accept a non-exclusive license to use the Software to reproduce and display weights, styles and versions of letters, numerals, characters and symbols ("Typefaces") solely for your own customary business or personal purposes at the address stated on the registration card you return to Monotype Imaging. Under the terms of this License Agreement, you have the right to use the Fonts on up to three printers. If you need to have access to the fonts on more than three printers, you need to acquire a multi-user license agreement which can be obtained from Monotype Imaging. Monotype Imaging retains all rights, title and interest to the Software and Typefaces and no rights are granted to you other than a License to use the Software on the terms expressly set forth in this Agreement.
3. To protect proprietary rights of Monotype Imaging, you agree to maintain the Software and other proprietary information concerning the Typefaces in strict confidence and to establish reasonable procedures regulating access to and use of the Software and Typefaces.
4. You agree not to duplicate or copy the Software or Typefaces, except that you may make one backup copy. You agree that any such copy shall contain the same proprietary notices as those appearing on the original.
5. This License shall continue until the last use of the Software and Typefaces, unless sooner terminated. This License may be terminated by Monotype Imaging if you fail to comply with the terms of this License and such failure is not remedied within thirty (30) days after notice from Monotype Imaging. When this License expires or is terminated, you shall either return to Monotype Imaging or destroy all copies of the Software and Typefaces and documentation as requested.
6. You agree that you will not modify, alter, disassemble, decrypt, reverse engineer or decompile the Software.
7. Monotype Imaging warrants that for ninety (90) days after delivery, the Software will perform in accordance with Monotype Imaging-published specifications, and the diskette will be free from defects in material and workmanship. Monotype Imaging does not warrant that the Software is free from all bugs, errors and omissions.
8. THE PARTIES AGREE THAT ALL OTHER WARRANTIES, EXPRESSED OR IMPLIED, INCLUDING WARRANTIES OF FITNESS FOR A PARTICULAR PURPOSE AND MERCHANTABILITY, ARE EXCLUDED.
9. Your exclusive remedy and the sole liability of Monotype Imaging in connection with the Software and Typefaces is repair or replacement of defective parts, upon their return to Monotype Imaging.
10. IN NO EVENT WILL MONOTYPE IMAGING BE LIABLE FOR LOST PROFITS, LOST DATA, OR ANY OTHER INCIDENTAL OR CONSEQUENTIAL DAMAGES, OR ANY DAMAGES CAUSED BY ABUSE OR MISAPPLICATION OF THE SOFTWARE AND TYPEFACES.
11. Massachusetts U.S.A. law governs this Agreement.
12. You shall not sublicense, sell, lease, or otherwise transfer the Software and/or Typefaces without the prior written consent of Monotype Imaging.
13. Use, duplication or disclosure by the Government is subject to restrictions as set forth in the Rights in Technical Data and Computer Software clause at FAR 252-227-7013, subdivision (b)(3)(ii) or subparagraph (c)(1)(ii), as appropriate. Further use, duplication or disclosure is subject to restrictions applicable to restricted rights software as set forth in FAR 52.227-19 (c)(2).
14. YOU ACKNOWLEDGE THAT YOU HAVE READ THIS AGREEMENT, UNDERSTAND IT, AND AGREE TO BE BOUND BY ITS TERMS AND CONDITIONS. NEITHER PARTY SHALL BE BOUND BY ANY STATEMENT OR REPRESENTATION NOT CONTAINED IN THIS AGREEMENT. NO CHANGE IN THIS AGREEMENT IS EFFECTIVE UNLESS WRITTEN AND SIGNED BY PROPERLY AUTHORIZED REPRESENTATIVES OF EACH PARTY. BY OPENING THIS DISKETTE PACKAGE, YOU AGREE TO ACCEPT THE TERMS AND CONDITIONS OF THIS AGREEMENT.

This page is left blank intentionally

Table of Contents

Overview of PRESCRIBE Commands by Function 2

Access Commands	2
Print System Setting Commands	2
Margin and Page Portrait/Landscape Orientation Commands	3
Font Commands	4
Cursor Movement Commands	5
Vector Graphics Commands	5
Path Mode Graphics Commands	6
Raster Graphics Commands	7
Color Commands	7
Barcode Commands	8
Macro Commands	8
Debug Commands	9
External Media Control Commands	9
<i>e-MPS</i> Commands	10

PRESCRIBE Commands 11

ACLI — Add CoLor by Index	12
ALTB A — [ALlocate TaBle] Assign user-defined character table	14
ALTB C — [ALlocate TaBle] Convert character code	15
ALTB D — [ALlocate TaBle] Delete user-defined character table	16
ALTB E — [ALlocate TaBle] End defining combination characters	17
ALTB G — [ALlocate TaBle] Generate user-defined table	18
ALTB R — [ALlocate TaBle] Release user-defined character table	19
ALTB S — [ALlocate TaBle] Start to define the combination character	20
ALTB T — [ALlocate TaBle] define combined character by Table	21
ALTF — change to ALTerNate Font	23
AMCR — call Automatic MaCRo	25
APSG — Assign Paper Source Group	27
ARC — draw filled-in ARC	28
ASFN — ASsign external characters for FoNt	31
ASTK — Assign STAcKer trays	32
BARC — draw BARCode	34
BKLT — print in BooKLeT binding	38
BLK — draw filled-in BLoCk	40
BOX — draw BOx	42
CALL — CALL macro	44
CCPY — Carbon CoPY	46
CDSK — Check hard DiSK	49
CID — Configure color-Image Data	50
CIR — draw CIRcle	52
CLIP — CLIP current path	54
CLPR — CLiP Rectangular area	55
CLSP — CLoSe Path	57
CMNT — CoMmeNT	59
CMOD — Color MODe	60
COPY — set number of COPIES	61
CPAL — Control PALETTE	63
CPTH — Character PaTH	64

CSET — Change symbol SET by symbol-set ID	65
CSTK — select Collator STacKer	68
CTXT — print Centered TeXT	69
DAF — Delete All Fonts	71
DAM — Delete All Macros	72
DAP — Draw to Absolute Position	73
DELf — DELEte Font	75
DELM — DELEte Macro	76
DPAT — select Dashed PATtern	78
DRP — Draw to Relative Position	80
DRPA — Draw to Relative Position specified by Angle	82
DUPX — select/deselect DUPlEX mode	84
DXPG — select DupleX PaGe side	85
DZP — Draw to Zero-relative Position	87
EMCR — Enable MaCRo depending on paper source	89
ENDB — END a two-dimensional Barcode string	91
ENDC — END carbon Copy	92
ENDD — END Dump	93
ENDM — END Macro	94
ENDR — END Raster data	96
EPL — select EcoPrint Level	98
EXIT — EXIT from PRESCRIBE mode	99
FDIR — MP tray Feed DIRection	100
FILL — FILL closed path	101
FLAT — set FLATness	104
FLST — print Font LiST	105
FOLD — FOLD printed pages	107
FONT — change current FONT	108
FPAT — generate Fill PATtern	110
FRPO — Firmware RePrOgram	112
FRPO INIT — FRPO-INITialize	114
FSET — change current Font SETting by characteristic	115
FTMD — bitmap FonT MoDe	119
GPAT — set Gray PATtern	122
GRAY — represent GRAY	124
GRRD — GRaphic data ReaD	125
HUE — adjust HUE	127
INTL — InterNaTionaL characters	129
JOBd — JOB Deletion	132
JOBL — print JOB List	134
JOBO — JOB Output	136
JOBp — JOB, print with Print options	138
JOBS — JOB Start	140
JOBT — JOB Terminate	144
JOG — JOG output stacks for separation	145
LAPI — LoAd API Program	146
LDfC — LoAd Font Character	147
LDfN C — generate bitmap character for LoADing FoNt	149
LDfN F — create header for LoADing FoNt	151
LDfN S — LoAd truetype FoNt	153
LGHT — adjust LiGHTness	155
MAP — Move to Absolute Position	157
MCLR — Match CoLoR	159
MCRO — define MaCRO	160
MDAT — set MeDia type ATtribute	162
MID — Multi-tray ID	164
MPSS — e-MPS Storage	165

MPTE — MP Tray Empty	166
MRP — Move to Relative Position	167
MRPA — Move to Relative Position specified by Angle	169
MSTK — select Mailbox STacKer	171
MTYP — select Media TYPe	172
MZP — Move to Zero-relative Position	174
NEWP — start NEW Path	176
OTRY — select paper Output TRaY	177
PAGE — start new PAGE	179
PANT — create PANTone color palette	180
PARC — in Path, draw ARC	182
PAT — select fill PATtern	184
PCRP — in Path, Curve to Relative Position	186
PCZP — in Path, Curve to Zero-relative Position	188
PDIR — set Print DIRection	190
PDRP — in Path, Draw to Relative Position	193
PDZP — in Path, Draw to Zero-relative Position	195
PELP — in Path, draw ELLipse	197
PIE — draw PIE chart	199
PMRA — in Path, Move to Relative position specified by Angle	201
PMRP — in Path, Move to Relative Position	203
PMZP — in Path, Move to Zero-relative Position	205
PNCH — PuNCH	207
PRBX — in Path, draw Round BoX	209
PRRC — in Path, at Relative position, draw aRC	211
PSRC — select Paper SouRCe	213
PXPL — PiXel PLacement	214
!R! — PRESCRIBE start sequence	215
RCLT — Rotated CoLLaTion	217
RDMP — Received-data DuMP	218
RES — RESet	220
RESL — select RESoLution	222
RGBL — control RGB Level	223
RGST — offset ReGiSTration	225
RPCS — Return to Previous Code Set	226
RPF — Return to Previous Font	227
RPG — Return to Previous Graphics state	228
RPP — Return to Previous Position	230
RPPL — Return to Previous PaLette	232
RPU — Return to Previous Unit	233
RTTX — RoTate TeXt	234
RTXT — print Right-aligned TeXT	236
RVCD — ReceiVe Compressed raster Data	238
RVCL — ReceiVe CoLor raster data	239
RVRD — ReceiVe Raster Data	240
RWER D — [Read/Write External Resource] Delete data on external media	242
RWER F — [Read/Write External Resource] Format external media	243
RWER I — [Read/Write External Resource] automatically print partition Information	244
RWER L — [Read/Write External Resource] print partition List	245
RWER R — [Read/Write External Resource] Read data from external media	246
RWER S — [Read/Write External Resource] Store TrueType font	247
RWER T — [Read/Write External Resource] set Terminate string	249
RWER W — [Read/Write External Resource] Write data to external media	250
RWRF D — [Read/Write Resource File] Delete data on external device	252
RWRF F — [Read/Write Resource File] Format external device	253
RWRF L — [Read/Write Resource File] print resource file List	254
RWRF P — [Read/Write Resource File] set hidden file	255

RWRF R — [Read/Write Resource File] Read	256
RWRF T — [Read/Write Resource File] set Terminate string	257
RWRF W — [Read/Write Resource File] Write data to external device (API program)	258
RWRF W — [Read/Write Resource File] Write data to external device (font)	259
RWRF W — [Read/Write Resource File] Write data to external device (macro)	261
RWRF W — [Read/Write Resource File] Write data to external device (printable data)	262
SATU — adjust SATUration level	263
SBM — Set Bottom Margin	264
SCAP — Set line CAP	266
SCCS — Save Current Code Set	268
SCF — Save Current Font	270
SCG — Save Current Graphics state	272
SCOL — Select COLor	274
SCP — Save Current Position	275
SCPI — Set Characters Per Inch	276
SCPL — Save Current PaLette	278
SCRC — Set Command Recognition Character	279
SCS — Set Character Spacing	281
SCSZ — Set Custom paper SiZe	282
SCU — Save Current Unit	283
SDP — Store Dash Pattern	285
SEM — Set Emulation Mode	287
SETF — SET alternate Font	288
SFA — Set bitmap Font Attributes	290
SFNT — Select current FoNT by typeface	294
SGPC — Set kcGI Pen Color	297
SHMI — Set HMI	298
SIMG — Set IMAge model	299
SIMP — create SIMPle color palette	302
SIR — Set Image Refinement level	304
SLJN — Set Line JoiN	305
SLM — Set Left Margin	307
SLPI — Set Lines Per Inch	309
SLPP — Set Lines Per Page	310
SLS — Set Line Spacing	312
SMLT — Set Miter LimiT	313
SMNT — Set MoNiTor simulation	315
SPAL — Select PALette	317
SPD — Set Pen Diameter	318
SPL — Set Page Length	320
SPO — Set Page Orientation	322
SPSZ — Set Paper SiZe	324
SPW — Set Page Width	326
SRM — Set Right Margin	328
SRO — Set Raster Options	330
SROP — Set Raster OPeration	332
SSTK — select Sorter STAcKer	334
STAK — select paper STAcKer	335
STAT — STATus	336
STM — Set Top Margin	338
STPC — set STaPle Counter	340
STPL — STaPLe	341
STR — SeT dot Resolution	343
STRK — STRoKe current path	344
SULP — Set UnderLine Parameters	346
TATR — apply Tray ATtributes	348
TEXT — print TEXT	349

TPRS — Text PaRSing	351
TRSM — TRansparency Separate Mode	352
UNIT — set UNIT of measurement	353
UOM — Unit Of Measurement per dots	355
VMAL — Virtual Mailbox ALias	357
VMOB — Virtual Mailbox Output Bin	358
VMPW — set Virtual Mailbox PassWord	359
WIDE — set WIDE A4 mode	361
WRED — WRite EnD	362
XBAR — print two dimensional barcode	363
XBCP 0 — select barcode type/reset all other XBCP parameters	365
XBCP 1 — specify narrowest element width	366
XBCP 2 — specify error correction level by percentage	367
XBCP 3 — set error correction level	368
XBCP 4 — set data code word rows	369
XBCP 5 — set data code word columns	370
XBCP 6 — determine aspect ratio of vertical height and horizontal width	371
XBCP 7 — determine bar height	373
XBCP 8 — automatically set rows and columns	374
XBCP 9 — enable truncation	375
XBCP 10 — enable file name	376
XBCP 11 — enable block count	377
XBCP 12 — enable time stamp	378
XBCP 13 — enable sender ID	379
XBCP 14 — enable addressee ID	380
XBCP 15 — enable file size	381
XBCP 16 — enable checksum	382
XBCP 17 — allow control of file ID	383
XBCP 18 — enable Macro PDF417 symbol mode	385
XBCP 19 — position symbols at the specified locations	387
XBUF — define a BUFFer name	391
XPAT — generate eXpanded fill PATtern	393

PRESCRIBE Command Reference

1. Overview of PRESCRIBE Commands by Function

The tables in this section list all the printing system commands supported by the various models. Support for certain commands varies depending on the printing system model.

Access Commands

These commands provide access in and out of PRESCRIBE mode.

Command	Function	Page
EXIT	EXIT from PRESCRIBE mode	99
!R!	PRESCRIBE start sequence	215
SCRC	Set Command Recognition Character	279

Print System Setting Commands

These commands establish general printing environment parameters.

Command	Function	Page
COPY	set number of COPIES	61
EPL	select EcoPrint Level	98
FDIR	MP tray Feed DIRection	100
FRPO	Firmware RePrOgram	112
FRPO INIT	FRPO-INITialize	114
MDAT	set MeDia type ATtribute	162
MPTE	MP Tray Empty	166
MTYP	select Media TYPe	172
OTRY	select paper Output TRaY	177
PSRC	select Paper SouRCE	213
RCLT	Rotated CoLLaTion	217
RES	RESet	220
RESL	select RESoLution	222
RGST	offset ReGiSTration	225
RPU	Return to Previous Unit	233
SCSZ	Set Custom paper SiZe	282
SCU	Save Current Unit	283
SEM	Set Emulation Mode	287
SIR	Set Image Refinement level	304
SPSZ	Set Paper SiZe	324
STAK	select paper STAcKer	335
STAT	STATus	336
TRSM	TRansparency Separate Mode	352
UNIT	set UNIT of measurement	352
UOM	Unit Of Measurement per dots	355
WIDE	set WIDE A4 mode	361

Page Control, Text, and Comment Commands

The placement commands determine where text will be placed on the page.

Command	Function	Page
CMNT	CoMmeNT	59
CTXT	print Centered TeXT	69
PAGE	start new PAGE	179
RTTX	RoTate TeXt	234
RTXT	print Right-aligned TeXT	236
TEXT	print TEXT	348

Margin and Page Portrait/Landscape Orientation Commands

These commands are used to set the top, bottom, right, and left margins, and to specify portrait or landscape page orientation. Normal text data is printed inside the margins, but the TEXT and RTEXT commands can be used to print character strings or graphics anywhere in the entire printable area of the page, regardless of the margin settings. There are several commands for setting margins, so it is possible to select the one that best suits the purpose at hand. Margin settings can be changed in the middle of a page.

Note that the following limitations apply to the margin setting commands.

- The right margin must be at least one space character to the right of the left margin.
- The bottom margin must be at least one line below the top margin.
- It is not possible to specify negative values as margin settings.

For example, a right margin setting beyond the printable area's right edge would be brought back just inside the right edge of the printable area. In like manner, it is not possible to use a bottom margin setting that would place the bottom margin past the bottom edge of the printable area. An easy way to set the right and bottom margins to the very edge of the printable area is to use a large value such as 100 (centimeters).

The margins and page orientation can be reset by using the RES command.

Command	Function	Page
SBM	Set Bottom Margin	264
SLM	Set Left Margin	307
SLPP	Set Lines Per Page	310
SPL	Set Page Length	320
SPO	Set Page Orientation	322
SPW	Set Page Width	326
SRM	Set Right Margin	328
STM	Set Top Margin	338

Margins and page orientation are more fully explained in Section 1.3. of the *Technical Reference* manual.

Text Spacing Commands

These commands control the line spacing, character spacing, underline spacing, and underline thickness. The SLS and SLPI commands are used to set the line spacing. The SCS and SCPI commands are used to set the character spacing. These spacing settings may be changed even in the middle of a page.

The table below lists the minimum, maximum, and default values for the text spacing command parameters. If values outside of the above ranges are used, the command is ignored. If a value smaller than one dot is specified for the character spacing, it is interpreted as 0, which results in proportional spacing.

	Minimum value	Maximum value	Default value
Line spacing	1 dot	2,047 dots	6 lpi
	0.01 cm	17.3 cm	0.423 cm
	0.004 inch	6.8 inch	0.167 inch
Character spacing	1 dot	2,047 dots	10 cpi
	0.01 cm	17.3 cm	0.254 cm
	0.004 inch	6.8 inch	0.1 inch

Using the font mode (FTMD command), the line spacing and character spacing can be set automatically when the bitmap fonts are selected. Refer to the sections on the FTMD and SFA commands for additional details.

Command	Function	Page
SCPI	Set Characters Per Inch	276
SCS	Set Character Spacing	281
SHMI	Set HMI	298
SLPI	Set Lines Per Inch	309
SLS	Set Line Spacing	312
SULP	Set UnderLine Parameters	346

Font Commands

The font commands are used to select fonts. Fonts can be selected using substitute numbers instead of a font's specific number. It is also possible to change the attributes of fonts and print them in a unique style. Additionally, there are commands to download fonts to the printing system from an external source, and to generate fonts.

Command	Function	Page
ALTB A	[ALlocate TaBle] Assign user-defined character table	14
ALTB C	[ALlocate TaBle] Convert character code	15
ALTB D	[ALlocate TaBle] Delete user-defined character table	16
ALTB E	[ALlocate TaBle] End defining combination characters	17
ALTB G	[ALlocate TaBle] Generate user-defined table	18
ALTB R	[ALlocate TaBle] Release user-defined character table	19
ALTB S	[ALlocate TaBle] Start to define the combination character	20
ALTB T	[ALlocate TaBle] define combined character by Table	21
ALTF	change to ALternate Font	23
ASFN	ASsign external characters for FoNt	31

Command	Function	Page
CSET	Change symbol SET by symbol-set ID	65
DAF	Delete All Fonts	71
DELf	DELEte Font	75
FLST	print Font LiST	105
FONT	change current FONT	108
FSET	change current Font SETting by characteristic	115
FTMD	bitmap FonT MoDe	119
INTL	InterNaTionaL characters	129
LDFC	LoaD Font Character	147
LDFN C	generate bitmap character for LoaDing FoNt	149
LDFN F	create header for LoaDing FoNt	151
LDFN S	LoaD truetype FoNt	153
RPCS	Return to Previous Code Set	226
RPF	Return to Previous Font	227
SCCS	Save Current Code Set	268
SCF	Save Current Font	270
SETF	SET alternate Font	288
SFA	Set bitmap Font Attributes	290
SFNT	Select current FoNT by typeface	294
TPRS	Text PaRSing	351

Fonts are more fully explained in *Chapter 4* of the *Technical Reference* manual.

Cursor Movement Commands

These commands move the cursor to specific positions. Cursor movements can be established for absolute positions based on the margins, zero point positions based on the printable area, or relative positions based on the current cursor position.

Cursor movement commands only change the position of the cursor. Also, the cursor may not be moved outside of the printable area (see MZP command). It is possible to store the cursor position in memory then to return to that position stored in memory.

Command	Function	Page
MAP	Move to Absolute Position	157
MRP	Move to Relative Position	167
MRPA	Move to Relative Position specified by Angle	169
MZP	Move to Zero-relative Position	174
RPP	Return to Previous Position	230
SCP	Save Current Position	275


Vector Graphics Commands

The commands below are used to create figures such as lines, circles, and rectangles, and to add shading, simply by adding parameters.

It is possible to specify colors for all vector graphics. For details, see a color specific command such as SCOL.

If an attempt is made to create figures outside of the printable area using these commands, the coordinates are adjusted automatically so that the figures print inside the printable area. As in the example below, this can produce printed results different from what is expected.

Print settings according to command



Printing extremely complex figures can exhaust printing system memory. Downloadable fonts can be particularly taxing on printing system memory, producing unexpected print results. It may be possible to correct the problem by removing some of the downloaded fonts or by simplifying the graphic. To determine current memory usage, refer to a status page.

Command	Function	Page
ARC	draw filled-in ARC	28
BLK	draw filled-in BLock	40
BOX	draw BOX	42
CIR	draw CIRcle	52
DAP	Draw to Absolute Position	73
DPAT	select Dashed PATtern	78
DRP	Draw to Relative Position	80
DRPA	Draw to Relative Position specified by Angle	82
DZP	Draw to Zero-relative Position	87
FPAT	generate Fill PATtern	110
GPAT	set Gray PATtern	122
PAT	select fill PATtern	184
PIE	draw PIE chart	199
SPD	Set Pen Diameter	318
XPAT	generate eXpanded fill PATtern	393

Path Mode Graphics Commands

The following commands provide a variety of path construction operators and painting operators for stroking or filling paths.

Command	Function	Page
CLIP	CLIP current path	54
CLPR	CLiP Rectangular area	55
CLSP	CLoSe Path	57
CPTH	Character PaTH	64

Command	Function	Page
FILL	FILL closed path	101
FLAT	set FLATness	104
NEWP	start NEW Path	176
PARC	in Path, draw ARC	182
PCRCP	iiin Path, Curve to Relative Position	186
PCZP	in Path, Curve to Zero-relative Position	188
PDIR	set Print DIRection	190
PDRP	in Path, Draw to Relative Position	193
PDZP	in Path, Draw to Zero-relative Position	195
PELP	in Path, draw ELLipse	197
PMRA	in Path, Move to Relative position specified by Angle	201
PMRP	in Path, Move to Relative Position	203
PMZP	in Path, Move to Zero-relative Position	205
PRBX	in Path, draw Round BoX	209
PRRC	in Path, at Relative position, draw aRC	211
RPG	Return to Previous Graphics state	228
SCAP	Set line CAP	266
SCG	Save Current Graphics state	272
SDP	Store Dash Pattern	285
SIMG	Set IMAge model	298
SLJN	Set Line JoiN	305
SMLT	Set Miter LimiT	313
STRK	STRoKe current path	344

Raster Graphics Commands

Raster graphics commands can be used to draw any graphic design by specifying the individual dots. The dot resolution is selectable from 75, 100, 150, 200, 300, 600, and 1200 dots per inch.

Command	Function	Page
ENDR	END Raster data	96
PXPL	PiXel PLacement	214
RVCD	ReceiVe Compressed raster Data	238
RVRD	ReceiVe Raster Data	240
SRO	Set Raster Options	330
SROP	Set Raster OPeration	332
STR	SeT dot Resolution	343

Color Commands

Color commands are used to create a color palette, assign colors to a palette, and specify colors. In this manual, color commands are marked with **COLOR** under the command name.

Command	Function	Page
ACLI	Add CoLoR by Index	12
CID	Configure color-Image Data	50
CMOD	Color MODe	60
CPAL	Control PALette	63
GRAY	represent GRAY	124
GRRD	GRaphic data ReaD	125
HUE	adjust HUE	127
LGHT	adjust LiGHtNess	155
MCLR	Match CoLoR	159
PANT	create PANTone color palette	180
RGBL	control RGB Level	223
RPPL	Return to Previous PaLette	232
RVCL	ReceiVe CoLoR raster data	239
SATU	adjust SATURation level	263
SCOL	Select COLor	274
SCPL	Save Current PaLette	278
SGPC	Set kcGI Pen Color	297
SIMP	create SIMPlE color palette	302
SMNT	Set MoNiTor simulation	315
SPAL	Select PALette	317

Barcode Commands

Barcodes conforming to various specifications can be printed simply by specifying the type of barcode and the barcode data. It is also possible to print barcodes with check digits added.

Command	Function	Page
BARC	draw BARCode	34
ENDB	END a two-dimensional Barcode string	91
XBAR	print two dimensional barcode	363
XBCP	select barcode type/reset all other XBCP parameters, etc.	365 to 387
XBUF	define a BUFFer name	391

Macro Commands

A series of PRESCRIBE commands can be defined for sequential processing as a macro. Macros are selected using the CALL command and automatically using the AMCR command.

Command	Function	Page
AMCR	call Automatic MaCRo	25
CALL	CALL macro	44
CCPY	Carbon CoPY	46

Command	Function	Page
DAM	Delete All Macros	72
DELM	DELeTe Macro	76
EMCR	Enable MaCRo depending on paper source	89
ENDC	END carbon Copy	92
ENDM	END Macro	94
MCRO	define MaCRO	160

Debug Commands

These commands can be used to check the print data in hexadecimal format for print jobs with unexpected output.

Command	Function	Page
ENDD	END Dump	93
RDMP	Received-data DuMP	218

External Media Control Commands

If an external storage, such as JEIDA Ver. 4 memory card or CompactFlash card, or a hard disk, is installed in the printing system, the following commands can be used to write data to or read from it. These commands do not apply to the models that do not have an external storage.

Command	Function	Page
RWER D	[Read/Write External Resource] Delete data on external media	242
RWER F	[Read/Write External Resource] Format external media	243
RWER I	[Read/Write External Resource] automatically print partition Information	244
RWER L	[Read/Write External Resource] print partition List	245
RWER R	[Read/Write External Resource] Read data from external media	246
RWER S	[Read/Write External Resource] Store TrueType font	247
RWER T	[Read/Write External Resource] set Terminate string	249
RWER W	[Read/Write External Resource] Write data to external media	250
RWRF D	[Read/Write Resource File] Delete data on external device	252
RWRF F	[Read/Write Resource File] Format external device	253
RWRF L	[Read/Write Resource File] print resource file List	254
RWRF P	[Read/Write Resource File] set hidden file	255
RWRF R	[Read/Write Resource File] Read	256
RWRF T	[Read/Write Resource File] set Terminate string	257
RWRF W	[Read/Write Resource File] Write data to external device (API program)	258
RWRF W	[Read/Write Resource File] Write data to external device (font)	259
RWRF W	[Read/Write Resource File] Write data to external device (macro)	261
RWRF W	[Read/Write Resource File] Write data to external device (printable data)	262
WRED	WRite EnD	362

Control Commands for Option Devices

A variety of options, such as feeders, sorters, document finishers, etc., are available for print systems. The commands below provides access to the optional equipment.

Command	Function	Page
APSG	Assign Paper Source Group	27
ASTK	Assign STacKer trays	32
BKLT	print in BooKLeT binding	38
CSTK	select Collator STacKer	68
DUPX	select/deselect DUPlEX mode	84
DXPG	select DupleX PaGe side	85
FOLD	FOLD printed pages	107
JOG	JOG output stacks for separation	145
MID	Multi-tray ID	164
MSTK	select Mailbox STacKer	171
PNCH	PuNCH	207
SSTK	select Sorter STacKer	334
STPC	set STaPle Counter	340
STPL	STaPLe	341

e-MPS Commands

The printing system features various job storing options – e-MPS (enhanced Multiple Printing System) – that utilize the hard disk. For this feature to be performed, a print job must be defined as an e-MPS job by the JOBS command. Once the job is defined and stored in the hard disk by the JOBS command, it can be retrieved from the hard disk for printing with various printing features such as stapling, duplex-printing, mailboxing, etc., added on the fly.

Command	Function	Page
CDSK	Check hard DiSK	49
JOB D	JOB Deletion	132
JOBL	print JOB List	134
JOBO	JOB Output	136
JOBP	JOB, print with Print options	138
JOBS	JOB Start	140
JOBT	JOB Terminate	144
MPSS	e-MPS Storage	165
VMAL	Virtual Mailbox ALias	357
VMOB	Virtual Mailbox Output Bin	358
VMPW	set Virtual Mailbox PassWord	359

2. PRESCRIBE Commands

In the command format descriptions below, the command is written in fixed-pitch CAPITAL letters. Parameters are indicated in lowercase *italics*. Optional parameters, which may be omitted, are enclosed in brackets []. Three dots (...) mean that the preceding parameters may be repeated. Except for the initial !R!, PRESCRIBE commands may be written in lowercase letters if preferred.

Example:

```
!R! SFNT 'Helvetica'; EXIT;
```

Refer to *Chapter 1* of the *Technical Reference* manual for the discussion on how PRESCRIBE follows the rule regarding upper- and lowercase letters.

The functions of some PRESCRIBE commands are duplicated by the panel keys. For example, the same status page can be printed by the STAT command or by pressing the panel keys. The *User's Manual* provides a full description of the control panel functions.

Many PRESCRIBE commands can be embedded commands within word processing software. This support depends upon the emulation mode and the software. See *Emulation* of the *Technical Reference* manual.

ACLI — Add CoLoR by Index

COLOR

Format

```
ACLI index, color1, color2, color3;
```

Parameters

index:

integer from 1 to 255

color1:

value for red, integer from 0 to 255, in RGB color space
value for lightness, number from 0.0 to 100.0, in L*a*b color space

color2:

value for green, integer from 0 to 255, in RGB color space
value for red/green, number from 0.0 to 100.0, in L*a*b color space

color3:

value for blue, integer from 0 to 255, in RGB color space
value for yellow/blue, number from 0.0 to 100.0, in L*a*b color space

Function

The ACLI command adds a color to the currently active palette. This command is ignored when the currently active palette is a simple color palette or the Pantone color palette. If an index number larger than the palette size is given, this command is also ignored. If a color is added to the existing index number, the current color for that index number is automatically replaced with the new color.

To add a color in a palette, the following guidelines must be used depending on the color space used:

For RGB color space, specify the values for red, green, and blue with an integer from 0 to 255, where 0 = 0% and 255 = 100%. Any value outside this range is rounded to the minimum or maximum value, namely 0 or 255. A fractional value beyond the decimal point is discarded.

For L*a*b color space, specify relative values for the lightness, red/green, and yellow/blue with a number from 0.0 to 100.0. Any value outside this range is rounded to the minimum or maximum value, namely 0.0 or 100.0.

File

```
!R! RES;

DAM;
UNIT C;
SFNT 'AntiqueOlive-Bd', 20;

CPAL C; SPAL 1; CID 0, 1;

ACLI 1, 0, 0, 255;
ACLI 2, 0, 255, 0;
ACLI 3, 255, 0, 0;

MZP 1, 1; SCOL 1;
TEXT 'This line is printed in blue.';
MZP 1, 2.4; SCOL 2;
TEXT 'This line is printed in green.';
MZP 1, 3.8; SCOL 3;
TEXT 'This line is printed in red.';

EXIT;
```

Printout



This line is printed in blue.

This line is printed in green.

This line is printed in red.

Related Commands

CPAL, SPAL

ALTB A — [ALlocate TaBle] Assign user-defined character table

Format

ALTB A, *table-id*;

Parameter

table-id:

number identifying the character table assigned to the resident font, from 1 to 65535

Function

The ALTB A command assigns the user-defined character table to the current resident font.

File

See *ALTB C* on page 15.

ALTB C — [ALlocate TaBle] Convert character code

Format

ALTB C, *table-id*, *source-character*, *target-character*;

Parameters

table-id:

number identifying the character table to convert the character, from 1 to 65535

source-character:

character code before conversion: 1-byte code = 0 to 255; 2-byte code = 0 to 65533, or in hexadecimal notation (\$xxxx).

target-character:

character code after conversion: 1-byte code = 0 to 255; 2-byte code = 0 to 65533, or in hexadecimal notation (\$xxxx).

Function

The ALTB C command converts the character code given as *source-character* to the character code given as *target-character* in the user-defined character table.

If the character table having the identical number already exists, the existing character table is replaced by the new character table.

File

In the example below, the character table number 56535 is generated by ALTB G, in which the character “\ (backslash, character code 92)” is replaced with the Euro currency symbol “a (character code 186 for Roman-9 symbol set).”

```
!R!
SFNT "TimesNewRoman", 14;
CSET 4U; CMNT Roman-9;
ALTB G, 56535, 1;
ALTB C, 56535, 92, 186;
ALTB A, 56535;
TEXT "The Euro currency is symbolized as \.";
ALTB R, 56535; CMNT Release user table;
PAGE; EXIT;
```

Printout

The Euro currency is symbolized as €.

ALTB D — [ALlocate TaBle] Delete user-defined character table

Format

ALTB D, *table-id*;

Parameter

table-id:

number identifying the character table, from 1 to 65535, to be deleted

Function

The ALTB D command deletes the character table defined by the user.

If the table to be deleted is currently allocated to the resident font (see ALTB A), terminate the allocation by using ALTB R before deleting. It is possible to delete all user-defined character tables at once by using ALTB D, 0;

File

See *ALTB C* on page 15.

Related Commands

ALTB A, ALTB R

ALTB E — [ALlocate TaBle] End defining combination characters

Format

ALTB E ;

Parameter

None

Function

This command signals the end of defining combination characters in an already generated user-defined character table which was started by the ALT B S command (page 20). The registered combined characters can be printed by specifying the character code which was assigned at the time of definition.

Related Command

ALT B S

ALTB G — [ALlocate TaBle] Generate user-defined table

Format

ALTB G, *table-id*, *table-format*;

Parameters

table-id:

table ID number: 1 to 65535

table-format:

1 = 1-byte, character codes 0 to 255

2 = 2-byte, character codes 0 to 65533

Function

The ALTB G command generates a user-defined table in either 1-byte or 2-byte format. If the same table ID number already exists, the format table is replaced with the newly generated table.

ALTB R — [ALlocate TaBle] Release user-defined character table

Format

ALTB R, *table-id*;

Parameter

table-id:

number identifying the character table to be released

Function

The ALTB R command terminates the assignment of the user-defined character table for the current resident font. It is also possible to terminate the assignment of all user-defined character tables by using ALTB R, 0;.

ALTB S — [ALlocate TaBle] Start to define the combination character

Format

```
ALTB S, table-id, code[, width, height[, x-coordinate, y-coordinate,
[paint-mode]]];
```

Parameters

table-id:

number identifying the character table to define the combination character, from 1 to 65535

code:

character code for which the combination character is defined. 1-byte code = 0 to 255; 2-byte code = 0 to 65533, or in hexadecimal notation (\$xxxx).

width:

overall width of the combined character, represented by number from 1 to 65535 in units where the width of the resident font character is 1000

height:

overall height of the combined character, represented by number from 1 to 65535 in units where the height of the resident font character is 1000

x-coordinate:

X coordinate of the reference point, integer value converted in reference to 1000 units

y-coordinate:

Y coordinate of the reference point, integer value converted in reference to 1000 units

paint-mode:

paint mode: 0 = normal (filled), non-zero = line width of outline character (stroked): 1 to 100 units

Function

This command starts to define the combination of characters by specifying the region and coordinates. The characters to combine in this region are registered using the ALTB T command (page 21). The ALTB S sequence is terminated by the ALTB E command. If the identical table number already exists, the existing character table is replaced with the new character table.

ALTB T — [ALlocate TaBle] define combined character by Table

Format

ALTB T, *character-code*, *width*, *height*, *x-coordinate*, *y-coordinate*;

Parameters

code:

character code for which the combined character is defined, or hexadecimal notation (\$xxxx), JIS code

width:

character width: the value from 1 to the character width specified by ALTB S (page 20)

height:

character height: the value from 1 to the character height specified by ALTB S (page 20)

x-coordinate:

X coordinate of the reference point, integer value calculated in 1000 unit basis


y-coordinate:

Y coordinate of the reference point, integer value calculated in 1000 unit basis

Function

This command defines the region in which to store the combination of characters within coordinates (0, 0) to (1000, 1000) by specifying a rectangular area and reference point for the characters. There are no restrictions on the number of characters to be combined.

Example:



File

```
!R! res; unit c;
SFNT 'TimesNewRoman', 48;
ALTB G, 1000, 1;
ALTB A, 1000;
ALTB S, 1000, 64, 1800, 1000;
ALTB T, 77, 700, 700, 0, 700; CMNT M;
ALTB T, 65, 700, 700, 630, 700; CMNT A;
ALTB T, 85, 700, 700, 1050, 700; CMNT U;
ALTB T, 73, 700, 700, 1570, 700; CMNT I;
ALTB T, 36, 700, 700, 0, 0; CMNT $;
ALTB T, 57, 700, 700, 400, 0; CMNT 9;
ALTB T, 57, 700, 700, 800, 0; CMNT 9;
ALTB T, 57, 700, 700, 1200, 0; CMNT 9;
ALTB E;
MZP 2, 4; SFNT 'TimesNewRoman', 48; TEXT "HOLIDAYS!";
PAGE; EXIT;
```

Printout

**MAUI
\$999 HOLIDAYS!**

ALTF — change to ALternate Font

Format

ALTF *alternate-number* ;

Parameter

alternate-number: a number from 0 to 47

Function

The ALTF command changes the current font to a font designated by an alternate number. The actual font assigned to this alternate number is set by a previous SETF command.

The alternate number can be a bitmap font number or a number specifically assigned to the scalable font using the SFNT command (See *File* on the next page).

If no font has been assigned to the specified alternate number, the ALTF command has no effect: the font does not change.

The ALTF command enables you to prepare a document using several fonts, without deciding in advance which specific fonts to use. For example, you could use ALTF 1 to designate plain text and ALTF 2 to designate italics, and allow the specific size and style of type to be determined by SETF commands at a later time.

File

```
!R! RES;
SFNT "Helvetica-Nr", 12, 1000;
SFNT "TimesNewRoman-It", 11, 1001;
SETF 1, 1000;
SETF 2, 1001;
MCRO BIBLIO;
    ALTF 1; TEXT %1, E; TEXT " ", E;
    ALTF 2; TEXT %2, E; TEXT " ", E;
    ALTF 1; TEXT %3, N;
ENDM;
ALTF 1; TEXT "Bibliography", N, U;
TEXT, N;
CALL BIBLIO,
    "Radin, Paul,",
    "Hero Cycles of the Winnebago,",
    "Indiana University Publications";
CALL BIBLIO,
    "Neumann, Erich,",
    "Amor and Psyche,", "Bollingen Press";
CALL BIBLIO,
    "Jung, Carl G.," "Man and His Symbols,",
    "Bollingen Press";
PAGE;
EXIT;
```

Printout

Bibliography

Radin, Paul, *Hero Cycles of the Winnebago*, Indiana University Publications
Neumann, Erich, *Amor and Psyche*, Bollingen Press
Jung, Carl G., *Man and His Symbols*, Bollingen Press

AMCR — call Automatic MaCRo

Format

`AMCR mode , [repeat-count] macro-name , parameter , ... ;`

Parameters

mode:

E = Enable automatic macro; Save and retain current printing parameters
 T = Enable automatic macro; Do not save current printing parameters
 D = Disable automatic macro

repeat-count:

number of times the macro is executed

macro-name:

name defined by the MCRO command

parameter:

values given to the dummy parameters in the macro definition

Function

The AMCR command enables the automatic macro (overlay). It automatically executes the macro defined by a MCRO command each time a page is printed. For example, it would be used to print multiple copies of the same form, filled in with different text.

The AMCR command closely resembles the CALL command; therefore, the user should refer to the section on the CALL command for further information.

When the automatic macro is enabled using mode E, the following printing system settings are saved and retained upon completion of the macro: font information, margins, cursor position, line spacing, character spacing, page orientation, unit of measurement, underline parameters, and fill pattern.

When the automatic macro is enabled with mode T, the printing system settings are modified according to the parameter settings contained in the macro.

Color specifications may be used within macros. When an auto-macro (AMCR) is executed, the active palette and color index number are saved and restored after the macro has executed.


The automatic macro remains enabled until the AMCR command with the mode set to D is encountered or the printing system is switched off. Note that mode D requires no other parameters such as repeat-count.

The following commands are invalid within a macro: ASTK, PSRC, CCPY, DUPX, DXPG, JOB, JOBL, JOBO, JOBP, JOBS, JOBT, JOG, LDFC, LDFN, MID, MSTK, RDMP, SEM, SPSZ, SSTK, STAK, STAT, STPL, VMAL, VMPW

File

```
!R! RES;  
MCRO CROPMARK;  
UNIT C;  
MZP 2.46, 3.96; SCP; SCP; SCP;  
MRP -.635, 0; DRP -1.9, 0; RPP;  
MRP 0, -.635; DRP 0, -1.9; RPP;  
MRP 14.85, 0; SCP; SCP;  
MRP .635, 0; DRP 1.9, 0; RPP;  
MRP 0, -.635; DRP 0, -1.9; RPP;  
MRP 0, 21; SCP; SCP;  
MRP .635, 0; DRP 1.9, 0; RPP;  
MRP 0, .635; DRP 0, 1.9; RPP;  
MRP -14.85, 0; SCP;  
MRP -.635, 0; DRP -1.9, 0; RPP;  
MRP 0, .635; DRP 0, 1.9;  
RPP;  
ENDM;  
  
AMCR E, CROPMARK;  
PAGE; UNIT C; SLM 3.6; STM 5.2; EXIT;  
The text starts here.
```

Printout



The text starts here.

Related Command

CCPY

APSG — Assign Paper Source Group

Format

APSG *group #* (cassette 1), *group #* (cassette 2), ...*group #* (cassette 6);

Parameters

group #:

Number representing the group to which the cassette is grouped, from the top (1) towards the bottom (6) cassette, as the following numbers:

- 0 = Ungroup the cassette
- 1 = Assign the cassette for group 1
- 2 = Assign the cassette for group 2
- 3 = Assign the cassette for group 3

Function

The APSG command assigns the paper cassettes with group numbers, between 1 through 3, so that the grouped cassettes behave as a single paper source for bulk printing. The MP tray cannot be assigned to a group. All 6 parameters must be given even though the printer has less than 6 cassettes installed. For missing cassette numbers, specify 0.

The printing system will automatically switch between the paper cassettes in the group when a cassette becomes empty. When a group contains paper of different paper sizes and types, printing will continue for the different paper sizes and types.

If the automatic paper selection feature of the printer driver is used, APSG is disabled. To utilize paper source grouping, the paper source must be explicitly commanded by the printer driver.

Example

To set six paper cassettes into group 1 (cassettes 1 and 4), group 2 (cassettes 2 and 5), and group 3 (cassettes 3 and 6), respectively:

```
!R! APSG 1,2,3,1,2,3; EXIT;
```

In this example, when cassette 1 becomes empty, printing will continue using cassette 4 which is the other part of group 1.

The APSG sequence is valid after the printing system is switched off. To upgroup paper cassettes, use parameter 0 as the following example:

```
!R! APSG 1,0,0,1,0,0; EXIT;
```

Paper cassettes of 2 and 5, and 3 and 6 are ungrouped, keeping paper cassettes 1 and 3 (group 1) intact.

ARC — draw filled-in ARC

Format

ARC *inner-radius, outer-radius, start-angle, end-angle;*

Parameters

inner-radius:

inner radius of arc

outer-radius:

outer radius of arc

start-angle:

starting angle

end-angle:

ending angle

Function

The ARC command works in the standard graphics mode. It draws a filled-in arc-shaped area extending from the inner radius to the outer radius, and from the starting angle to the ending angle. The area is filled with the pattern designated by the previous PAT, FPAT, or GPAT command. The starting and ending angles are specified in integer degrees measured clockwise from the vertical. (Note that the PARC command measures the angles from the positive x axis.) The cursor is located at the center of curvature of the arc, and does not move.

If the inner radius is zero, the area has a pie-slice shape, and can be used to fill in one sector of a pie chart.



If the inner and outer radii are nearly equal and the fill pattern is the all-black pattern (PAT 1, for example), the result is a line arc. The line thickness is the difference between the two radii, rather than the value designated by the SPD (set pen diameter) command.

The ARC command does not draw the boundary of the filled-in area.

File

```
!R! RES; UNIT C;
MAP 5, 8;
PAT 52;
ARC 1, 2, -45, 45;
MAP 5, 10;
PIE 2, 0, 10, 20, 30;
ARC 0, 2, 0, 60;
PAT 20;
ARC 0, 2, 60, 180;
PAT 10;
ARC 0, 2, 180, 360;
PAT 1;
NEWP;
PMZP 4, 2;
PARC 4, 3, 1, 90, 270;
PARC 5, 4, 1, 180, 360;
PARC 6, 3, 1, 270, 90;
PARC 5, 2, 1, 0, 180;
STRK;
MRP 0.6, 1.1;
SFNT "Univers-Md";
TEXT "ARC";
PAGE; EXIT;
```

Printout



Related Command

PARC

ASFN — ASsign external characters for FoNt

Format

ASFN mode , font-number ;

Parameters

mode:

A = assign external characters
R = release external characters

font-number:

number identifying the external character font: 1000 to 65535

Function

The ASFN command assigns external characters which are defined by the user to the current resident font. The external characters should be generated and downloaded in the printing system as an external font by using the LDFC or LDFN command.

For the print systems supporting 2-byte font system, if the user-defined font is a 1-byte code font, it is assigned to the current resident roman (ANK) font. If it is a 2-byte code font, it is assigned to the current 2-byte font.

This command has no effect if the resident font is a bitmap font and the font to be assigned is an outline font, or vice versa.

mode = R is used to release the current resident font from the external characters. Note that it is necessary to release the external font assignment if you want to delete the external font.

ASTK — Assign STackEr trays

Format

ASTK *tray1*, *tray2*, *tray3*, ...;

Parameters

tray1, *tray2*, *tray3*, ...:

Number representing the tray in the optional sorter which becomes the last (bottom) tray in each group, as counted from the top towards the bottom tray.

Function

The ASTK command assigns the sorter trays in groups so that the sorter can be shared among the multiple users. Each parameter represents the last tray in each group of trays. The maximum number of parameters must be equal to the number of the trays the sorter is equipped with. By the factory default, the trays are grouped as the following table:

Group	Sorter model		
	SO-6	SO-30	SO-60
1	1, 2	1, 2	1, 2
2	3, 4	3, 4	3, 4
3	5, 6	5, 6	5
4	7, 8	7, 8	—
5	9 to 15	9	—
Equiv. ASTK command	ASTK 2, 4, 6, 8;	ASTK 2, 4, 6, 8;	ASTK 2, 4;

The ASTK command is ignored if parameters include the value greater than the actual number of trays. If the number of parameters is less than the maximum number of sorter trays, the printing system automatically groups the remaining trays in one, making it the last group. For example, the command sequence ASTK 2, 4, 6; is equivalent to ASTK 2, 4, 6, (bottom tray number);.

The tray group setting made by ASTK is stored indefinitely in the printing system's permanent memory while the power is off, or until it is changed by another ASTK command. It is ignored in the normal sorter mode (FRPO S0, 0;) but the printing system automatically sets up the previous tray group setting when the sorter is again switched to the multiple environment mode (FRPO S0, 1; or FRPO S0, 2;).

File

The following command sequence delivers print jobs to trays 4, 5, 6, and 7 (the second group).

```
!R! FRPO S0, 1; CMNT Enter multiple sorter mode;  
    ASTK 3, 7, 9; CMNT Create three groups;  
    MID 2; CMNT Send the output to 2nd grp.;  
    SSTK; CMNT Sort;  
EXIT;
```

Related Commands

CSTK, MID, MSTK, SSTK

BARC — draw BARCode

Format

`BARC type, flag, 'string' [, short, tall [, bar1, bar2, bar3, bar4, space1, space2, space3, space4]] ;`

Parameters

type:

There are the following barcode types:

- 0 = UPC A
- 1 = UPC A with two-digit supplement
- 2 = UPC A with five-digit supplement
- 3 = UPC D-1
- 4 = UPC D-2
- 5 = UPC D-3
- 6 = UPC D-4
- 7 = UPC D-5
- 8 = UPC E
- 9 = UPC E with two-digit supplement
- 10 = UPC E with five-digit supplement
- 11 = EAN-8 (Truncated JAN)
- 12 = EAN-13 (Truncated JAN)
- 13 = DUN-14 (Distribution Unit Number, EAN)
- 14 = DUN-16 (Distribution Unit Number, EAN)
- 15 = MSI with no check digit
- 16 = MSI with single mod-10 check digit
- 17 = MSI mod-10 followed by mod-10 check digit
- 18 = MSI mod-11 followed by mod-10 check digit
- 19 = CODE 39 with no check digit
- 20 = CODE 39 with mod-43 check digit
- 21 = Interleaved two of five with no check digit
- 22 = Identicon two of five with no checksum
- 23 = Code 128 (manual mode change)
- 24 = Code 128 (auto mode change)
- 25 = Code 11 with only 'c'
- 26 = Code 11 with both 'c' and 'k' checksum
- 27 = Code 93 with both 'c' and 'k' checksum
- 28 = CODABAR with no check digit (narrow gap)
- 29 = Matrix two of five with no checksum
- 31 = Industrial two of five with no checksum
- 32 = Ames with no checksum
- 33 = Delta distance 'a' (IBM) with no checksum
- 34 = Delta distance 'a' (IBM) with no checksum
- 35 = EAN 8 with two-digit supplement
- 36 = EAN 8 with five-digit supplement
- 37 = EAN 13 with two-digit supplement
- 38 = EAN 13 with five-digit supplement
- 39 = POSTNET
- 40 = FIM
- 41 = Interleaved two of five with check digit
- 42 = UPC/EAN128

43 = CUSTOM CODE (Japan)¹
 44 = CODABAR with no check digit (wide gap)

flag:

Y = Print the string under the barcode
 N = Do not print the string

string:

Text to be converted to a barcode. The text must be enclosed with single or double quote marks.

short:

Height of short bars in barcode types that have bars in two different heights. Height of all bars in other cases. Default value is 1.52cm (0.6 inches, 180 dots). The unit of measurement is determined by the UNIT command.
 Minimum value 0.254cm (0.1 inch, 30 dots)
 Maximum value 27.9cm (11 inches, 3300 dots)

tall:

Height of tall bars in barcode types that have bars in two different heights. Default values are the same as that of the short height.

bar1/2/3/4:

Width of a bar.
 Minimum value 0.0085cm (0.0033 inches, 1 dot)
 Maximum value 1.6933cm (0.6667 inches, 200 dots)

space1/2/3/4:

The width of a space.
 The minimum and maximum values are the same as that of the bar width.

Function

The BARC command converts a string to a barcode and prints it. The specification of the type, flag and string parameters is necessary. After printing, the cursor positions at the upper left corner of the barcode.

If a value not mentioned above is specified as the barcode type, 12 EAN-13 will be automatically selected.

Specify the flag parameter to Y when printing the text under the printed barcode or N when not printing the text. The text will be printed in the font currently selected for the printing system. However, a font selection command can be written ahead of the BARC command to print the text in an arbitrary font.

Characters that can be converted

The length of the string varies depending on the type of barcode to use. Characters that can be used for the most barcode type are described in the table below.

¹ FS-600/FS-7000/FS-9100/FS-9500 only

Table 20 Characters that can be used

Type	Max	Characters to be used	Default
0	11	0123456789	0
1	13	0123456789	0
2	16	0123456789	0
3	13	0123456789	0
4	18	0123456789	0
5	22	0123456789	0
6	25	0123456789	0
7	29	0123456789	0
8	6	0123456789	0
9	8	0123456789	0
10	11	0123456789	0
11	7	0123456789	0
12	12	0123456789	0
13	13	0123456789	0
14	15	0123456789	0
15	1 - 14	0123456789	0
16	1 - 14	0123456789	0
17	1 - 14	0123456789	0
18	1 - 14	0123456789	0
19	1 - 40	\$%+-. /0123456789ABCDEFGHIJKLMNOPQRSTUVWXYZ and spaces	space
20	1 - 40	\$%+-. /0123456789ABCDEFGHIJKLMNOPQRSTUVWXYZ and spaces	space
21	2 - 26	0123456789	0
22	1 - 25	0123456789	0
23	1 - 40	All (ASCII characters 32 - 126)	space
24	2 - 40	All (ASCII characters 32 - 126)	space
25	1 - 45	-0123456789	space
26	1 - 45	-0123456789	space
27	1 - 50	All (ASCII characters 32 - 126)	space
28	3 - 32	\$+-. /0123456789:abcdetn*	-
29	1 - 25	0123456789	0
30	1 - 25	0123456789	0
31	1 - 25	0123456789	0
32	1 - 25	-0123456789	0
33	1 - 25	0123456789KLMO	0
34	1 - 25	0123456789KLMO	0
35	9	0123456789	0
36	12	0123456789	0
37	14	0123456789	0
38	17	0123456789	0
39	139	0123456789	-

Type	Max	Characters to be used	Default
40	1	The following FIM pattern type A. Country Reply Mail with POSTNET B. Business Reply, Penalty and Franked Mail without POSTNET C. Business Reply, Penalty and Franked Mail with POSTNET D. OCR Readable Mail without POSTNET	0
41	125	0123456789	0
42	80	All (ASCII characters 32 - 126)	
43	1 - 20	0123456789ABCDEFGHIJKLMNPOQRSTUVWXYZ	-
44	3 - 32	\$+-. /0123456789:abcdetn* and spaces	-

Length of the bars

For the UPC and EAN (JAN) barcode type, the length of the short and tall bars can be adjusted. When adjusting the length of the tall or short bars, specify that of both tall and short bars. If the length of only the tall or short bars is adjusted, the bars' length does not change. It is recommended to use the dot for the unit of measurement in order to avoid the translation error by the barcode reader. In the example below, the length of the short bar is specified as 1.5cm and the tall bar as 1.7cm.

```
!R! UNIT D; BARC 8, Y, '123456', 180, 200; PAGE; EXIT;
```



Width and space of the bar

It is possible to adjust the widths of the bars and space between the bars. When adjusting these widths, specify all of the eight width and space parameters. Specify the bar 1 to bar 4 in the ascending order. The values can be specified between 1 and 200 dots.

```
1dot ≤ bar1 < bar2 < bar3 < bar4 ≤ 200dots  
1dot ≤ space1 < space2 < space3 < space4 ≤ 200dots
```

It is recommended to use the dot for the unit of width in order to avoid the translation error by the barcode reader. The example below specifies the width of the first bar and space as 5 dots and that of the remaining bars as 10 dots.

```
!R! UNIT D;  
BARC 19, Y, '0123ABC', 60, 60, 5, 10, 10, 10, 10, 5, 10, 10;  
PAGE;  
EXIT;
```



BKLT — print in BookLeT binding

Format

BKLT *binding* ;

Parameter

binding:

number representing binding mode:

0 = None


1 = Left-to-right binding

2 = Right-to-left binding

Function

The BKLT command allows printing a job as a booklet, reordering and printing the pages two-up on a sheet of paper. Booklet printing requires the hard disk to be installed in the printing system.


a = First page
b = Last page



The job to booklet-print must be defined using the JOBS command. Used together with the FOLD command, the paper can be folded in the middle to form a booklet.

Booklet printing is possible for documents of A4, A5, B5, letter, statement, or 16K size documents and are automatically printed on double size paper, namely A3, A4-R, B4, ledger, letter-R, or 8K. The original page size must be specified by the SPSZ command, for example, SPSZ 8; for A4 page size. It is also possible to reduce the original page size by commanding the second parameter for SPSZ. For example, to print two A4 pages on a A4 sheet, command SPSZ 8, 5;. For details on page size, refer to the SPSZ command page in this manual. The document must have the same page size for all pages. If the double size paper for the document is not available in the printing system, the printing system ignores the BKLT command and prints in the original paper size. Duplex printing is also possible by adding the DUPX command. To correctly arrange two neighboring pages in line, short-edge binding duplex must be used by commanding DUPX 2;.

The order of print pages can be selected by *binding* in either of the following manners:



This command is not supported by some photocopier models.

File

The following command sequence prints a copy of a 12-page A4 document in booklet mode, folding the stack in the middle. The printing system will require A4-R paper to be loaded.

```
!R! RES; SPSZ 8, 5;

UNIT C;
BKLT 1;
DUPX 2;
FOLD;
SFNT 'Helvetica-Bd', 72;
JOBS 0, 1, 0, 0;
MZP 4, 5; TEXT "Page 1";
PAGE;
MZP 4, 5; TEXT "Page 2";
PAGE;
MZP 4, 5; TEXT "Page 3";
PAGE;
MZP 4, 5; TEXT "Page 4";
PAGE;
MZP 4, 5; TEXT "Page 5";
PAGE;
MZP 4, 5; TEXT "Page 6";
PAGE;
MZP 4, 5; TEXT "Page 7";
PAGE;
MZP 4, 5; TEXT "Page 8";
PAGE;
MZP 4, 5; TEXT "Page 9";
PAGE;
MZP 4, 5; TEXT "Page 10";
PAGE;
MZP 4, 5; TEXT "Page 11";
PAGE;
MZP 4, 5; TEXT "Page 12";
PAGE;
JOBT;
EXIT;
```

BLK — draw filled-in BLock

Format

BLK *width*, *depth*[, *option*];

Parameters

width:


width of the block

depth:

depth of the block

option:

- B = Leave the cursor unmoved (Default)
- H = Move the cursor to the horizontally adjacent corner
- V = Move the cursor to the vertically adjacent corner
- E = Move the cursor to the diagonally opposite corner
- L = Move the cursor one line down
- N = Move the cursor to the beginning of the next line




Function

The BLK command draws a block of a specified size with one corner at the current cursor position. The block is filled in with the pattern designated by the previous PAT, FPAT, or GPAT command. The cursor is moved as specified by the option. If the option is omitted, the cursor is not moved (same as option *B*).

File

```
!R! RES; UNIT C;  
MAP 5, 5;  
PAT 20;  
BOX 4, 1; BLK 4, 1;  
MRP 0, 1.5;  
BOX 4, 1; BLK 4, 1;  
MRP .9, .6;  
SFNT "Helvetica-Nr";  
TEXT 'OVERPRINTING';  
MRP 0, 5;  
PAT 29; BOX 1, -4; BLK 1, -4, H;  
PAT 39; BOX 1, -3; BLK 1, -3, H;  
PAT 49; BOX 1, -2; BLK 1, -2, H;  
PAT 59; BOX 1, -1; BLK 1, -1, H;  
PAGE;  
EXIT;
```

Printout

BOX — draw BOX

Format

BOX *width*, *depth*[, *option*];

Parameters

width:


width of the box

depth:

depth of the box

option:

- B = Leave the cursor unmoved (Default)
- H = Move the cursor to the horizontally adjacent corner
- V = Move the cursor to the vertically adjacent corner
- E = Move the cursor to the diagonally opposite corner
- L = Move the cursor one line down
- N = Move the cursor to the beginning of the next line



Function

The BOX command draws a box of a specified size with one corner at the current cursor position. The line thickness is designated by the SPD (set pen diameter) command. The size and thickness is measured on the basis of value established in the UNIT command. The cursor is moved as specified by the option. If the option is omitted, the cursor is not moved (same as option B).

The BOX command draws only the outline of the box. To fill in the box, use the BLK command.

The width and depth may use either a positive or negative value.


File

```
!R! RES; UNIT C; SPD .1;  
SLM 2;
```

```
MAP 4, 0;  
SCP;  
SFNT "Helvetica-Nr", 10;  
BOX 4, 2, L; TEXT 'Text printed with option L';  
RPP;  
BOX 4, 2, N; TEXT 'Text printed with option N';  
PAGE;  
EXIT;
```

Printout

Text printed with option N



CALL — CALL macro

Format

CALL[*repeat-count*] *macro-name*, *parameter-value*, . . . ;

Parameters

repeat-count:

number of times the macro is to be executed

macro-name:

name defined by a MCRO command

parameter-values:

values given to the dummy parameters in the macro definition

Function

The CALL command executes a macro previously defined by a MCRO command. The macro may be executed any desired number of times by specifying a repeat count. Note that the repeat count is not followed by a comma. If the repeat count is omitted, the macro is executed once.

Only the first four characters of the macro name are significant. They must match the first four characters of a macro name defined by a previous MCRO command.

A parameter value consists of the characters that occur between the delimiting commas, excluding the commas. These parameters are substituted without change for the corresponding dummy parameter in the macro definition, except that consecutive spaces are compressed to a single space.




When a character-string parameter is enclosed in apostrophes or quotation marks, the enclosed string together with the enclosing apostrophes or quotation marks is substituted for the dummy parameter. In this case, consecutive spaces are not compressed.

If a CALL command has too few parameters, it is still valid. The macro is partially executed; any commands in the macro body without appropriate parameters will be ignored.

The CALL command, like other commands, is limited to 255 characters in length.

File

```
!R! RES; UNIT C; SLS .5;
CMNT Requires BX and SCREEN macros;
CMNT See ENDM command page;
MAP 9, 3; CALL BX;
MAP 7, 4; CALL 5 BX;
MAP 5, 5; CALL 9 BX;
MAP 8, 6; CALL SCREEN, 2.2, 1.7;
MAP 8.6, 7;
TEXT 'Narrow', L;
TEXT 'screen';
MAP 5,9; CALL SCREEN, 8.2,1.2;
MAP 8,10;
TEXT 'Wide screen';
PAGE;
EXIT;
```

Printout

CCPY — Carbon CoPY

Format

CCPY mode ;

Parameter

mode:

E = enable carbon copy

D = disable carbon copy

Function

The CCPY command continuously executes different automacros using different paper sources, enabling to apply the different form overlay using the different paper type to a document to be carbon-copied, similarly to hand-writing carbon copy forms. The CCPY statement should be terminated with the ENDC command.

The macro and paper source for each form should be described following the CCPY command in a separate line using the PSRC and AMCR command, and terminated by the TERM command. For example, the second line in the following example is applied to the first page of the document and executes the CROPMARK macro printing the page on the paper fed from cassette #2. The third line applies to the second page of the document, and so on. It is possible to add as many lines as required to match the number of pages of the document to print.

```
CCPY E ;
    PSRC 2 ; AMCR E, CROPMARK ; TERM ;
    PSRC 3 ; AMCR E, SIGNATURE ; TERM ;
    PSRC 4 ; AMCR E, COLOPHON ; TERM ;
ENDC ;
```

During the carbon copy is executed, the copy count (COPY command) is forced to be 1. To apply no overlay to a specific page, describe only the TERM command in the line corresponding to the page. Note that the same paper source as the previous line is applied to that line. For example:

```
CCPY E ;
    PSRC 2 ; AMCR E, CROPMARK ; TERM ;
    TERM ;
    PSRC 4 ; AMCR E, COLOPHON ; TERM ;
ENDC ;
```

In the example above, the second page is printed using cassette #2 with no form overlay.

In carbon copy, duplex printing is not possible. The paper size and the page orientation should be consistent for all pages in the document to be carbon copied. Otherwise, the carbon copy is cancelled.

To terminate carbon copy, send CCPY D to the printing system.

File

```
!R!  
RES; DAM; CCPY D; UNIT C;  
  
MCRO COVER;  
UNIT C;  
SFNT 'Helvetica-Bd',14;  
TEXT 'Registration Form', L;  
SFNT 'Helvetica', 9;  
TEXT 'Please provide your personal details:';  
MRP 0, .2;  
BOX 6, 1;  
MRP 0, .35;  
TEXT ' Your Name';  
ENDM;  
  
MCRO MAIL;  
UNIT C;  
SFNT 'Helvetica-Bd',14;  
TEXT 'Mailing Information', L;  
SFNT 'Helvetica', 9;  
TEXT 'Shipping Department Use Only';  
MRP 0, .2;  
BOX 6, 1;  
MRP 0, .35;  
TEXT " Client's Name";  
ENDM;  
  
MCRO COPY;  
UNIT C;  
SFNT 'Helvetica-Bd',14;  
TEXT "Client's Copy", L;  
SFNT 'Helvetica', 9;  
TEXT 'Please retain this copy for future reference.';  
MRP 0, .2;  
BOX 6, 1;  
MRP 0, .35;  
TEXT ' Registered Client';  
ENDM;  
  
CCPY E;  
PSRC 1; AMCR E,COVER; TERM;  
PSRC 0; AMCR E,MAIL; TERM;  
PSRC 1; AMCR E,COPY; TERM;  
ENDC;  
  
EXIT;  
  
!R! CMNT Document to be carbon-copied;  
SFNT 'Helvetica', 12;  
MZP 0, 1.9;  
TEXT ' David Sumimoto';  
PAGE;  
EXIT;
```

Printout

Client's Copy

Please retain this copy for future reference.

Registered Client David Sumimoto

Mailing Information

Shipping Department Use Only

Client's Name David Sumimoto

Registration Form

Please provide your personal details:

Your Name David Sumimoto

CDSK — Check hard DiSK

Format

CDSK ;

Parameter

None

Function

The CDSK command verifies data in the hard disk (garbage collection). It deletes garbage jobs such as page image data that were accidentally generated in the hard disk and normally inaccessible for deletion.

File

!R! CDSK ; EXIT ;

CID — Configure color-Image Data

COLOR

Format

CID *color-space* , *encoding* ;

Parameters

color-space:

- 0 = Device-dependent RGB (Default)
- 1 = Device-dependent CMY
- 2 = Void
- 3 = CIE L*a*b

encoding:

- Pixel encoding
- 1 = Indexed by Pixel (Default)
- 3 = Direct by Pixel

Function

The CID command creates a new color palette using the ID which is assigned to the currently active palette. Note that the currently active palette is deleted by this command.

It is possible to add and change colors on this palette. However, the currently active colors do not change even though a new palette is created with this command.

The same colors as simple colors are defined for index numbers 0 through 7 when the RGB color space is specified. The color assignments for these indices can be changed using the ACLI command.

Indexed by Pixel encoding assigns colors onto the palette. This is used for rendering text and vector objects. Direct by Pixel encoding assigns colors directly to pixels. This is used for bitmap images.

Changing a color palette does not affect color print quality on output pages.

File

```
!R! RES;
DAM;
UNIT C;
SFNT 'Helvetica-Bd', 20;

CPAL C, 1; SPAL 1;

CID 0, 1; CMNT RGB Color Space used;
ACLI 1, 0, 0, 255;

MZP 1, 1; SCOL 1;
TEXT 'An RGB color palette is used.'

CID 3, 1; CMNT L*a*b Color Space used;
ACLI 2, 100, 75, 0;

MZP 1, 2; SCOL 2;
TEXT 'An L*a*b color palette is used.';

PAGE;
EXIT;
```

Printout

An RGB color palette is used.
An L*a*b color palette is used.

CIR — draw CIRcle

Format

CIR *radius* ;

Parameter

radius:

radius of the circle

Function


In standard graphics mode, the CIR command draws a circle centered at the current cursor position. The line thickness used is designated by the SPD command.

A circle can also be drawn by using the path mode graphics. Refer to the PARC command.

File

```
!R! RES; UNIT C; SLS .5;
SPD .01; MAP 10, 1;
CIR .1; MRP 0, .1;
CIR .2; MRP 0, .2;
CIR .4; MRP 0, .4;
CIR .8; MRP 0, .8;
SPD .02;
CIR 1.6; MRP 0, 1.6;
CIR 3.2; MRP 0, 3.2;
SPD .2;
CIR .1; MRP .2, .4;
SFNT "Helvetica-Nr"; TEXT 'Cursor now here';
PAGE;
EXIT;
```

Printout



CLIP — CLIP current path

Format

```
CLIP [ clip-mode ] ;
```

Parameter

clip-mode:

- 1: EO clip
- 2: Non-zero winding clip

Function

The CLIP command sets the current path as the clipped path. After this command is issued, no painting is done outside the area of the path. Executing this command does not delete the current path.

The RES command initializes paths to a state with no clip area.

CLPR — CLIP Rectangular area

Format

```
CLPR [x1, y1, x2, y2];
```

Parameters

x1:

x coordinate of the top-left corner of the rectangle

y1:

y coordinate of the top-left corner of the rectangle

x2:


x coordinate of the bottom-right corner of the rectangle

y2:

y coordinate of the bottom-right corner of the rectangle

Function

The CLPR command describes a rectangular path which intersects the inside of the texts, current path, or raster graphics objects. The clipping rectangle does not clip those graphics segments drawn in the standard mode graphics. Any painting operations outside the specified rectangle is erased and those segments within it are painted on the page.




The coordinates are measured from the top and left edge limits in the units designated by the UNIT command. If the parameters are omitted, the rectangle is equal to the size of the current page size. If an attempt is made to position the *top-left* corner to the right of the *bottom-left* corner [greater values for $(x1, y1)$ than for $(x2, y2)$], the printing system exchanges the coordinate values with each other.

The printing system resets the size of the rectangular clipping path to be equal to the maximum printable area of the current page size when a PSRC, RES, SPO, or SPSZ (Set Page SiZe) command is encountered.

File

```
!R! RES; UNIT C;  
NEWP;  
SPD 1;  
PMRA 6, 9, 3, 0;  
PAT 1;  
PARC 6, 9, 3, 0, 360;  
CLPR 3, 6, 9, 12;  
STRK;  
PAGE;  
EXIT;
```

Printout



CLSP — CLoSe Path

Format

```
CLSP ;
```

Parameter

None

Function

The CLSP command closes an open path by adding a line segment extending from the end of the path to the beginning of the path.

Simply ending a path at the path's starting point does not make it a closed path; such a path remains open until it is closed. Closure can be made either explicitly, by means of the CLSP command, or implicitly by means of the FILL (FILL current path) command.

File


```
!R! RES; UNIT C;

NEWP;
CMNT The 1 st path below is stroked without closing;
SPD .25;
PMZP 5, 10;
PMRA 5, 10, 1, 90;
PARC 5, 10, 1, 90, 270;
PARC 7, 10, 1, 270, 90;
STRK;

NEWP;
CMNT Next path is closed by CLSP before stroked;
PMZP 5, 14;
PMRA 5, 14, 1, 90;
PARC 5, 14, 1, 90, 270;
PARC 7, 14, 1, 270, 90;
CLSP;
STRK;

PAGE; EXIT;
```

Printout



Related Command

NEWP

CMNT — CoMmeNT

Format

```
CMNT string;
```

Parameter

string:

any character string

Function

The CMNT command adds a character string, typically a comment explaining the purpose of the following PRESCRIBE commands. The CMNT command statement will be ignored by the printing system.

Although character strings are normally enclosed in apostrophes or quotation marks, this is unnecessary for comment strings as long as the string does not itself contain an apostrophe, quotation mark, or semicolon.

File

```
!R! RES;  
SLM 2; STM 2;  
SFNT 'TimesNewRoman'; CMNT TimesNewRoman;  
EXIT;  
The CMNT command in the above program reminds  
you that SFNT represents the TimesNewRoman font.
```

Printout

The CMNT command in the above program reminds you that SFNT represents the TimesNewRoman font.

CMOD — Color MODE

COLOR

Format

CMOD *mode* ;

Parameter

mode:

color mode:
0 = Monochrome
1 = Color (Default)

Function

The CMOD command switches between monochrome and color modes. In monochrome mode, other color commands are still enabled, but printing is converted to monochrome or grayscale.

This command causes a page break.

COPY — set number of COPIES

Format

COPY number ;

Parameter

number:

number of copies

Function

The COPY command designates the number of copies for each page in a print job. This copy count value can be changed by sending the printing system another COPY command or by issuing a reset command. For best results, place the copy command near the start of a job or at the top of a page. If two or more copy commands are placed on the same page, the last one will be recognized.

File

```
!R! RES;  
SLM 0.5;  
UNIT P;  
SFNT 'Helvetica-Bd', 14.4;  
SLS 17;  
COPY 3;  
EXIT;  
    Announcement  
    Party Tonight at 6:00  
!R! PAGE; EXIT;
```

Printout

**Announcement
Party Tonight at 6:00**

**Announcement
Party Tonight at 6:00**

**Announcement
Party Tonight at 6:00**

CPAL — Control PALETTE

COLOR

Format

```
CPAL mode [ , palette-id ] ;
```

Parameters

mode:

palette control mode:

A = Delete all palettes

S = Delete all stacked palettes

D = Delete the palette having the specified ID

C = Copy the active palette to the specified ID

palette-id:

number from 0 to 32767 (Default is 0)

Function

The CPAL command controls the palette identified by *palette-id* in the way described by the control mode. If *palette-id* is not specified, the command is applied to palette of ID 0. No check is made whether the palette ID number given already exists for another palette. The palette ID number is reset to 0 by RES command.

Mode *A* deletes all existing palettes and brings the default palette back to the default state (ID 0). Note that non-active palettes are not deleted. Also, *palette-id* is not used when *mode* is *A*.

Mode *S* deletes all palettes that are not active. The active palette is not affected. Also, any ID specified by *palette-id* is not used.

Mode *D* deletes the palette having the ID specified by *palette-id* regardless of being active or inactive. If the palette deleted was active, then the current palette will be a monochrome palette having the same ID as that of the deleted palette.

Mode *C* copies the currently active palette to a new palette with the new ID specified by *palette-id*. If the palette having the same ID already exists, that palette is overwritten.

Deleting or copying a color palette does not affect color print quality on output pages.

CPTH — Character PaTH

Format

```
CPTH 'string' ;
```

Parameter

string:

any character string except from ASCII code 31 (hex 1F) and lower

Function

The CPTH command constructs a path which adds the outlines of characters to the current path. When stroked or filled, it prints the given *string* at the current cursor position and moves the cursor to the end of the string. The current font preceding this command must be a resident or a soft scalable font.

The character path can be filled in with the pattern or shade of gray designated by the previous PAT, FPAT, or GPAT command.

File

```
!R! FSET 1p72v1s0b4148T; CMNT Univers-MdIt;  
NEWP;  
PMZP 2, 2;  
CPTH 'Enjoy' ;  
STRK;  
PAGE;  
EXIT;
```

Printout



The word "Enjoy" is displayed in a large, outlined, serif font. The letters are hollow, showing the internal structure of the font. The word is centered on the page.

CSET — Change symbol SET by symbol-set ID

Format

CSET *symbol-set-id* ;

Parameter

symbol-set-id:

symbol set ID:

0D = ISO 60 Norway
0E = Roman Extension
0F = ISO 25 French
0G = HP German
0I = ISO 15 Italian
0K = ISO 14 JIS ASCII
0N = ECMA-94 Latin 1
0S = ISO 11 Swedish
0U = ISO 6 ASCII
1D = ISO 61 Norwegian
1E = ISO 4 U.K.
1F = ISO 69 France
1G = ISO 21 Germany
1S = ISO 15 Spanish
1U = US Legal
2K = ISO 57 Chinese
2N = ISO Latin 2
2S = ISO 17 Spain
2U = ISO 2 IRV
3R = PC Cyrillic*
3S = ISO 10 Swedish
4S = ISO 16 Portuguese
4U = Roman-9
5M = PS Math
5N = ISO Latin 5
5S = ISO 84 Portuguese
5T = ISO Windows Latin 5
6J = MS Publishing
6M = Ventura Math
6N = ISO8859/10 Latin 6
6S = ISO 85 Spanish
7J = Desktop
8G = Greek-8
8M = Math 8
8Q = PC Set 1 IBM
8T = Turkish-8
8U = Roman-8
9E = Windows Latin 2
9G = Windows 3.1 Latin/Greek
9J = PC-1004
9L = Ventura ITC Zapf Dingbats
9N = ISO Latin 9
9Q = PC Extension

9R = Windows 3.1 Latin/Cyrillic
9T = PC-Turkish
9U = Windows 3.0
10G = PC-851 Latin/Greek
10J = PS Text
10N = ISO8859/5 Latin/Cyrillic
10R = PC-855 Cyrillic/Serbia
10U = IBM PC-8
11G = PC-869 Latin/Greek
11U = IBM PC-8 Danish/Norwegian
12G = PC-8 Latin/Greek
12J = Macintosh
12N = ISO8859/7 Latin/Greek
12R = USSR-GOST
12U = IBM PC-850
13P = ABICOMP Brazil/Portugal
13R = PC-8 Bulgarian
13U = PC-858
14G = Greek Alternate 437G
14L = PS ITC Zapf Dingbats¹
14P = ABICOMP International
14R = PC Ukrainian
15U = Pi Font
16U = PC-857
17U = PC 852
18N = ISO-10646
18U = PC-853 Turkish/Latin 3
19L = Windows 3.1 Baltic[†]
19M = Symbol (Symbol-Set-SWM font only)
19U = Windows Latin 1
20U = PC-860 Portugal[†]
21U = PC-861 Iceland[†]
23U = PC-863 Canada-French[†]
24Q = PC-8 Polish Mazovia[†]
25U = PC-865 Norway[†]
26U = PC-775 Baltic[†]
27Q = PC-8 PC Nova[†]
579L = Windings (More Wingbats-RegularSWM font only)

¹ Available only with models having Agfa fonts

Function

'Symbol set' refers to a specific group of alphabetic, numeric, and symbolic characters. The CSET command summons the symbol set to suit the various languages with the current font. The factory default symbol set is 0U (US ASCII).

When the FSET command is used to select a font, the printing system selects a font based on a prioritization of the font characteristics. Among the given characteristics, the symbol set has the highest priority in the font selection process. Refer to the FSET command page.

The available symbol sets depend on the printing system model and the fonts. If the specified symbol set does not exist, the symbol set remains unchanged.

Files

```
!R! RES;
SFNT 'Helvetica', 24;
SLPI 3;
CMNT Current symbol set is ISO 6: ASCII;
TEXT '@ABCDEFG', N;
CSET 5M; CMNT Changed to PS math;
TEXT '@ABCDEFG';
PAGE;
EXIT;
```

Printout

@ ABCDEFG
 ≅ ABXΔEΦΓ

CSTK — select Collator STacker

Format

CSTK [*tray, tray, ...*]; or CSTK 0;

Parameter

tray:

number from 1 to the maximum (bottom)

Function

The CSTK command is effective only when the optional sorter is installed to the printing system. It instructs the printing system to deliver printed pages to the sorter in the collator mode. Each tray specified receives all copies of one page. The copies of the first page go to the first tray, the copies of the second page to the second tray, and so on. If the tray parameter is omitted, the trays are used in order from top to bottom.

The copy count must be selected with the COPY command. The maximum copy count is the maximum capacity of the sorter trays (which varies from sorter model to model).

Upon receiving the CSTK command, the printing system automatically performs a form feed.

The tray parameter of 0 (CSTK 0;) should be used only to deselect the collator mode and to receive the pages in the sorter's bulk tray (if available).



Note

This command is irrelevant to certain combination of photocopier models and option finisher models.

File

```
!R! RES; CSTK 1, 2; COPY 2;  
EXIT;  
Page 1 (2 copies, tray 1) !R! PAGE; EXIT;  
Page 2 (2 copies, tray 2) !R! PAGE; EXIT;  
Page 3 (2 copies, tray 1) !R! PAGE; EXIT;  
Page 4 (2 copies, tray 2) !R! PAGE; EXIT;
```

CTXT — print Centered TeXT

Format

```
CTXT [ , string ][ , option ][ , U-option ] ;
```

Parameters

string:

any character string

option:

B = Leave cursor at center of string
 E = Move cursor at end of string
 L = Move cursor down one line
 N = Move cursor to beginning of next line

U-option:

Underline text

Function

The CTXT command prints the given string centered at the cursor location, then moves the cursor as designated by the option. The string must be enclosed in either single or double quotation marks. (If the string includes one kind of quotation marks, use the other kind to enclose it.) If the string is omitted, nothing is printed. If the option is omitted, the cursor is not moved. If the U-option is specified, the text is underlined. The underline thickness and location can be set by the Sulp command.

The string is not centered between the right and left margins unless the cursor is located halfway between the margins. Before the CTXT command, the cursor should be moved to the desired center position by a command such as MAP, MRP, or PMRP.

The length of the string is unlimited, except that the command as a whole may not be longer than 255 characters. Carriage-return and linefeed codes in the string are ignored.

File

```
!R! RES; SLPI 3; SFNT "Helvetica-Bd", 16;
MAP 4, 1;
CTXT 'NEWS!', L, U;
SFNT "TimesNewRoman-It", 16;
CTXT 'Unforgettable Holidays in Sydney', L;
CTXT 'For 6 Days Only at $1000!';
PAGE;
EXIT;
```

Printout

NEWS!

*Unforgettable Holidays in Sydney
For 6 Days Only at \$1000!*

DAF — Delete All Fonts

Format

DAF;

Parameter

None

Function

The DAF command deletes all nonresident fonts and the font numbers assigned to scalable fonts by the SFNT command from the printing system's memory. This enables the user to make room for new fonts, macros, and graphics.

File

```
!R! UNIT P; SLS 16;
SFNT 'Helvetica-Bd', 15, 2000, 277, 2, 0;
SFNT 'Helvetica', 15, 2001, 277, 1, .5;

FONT 10; TEXT '1. FAREWELL TO ', E;
FONT 2000; TEXT 'FONT 2000', N;
FONT 2001; TEXT 'AND FONT 2001', N;
TEXT, N;

DAF;
FONT 10; TEXT '2. FAREWELL TO ', E;
FONT 2000; TEXT 'FONT 2000', N;
FONT 2001; TEXT 'AND FONT 2001';

PAGE;
EXIT;
```

Printout

1. FAREWELL TO FONT 2000
AND FONT 2001

2. FAREWELL TO FONT 2000
AND FONT 2001

DAM — Delete All Macros

Format

DAM;

Parameter

None

Function

The DAM command deletes all macro definitions made by using the PRESCRIBE MCRO command from the printing system's memory. This enables the user to make room for new macros, fonts, or graphics.

File

```
!R! RES; UNIT C;
CMNT Print ENDM file first.;
SFNT "Helvetica-Bd";
MAP 5, 4; CALL SCREEN, 9.2, 1.7;
MAP 7, 5; TEXT 'This text is enclosed by', L;
          TEXT 'SCREEN macro.';

DAM;
MAP 5, 6; CALL 10 BX;
MAP 5, 7; CALL SCREEN, 9.2, 1.7;
MAP 6, 8; TEXT 'SCREEN macro was deleted by DAM.';
PAGE;
EXIT;
```

Printout



**This text is enclosed by
SCREEN macro.**

SCREEN macro was deleted by DAM.

DAP — Draw to Absolute Position

Format

DAP *x-coordinate* , *y-coordinate* ;

Parameters

x-coordinate:

distance from the left margin

y-coordinate:

distance from the top margin

Function

The DAP command draws a line from the current cursor position to a position specified relative to the top and left margins. The thickness of the line is designated by the SPD (set pen diameter) command. The coordinates and line thickness are measured in the units designated by the UNIT command.


The coordinates may be positive or negative. Negative coordinates cause the line to extend outside the margin.

The DAP command is ignored in the path mode graphics.

File

```
!R! RES;
  UNIT C;
  CMNT Tree Diagram;
  STM 8; SLM 8;
  MAP 0, 0; DAP 2, 0.5;
  MAP 0, 1; DAP 2, 0.5;
  MAP 0, 1; DAP 2, 1.5;
  MAP 0, 2; DAP 2, 1.5;
  MAP 0, 2; DAP 2, 2.5;
  MAP 0, 3; DAP 2, 2.5;
  MAP 2, 0.5; DAP 4, 1;
  MAP 2, 1.5; DAP 4, 1;
  MAP 2, 1.5; DAP 4, 2;
  MAP 2, 2.5; DAP 4, 2;
  MAP 4, 1; DAP 6, 1.5;
  MAP 4, 2; DAP 6, 1.5;
PAGE; EXIT;
```

Printout



DELFF — DELEte Font

Format

```
DELFF font-number ;
```

Parameter

font-number:

number identifying the font to be deleted

Function

The DELFF command deletes any designated non-resident font from the printing system's memory. This enables the user to make room for new fonts, macros, and graphics.

The printing system's memory is large but not unlimited, and judicious use of the DELFF command may prevent memory overflow. (If a memory overflow occurs, the printing system gives an overflow warning message.) The amount of memory still available can be viewed on a status page, available by executing the STAT command.

File

```
!R! UNIT P; SLS 16;
SFNT 'Helvetica-Bd', 15, 2000, 277, 1.5, 0;
SFNT 'Helvetica', 15, 2001, 277, 1, .5;

FONT 2000; TEXT '1. FAREWELL TO ', E;
FONT 2001; TEXT 'ITALICS', N;
TEXT, N;

DELFF 2001;

FONT 2000; TEXT '2. FAREWELL TO ', E;
FONT 2001; TEXT 'ITALICS';

PAGE;
EXIT;
```

Printout

1. FAREWELL TO *ITALICS*

2. FAREWELL TO ITALICS

DELM — DElete Macro

Format

DELM *macro-name* ;

Parameter

macro-name:

name defined by a previous MCRO command

Function

The DELM command deletes a specified macro definition from the printing system's memory to make room for new macros, fonts, or graphics, or so that the same name can be used for another macro.

File

```

!R! UNIT C;
MCRO BX;
SPD .05; BOX .8, -.8; MRP 1, 0;
ENDM;
MCRO SCREEN %, width, depth;
CMNT Draws box with rounded corners;
CMNT Leaves cursor at previous position;
SPD .08;
SCP;
MRP .25, .25; ARC .23, .30, 270, 0;
MRP 0, -.25; DRP %1, 0;
MRP 0, .25; ARC .23, .30, 0, 90;
MRP .27, 0; DRP 0, %2;
MRP -.27, 0; ARC .23, .30, 90, 180;
RPP;
MRP 0, .25; DRP 0, %2;
MRP .25, 0; ARC .23, .30, 180, 270;
MRP 0, .27; DRP %1, 0;
RPP;
ENDM;
SFNT "Helvetica-Bd";
MAP 5, 4;
CALL SCREEN, 9.2, 1.7;
MAP 7, 5;
TEXT 'This text is enclosed by', L;
Text 'SCREEN macro.';
DELM SCREEN;
MAP 5,8;
CALL SCREEN, 9.2, 1.7;
MAP 6, 10;
TEXT 'SCREEN macro was deleted by DELM.';
PAGE;
EXIT;

```

Printout

**This text is enclosed by
SCREEN macro.**

SCREEN macro was deleted by DELM.

DPAT — select Dashed PATtern

Format

DPAT *number* ;

Parameter

number:

number from 1 to 20 (11 to 20 are for user-defined patterns.)

Function

The DPAT command specifies the pattern of alternating black and white dashes to stroke the current path.


The resident patterns range in value between 1 and 10. Refer to resident patterns list on page 79. Numbers 11 to 20 are reserved for use with defining particular patterns using the SDP (Store Dash Pattern) command. If you specify a value between 11 to 20 when no user pattern has been defined, the path will be stroked in solid black.

The DPAT command does not affect the line drawn by the standard graphics commands such as DRP, BOX, ARC, etc.

File

```
!R! RES; UNIT C;
MCRO DPAT;
NEWP;
SPD .2;
SCP;
SFNT "Helvetica-Nr", 10;
TEXT 'DPAT %1';
PMRP 1.4, -.1; DPAT %1; PDRP 9, 0;
RPP; PMRP 0, .7;
STRK;
ENDM;
MZP 5, 5;
FONT 13;
CALL DPAT,1;
CALL DPAT,2;
CALL DPAT,3;
CALL DPAT,4;
CALL DPAT,5;
CALL DPAT,6;
CALL DPAT,7;
CALL DPAT,8;
CALL DPAT,9;
CALL DPAT,10;
PAGE;
EXIT;
```


Printout



Related Command

SDP

DRP — Draw to Relative Position

Format

DRP *x-coordinate* , *y-coordinate* , . . . ;

Parameter

x-coordinate/y-coordinate:

coordinates as measured from the current cursor position

Function

The DRP command draws a line from the current cursor position to a specified relative position. The thickness of the line is designated by the SPD (set pen diameter) command. The coordinates and line thickness are measured by the value designated by the UNIT command.


Multiple pairs of coordinates may be specified. The printing system then draws a connected series of line segments, proceeding from one point to the next. For example, DRP 1, 2, 3, 4; is equivalent to DRP 1, 2; DRP 3, 4;. The coordinates may be positive or negative.

The DRP command is ignored in the path mode graphics. To add a line segment to a path, use the PDRP (in Path, Draw to Relative Position) command.

File

```
!R! RES;  
UNIT C; SPD .3;  
CMNT Cube;  
MZP 5, 10;  
DRP 4, 0, 2, -2, 0, -4, -4, 0, -2, 2, 0, 4;  
MRP 4, 0;  
DRP 0, -4, -4, 0, 4, 0, 2, -2;  
  
PAGE;  
EXIT;
```

Printout



Related Command

PDRP

DRPA — Draw to Relative Position specified by Angle

Format

DRPA *length*, *angle*;

Parameters

length:

length of the line

angle:

angle in degrees, measured clockwise from the vertical

Function


The DRPA command draws a line of a specified length at a specified angle, starting from the current cursor position, with the thickness designated by the SPD command.

This command is ignored if placed within a path.

File

```
!R! RES;   UNIT C;  
  MAP 2, 11;  
  DRPA 2, 72;  
  DRPA 2, 0;  
  DRPA 2, 144;  
  DRPA 2, 72;  
  DRPA 2, 216;  
  DRPA 2, 144;  
  DRPA 2, 288;  
  DRPA 2, 216;  
  DRPA 2, 0;  
  DRPA 2, 288;  
  PAGE;  
EXIT;
```

Printout



DUPX — select/deselect DUPlEX mode

Format

DUPX *mode* ;

Parameter

mode:

a number representing one of the following binding mode:


0 = simplex mode

1 = duplex mode, long-edge binding

2 = duplex mode, short-edge binding

Function

The DUPX command is effective only when the option duplexer is installed to the printing system. It instructs the printing system to select or deselect duplex printing and sets the binding mode.



File/Printout

See *DXPG* on page 85.

DXPG — select Duplex PaGe side

Format

DXPG *side* ;

Parameter

side:

a number representing the side of the ensuing page:

0 = Select next side (the same as form feed)

1 = Select front side

2 = Select back side

Function

The DXPG command is effective only when the option duplexer is installed to the printing system. It ends printing on the current page in the same manner as a form feed. However, unlike the form feed, it also selects the paper side, front or back, on which printing of the ensuing page is printed.

The *side* parameter 1 prints the ensuing page begins on the front side of the next sheet, regardless of whether the current page was printed on the front or back side of the current sheet.


The *side* parameter 2 prints the ensuing page begins on the back side of the current sheet if the current page was printed on the front side, or on the back side of the next sheet if the current page was printed on the back side.

File

The following example selects the duplex mode with short-edge binding, a portrait page orientation which results in text printing on both sides of the page.

```
!R! CMNT Initiates a PRESCRIBE command sequence; RES;  
CMNT Causes a page break and resets parameters;  
DUPX 2;  
CMNT Selects duplex mode, short-edge binding;  
SPO P;  
CMNT Selects portrait orientation;  
STAK 1;  
CMNT Select face-down delivery;  
EXIT;  
This text prints right-side up on the front side of the page  
!R! DXPG 0;  
CMNT Ends current page (front side of the paper);  
EXIT;  
This text prints upside-down on the reverse side of the page  
!R! RES; EXIT;
```

Printout



DZP — Draw to Zero-relative Position

Format

DZP *x-coordinate*, *y-coordinate* ;

Parameters

x-coordinate:

distance from the left edge limit of paper

y-coordinate:

distance from the top edge limit of paper

Function

The DZP command draws a line from the current cursor position to a position specified relative to the top and left edge limits of the paper.


The thickness of the line is designated by the SPD (set pen diameter) command. The coordinates and line thickness are measured in the value designated by the UNIT command. The coordinates should be positive or zero. Negative coordinates are treated as zero.

The DZP command is ignored within a path construction. To add a zero-relative line segment in a path, use the PDZP (in Path, Draw to Zero-relative Position) command.

File

```
!R! RES;UNIT C;  
  MZP 0, 0; DZP 0, 5;  
  MZP 1, 0; DZP 0, 4;  
  MZP 2, 0; DZP 0, 3;  
  MZP 3, 0; DZP 0, 2;  
  MZP 4, 0; DZP 0, 1;  
  MZP 5, 0; DZP 0, 0;  
  PAGE;  
EXIT;
```

Printout



Related Command

PDZP

EMCR — Enable MaCRo depending on paper source

Format

```
EMCR mode , paper-source [ = print-side ] , [ repeat-count ] macro-name  
[ , parameter . . . ] ;
```

Parameters

mode:

E = enable automatic macro
D = disable automatic macro

paper-source:

0 = multi-purpose feeder
1 = cassette 1 (printing system's cassette)
2 = cassette 2
3 = cassette 3
4 = cassette 4
5 = cassette 5
6 = cassette 6 (bottom)
99 = envelope or universal feeder

print-side:

side of the page the macro is executed when duplex-printing
0 = front
1 = back

repeat-count:

number of times the macro is executed

macro-name:

name defined by the MCRO command

parameter:

values given to the dummy parameters in the macro definition

Function

The EMCR command allows separate auto macros to be assigned for each of the separate paper sources. In duplex printing, it is also possible to assign separate auto macros for both the front and back sides of a page.

The following commands are invalid within a macro:

AMCR, ASTK, PSRC, CCPY, DUPX, DXPG, JOB, JOBL, JOBO, JOBP, JOBS, JOBT, JOG, LDFC, LDFN, MID, MSTK, RDMP, SEM, SPSZ, SSTK, STAK, STAT, STPL, VMAL, VMPW

To disable the auto macro, use EMCR D. Note that EMCD D does not require other parameters such as paper-source.

ENDB — END a two-dimensional Barcode string

Format

ENDB ;

Parameter

None

Function

The ENDB command terminates a barcode data string preceded by a XBAR or XBUF command. If there is no ENDB command after data input for the XBAR and XBUF commands, a malfunction may occur. As in the case of XBAR or XBUF command, if data length is omitted, 0, negative value or non integer, data string is counted, all characters until the program sees ;ENDB;. It is necessary to have no space and line feed between ';' and ENDB.

If the XBCP 18 command is executed, the second and subsequent macro PDF417 symbols will be output. When ENDB is used immediately after XBCP 18 is executed, it is used alone, without association with XBAR or XBUF.

For details on two-dimensional barcodes, see Section 5.2. of the *Technical Reference* manual.

ENDC — END carbon Copy

Format

ENDC ;

Function

The ENDC command ends definition of the CCPY E command sequence. For details, refer to the CCPY command.

ENDD — END Dump

Format

ENDD ;

Parameter

None

Function

The ENDD command ends a hexadecimal dump initiated by the RDMP (Received-data DuMP) command and returns to the PRESCRIBE command mode.

File

See *RDMP* on page 218.

Related Command

RDMP

ENDM — END Macro

Format

ENDM;

Parameter

None




Function

The ENDM command ends a macro definition and returns to the PRESCRIBE command mode.

File

```
!R! RES; DAM; UNIT C;
MCRO BX;
SPD .05; BOX .8, -.8; MRP 1, 0;
ENDM;
MCRO SCREEN %, width, depth;
CMNT Draws box with rounded corners;
CMNT Leaves cursor at previous position;
SPD .08;
SCP; SCP;
MRP .25, .25; ARC .23, .30, 270, 0;
MRP 0, -.25; DRP %1, 0;
MRP 0, .25; ARC .23, .30, 0, 90;
MRP .27, 0; DRP 0, %2;
MRP -.27, 0; ARC .23, .30, 90, 180;
RPP;
MRP 0, .25; DRP 0, %2;
MRP .25, 0; ARC .23, .30, 180, 270;
MRP 0, .27; DRP %1, 0;
RPP;
ENDM;
PAGE;
EXIT;
```


Printout



Related Commands

CALL, DAM, DELM, MCRO

ENDR — END Raster data

Format

ENDR ;

Parameter

None

Function

The ENDR command declares the end of raster data. To declare the start of raster data, a RVRD command is used.

File

The following example prints an illustrated crown using the RVRD and ENDR.

```
!R! RES; DAM; DAF; UNIT C; SRO 1;
MAP 5, 5;
RVRD;
  5, 0, 0, 7, 0, 0;
  5, 0, 0, 24, 192, 0; 5, 0, 0, 32, 32, 0;
  5, 0, 0, 64, 16, 0; 5, 0, 0, 64, 16, 0;
  5, 0, 0, 128, 8, 0; 5, 0, 0, 128, 8, 0;
  5, 0, 0, 128, 8, 0; 5, 0, 0, 64, 16, 0;
  5, 0, 0, 64, 16, 0; 5, 0, 0, 32, 32, 0;
  5, 0, 0, 120, 192, 0; 5, 0, 0, 103, 0, 0;
  5, 0, 0, 96, 0, 0; 5, 0, 0, 144, 0, 0;
  5, 0, 0, 144, 0, 0; 5, 0, 0, 144, 0, 0;
  5, 0, 1, 8, 0, 0; 5, 0, 1, 8, 0, 0;
  5, 0, 1, 136, 0, 0; 5, 0, 2, 68, 0, 0;
  5, 0, 2, 36, 0, 0; 5, 0, 2, 36, 0, 0;
  5, 0, 6, 66, 0, 0; 5, 0, 5, 130, 0, 0;
  5, 0, 4, 6, 0, 0; 5, 0, 8, 9, 0, 0;
  5, 0, 8, 17, 0, 0; 5, 0, 14, 17, 0, 0;
  5, 0, 17, 9, 128, 0; 5, 0, 16, 134, 128, 0;
  5, 0, 16, 128, 128, 0; 5, 0, 32, 128, 64, 0;
  5, 0, 49, 0, 64, 0; 5, 0, 46, 0, 64, 0;
  5, 0, 64, 28, 32, 0; 5, 0, 64, 34, 32, 0;
  5, 0, 64, 65, 32, 0; 5, 0, 128, 65, 16, 0;
  5, 0, 184, 65, 16, 0; 5, 0, 196, 34, 16, 0;
  5, 1, 130, 28, 8, 0; 5, 1, 130, 0, 8, 0;
  5, 1, 130, 0, 120, 0; 5, 2, 68, 0, 132, 0;
  5, 14, 56, 97, 7, 0; 5, 10, 0, 145, 5, 0;
  5, 11, 255, 255, 253, 0; 5, 8, 0, 0, 1, 0;
  5, 15, 255, 255, 255, 0; 5, 4, 0, 0, 2, 0;
  5, 52, 8, 1, 2, 192; 5, 76, 8, 1, 3, 32;
  5, 76, 8, 1, 3, 32; 5, 68, 62, 7, 194, 32;
  5, 34, 8, 1, 4, 64; 5, 34, 8, 1, 4, 64;
  5, 18, 8, 1, 4, 128; 5, 13, 0, 0, 11, 0;
  5, 1, 0, 0, 8, 0; 5, 0, 128, 0, 16, 0;
  5, 0, 131, 12, 16, 0; 5, 0, 64, 144, 32, 0;
```

```
5, 0, 160, 96, 80, 0; 5, 1, 24, 1, 136, 0;  
5, 14, 134, 6, 23, 0; 5, 48, 65, 248, 32, 192;  
5, 192, 129, 8, 16, 48; 5, 0, 64, 144, 32, 0;  
5, 0, 32, 96, 64, 0; 5, 0, 16, 0, 128, 0;  
5, 0, 32, 0, 64, 0; 5, 0, 16, 0, 128, 0;  
5, 0, 8, 1, 0, 0; 5, 0, 6, 6, 0, 0;  
5, 0, 1, 104, 0, 0; 5, 0, 0, 144, 0, 0;  
ENDR;  
PAGE;  
EXIT;
```

Printout



EPL — select EcoPrint Level

Format

EPL *level*;

Parameter

level:

0 = off
2 = on

Function

The EPL command switches the Ecoprint on or off (default). Ecoprint is a standard feature of the print systems that enables the user to reduce the amount of toner placed on the page which saves on printing costs. With EPL 2;, the printing image appears lighter, yet remains readable.

The following example simulates typical Ecoprint effect:



Only one EPL level is possible for a page. If more than one EPL command are put on a page, the last one is effective.

The Ecoprint setting does not effect print speed.

File

```
!R! RES; SFNT "Helvetica-Bd", 24;  
      SLPI 3;  
      EPL 2; TEXT "Ecoprint is ON.";  
      PAGE;  
EXIT;
```

Sample

Ecoprint is ON.

EXIT — EXIT from PRESCRIBE mode

Format

```
EXIT [, E];
```

Parameter

None

Function

When the printing system receives the EXIT command, it exits from the PRESCRIBE mode and treats the following information as characters to be printed.

If the parameter E is added (EXIT, E;), the printing system ignores one carriage-return/linefeed immediately following the command.

File

```
!R! RES; SFNT "Helvetica"; EXIT;  
Commands ending in EXIT; leave a gap  
in the output,  
!R! CMNT Like this; EXIT;  
while commands ending in EXIT, E;  
!R! CMNT Like this; EXIT, E;  
leave no trace of their presence except  
possibly a shortage of lines on the page.  
!R! PAGE; EXIT;
```

Printout

Commands ending in EXIT; leave a gap
in the output,

while commands ending in EXIT, E;
leave no trace of their presence except
possibly a shortage of lines on the page.

FDIR — MP tray Feed DIRection

Format

FDIR *direction*;

Parameter

direction:

paper orientation on the MP tray

0 = Lengthwise

1 = Widthwise

Function

The FDIR command specifies the paper orientation during feeding paper on the MP tray. This is possible when the paper is A4 or letter or B5 size although this depends on the printing system model.

This command is enabled only when the MP tray's operational mode is *cassette* or *manual feed*. Printing on A4 and letter size paper can only be done widthwise in *first* mode.

This command causes a page break.

This command is irrelevant to paper cassettes and envelope feeders.

The paper orientation setting is restored to the default orientation (FRPO T8) when the printing system receives the RES command.

FILL — FILL closed path

Format

FILL *rule*;

Parameter

rule:

drawing rule:

1 = use even-odd rule

2 = use non-zero winding number rule

Function

The FILL command fills the area enclosed by the current path with the current drawing rule. It also fills all independent subpaths contained in a path. An open subpath is implicitly closed before being painted (as done by a CLSP command). This command clears the current path after filling and moves the cursor to the end of that path.

With drawing rule 1, even-odd, whether or not a point is deemed to be inside the path is determined by drawing a ray from that point in any direction and counting the number of times the ray crosses segments of the path. The point is deemed to be inside the path if the result is an odd number; if the result is an even number, the point is deemed to be outside the path.

The non-zero winding rule also draws a ray from a point in any direction to determine whether or not that point is deemed to be inside the path and examines the points at which a segment of the path crosses the ray. However, it then starts counting from zero and adds one each time a segment in the path crosses the ray from left to right; and subtract one each time a segment in the path crosses the ray from right to left. If the result of counting all the crossings is zero, the point is deemed to be outside; otherwise the point is deemed to be inside the path.

The way in which several filled-in parts are stacked with each other to produce transparent and opaque effects is controlled by the SIMG (Set IMaGe model) command.


Related Commands

FPAT, GPAT, PAT, SIMG, XPAT

File

```
!R! RES; UNIT C; SIMG 1;
CMNT Drawing clusters;
NEWP;
PMZP 10, 12; PDRP -5.7, -1.75;
PCRP 0, 0, 0, -3.4, 3.1, -3.6; CLSP; PAT 2; FILL;
PMZP 10, 12; PDRP -2.6, -5.35;
PCRP 0, 0, 2.5, -2.5, 4.9, .1; CLSP; PAT 3; FILL;
PMZP 10, 12; PDRP 5.7, -1.75;
PCRP 0, 0, 0, -3.5, -3.4, -3.5; CLSP; PAT 4; FILL;
PMZP 12, 12; PDRP 5.7, -1.75;
PCRP 0, 0, 1.5, 1.75, 0, 3.5; CLSP; PAT 5; FILL;
PMZP 10, 12; PDRP -5.7, 1.75;
PCRP 0, 0, 0, 3.4, 3.1, 3.6; CLSP; PAT 6; FILL;
PMZP 10, 12; PDRP -2.6, 5.35;
PCRP 0, 0, 2.5, 2.5, 4.9, -.1; CLSP; PAT 7; FILL;
PMZP 10, 12; PDRP 5.7, 1.75;
PCRP 0, 0, 0, 3.5, -3.4, 3.5; CLSP; PAT 8; FILL;
PMZP 10, 12; PDRP -5.7, 1.75;
PCRP 0, 0, -1.5, -1.75, 0, -3.5; CLSP; PAT 9; FILL;
PAGE;
EXIT;
```


Printout



FLAT — set FLATness

Format

FLAT *flatness*;

Parameters

flatness:

number from 1 to 100

Function


The FLAT command determines how smoothly curves in paths are to be drawn when a STRK or FILL command is used. Higher number values produce rougher curves and lower number values create smoother curves which require greater computation. The number actually specifies the maximum error tolerance of a straight line segment approximation of any portion of the curve.

The printing system resets the flatness to 1 when it receives the RES command.

File

```
!R! RES; FRPO N8,0; CMNT 300 dpi;
UNIT C; SPD 0.02;
NEWP; CMNT 1st circle (FLAT 1);
PMRA 7, 16.5, 2, 0;
PARC 7, 16.5, 2, 0, 360;
STRK;
NEWP; CMNT 2nd circle (FLAT 50);
FLAT 50;
PMRA 12, 16.5, 2, 0;
PARC 12, 16.5, 2, 0, 360;
STRK;
PAGE; EXIT;
```

Printout



FLST — print Font LiST

Format

```
FLST;
```

Parameter

None

Function

The FLST command prints a list of all resident and downloaded (user-defined) fonts in the printing system. The list provides information on each font such as font name, whether it is scalable or bitmap, font selection sequence using FSET, font ID, etc. The resident fonts are identified by the letter I (internal) in the font ID; the downloaded fonts are identified by the letter M or D depending on whether the font is stored in the memory card or the hard disk, respectively.

The font list may differ depending on models.

Related Commands

FSET, SFNT, STAT

File

```
!R! FLST; EXIT;
```


FOLD — FOLD printed pages

Format

```
FOLD [staple];
```

Parameter

staple:


1 = stapling

Function

The FOLD command is effective only when the optional document finisher having the booklet stitcher is installed to the printing system. This command is most conveniently used with the BKLT command that allows printing a job as a booklet, reordering and printing the pages 'two-up' on a sheet of paper.

The FOLD command outputs printed pages stapling and binding them together at its center. The job to be stapled and folded should be initiated by the JOBS command (See page 140).

With the *staple* value given, the pages are stapled (two) and folded at its center in reference to the short edge of the paper.



Paper sizes available for this mode include A4-R, letter-R, B4, A3 and ledger. The job is automatically delivered on the face-down output tray of the document finisher.

File

```
!R! FOLD 1;
    JOBS 1,1,2,0,0;
    TEXT 'Stapled and folded with the next page';
    PAGE;
    TEXT 'Stapled and folded with the preceding page';
    JOBT;
EXIT;
```

FONT — change current FONT

Format

```
FONT font-number;
```

Parameter

font-number:

number identifying the font

Function

The FONT command specifies a typeface for all text that follows.

The font-number should designate a resident bitmap font, a scalable font selected by the SFNT command, or a nonresident font generated by the LDFC command or downloaded from the computer. If an undefined font-number is specified, the font will not change. To print a font list, use the FLST command.

For a bitmap font, depending on the font mode (FTMD) value, the FONT command also sets the line and character spacing and page orientation to match the attributes of the selected font. The scalable fonts, however, ignore the font mode setting. Line spacing must be specified by the user in this case using one of the line spacing commands (SLS, SLPI, etc.).

File

```
!R! RES; UNIT P;  
SFNT 'Univers-BdIt', 18, 2000;  
SFNT 'Univers-BdIt', 14, 2001;  
FONT 2000;  
EXIT;  
---- Font Guideline ----  
!R! FONT 2001; EXIT;  
Be distinctive.  
!R! FONT 13; FTMD 15; EXIT;  
Use different font sizes and styles to set off titles,  
headings, footnotes, etc.  
!R! PAGE; EXIT;
```

Printout

--- Font Guideline ---

Be distinctive.

Use different font sizes and styles to set off titles, headings, footnotes, etc.

Related Commands

ALTF, FLST, SETF

FPAT — generate Fill PATtern

Format

FPAT #, #, #, #, #, #, #, #, #;

Parameters


#:

number from 0 to 255

Function

The FPAT command establishes a new shading pattern, for use by the BLK, ARC, and FILL commands, as described in a 8 × 8 matrix of dots.

Each number corresponds to one row of 8 dots. The value of the number is calculated according to the black dots in its row, counting 1 for a black dot in the rightmost column, 2 for a black dot in the next column left, 4 for a black dot in the next column, and so on.



To calculate the third number in the example above, add up the values of the columns that have black dots in the first row: 16+8 = 24.

To calculate the fourth number in the example above, add up the values of the columns that have black dots in the second row: 32+16+8+4 = 60, and so on.

For printing system models that support 600/1200 dpi resolution printing, the default resolution generated by the FPAT command is 300 dpi; however, 600/1200 dpi can be selected using the FRPO B8 command. See Section 6.2. of the *Technical Reference manual (FRPO B8)*. The printing system recognizes a dot in the pattern as having the size of 1/300 inch in the 300 dpi resolution, and 1/600 inch in the 600 dpi resolution, and 1/1200 inch in the 1200 dpi resolution.

For an all-white row, specify 0. The FPAT command for creating all white rows is specified below:

FPAT 0, 0, 24, 60, 60, 24, 0, 0;

For an all-black row, specify 255 (1 + 2 + 4 + 8 + 16 + 32 + 64 + 128 = 255). The default value is FPAT 255, 255, 255, 255, 255, 255, 255, 255; giving a solid black pattern which is equivalent to the PAT 1; command.


File

```

!R! RES; DAM; UNIT C;
FSET 1p08V0s0b5T;
MCRO PATTERN;
FPAT %1, %2, %3, %4, %5, %6, %7, %8;
BOX 5, 1; BLK 5, 1; MRP 0, 1.55;
TEXT 'FPAT %1, %2, %3, %4, %5, %6, %7, %8;', L;
ENDM;
MAP 2, 2; SCP;
CALL PATTERN, 3, 3, 0, 0, 0, 0, 0, 0;
CALL PATTERN, 255, 0, 0, 0, 0, 0, 0, 0;
CALL PATTERN, 1, 1, 1, 1, 1, 1, 1, 255;
RPP; MRP 5.5, 0;
CALL PATTERN, 0, 0, 24, 60, 60, 24, 0, 0;
CALL PATTERN, 1, 2, 4, 8, 16, 32, 64, 128;
CALL PATTERN, 8, 8, 8, 8, 8, 8, 8, 8;
PAGE;
EXIT;

```

Printout (300 dpi)



FPAT 3, 3, 0, 0, 0, 0, 0, 0;




FPAT 0, 0, 24, 60, 60, 24, 0, 0;



FPAT 255, 0, 0, 0, 0, 0, 0, 0;



FPAT 1, 2, 4, 8, 16, 32, 64, 128;



FPAT 1, 1, 1, 1, 1, 1, 1, 255;



FPAT 8, 8, 8, 8, 8, 8, 8, 8;

FRPO — Firmware RePrOgram

Format

```
FRPO parameter, value;
```

Parameters

parameter:

parameter stored in permanent memory

value:

new parameter value

Function

The FRPO command sets the various printing parameters stored in permanent memory. Each parameter is identified by a code consisting of a letter and a numeral, such as P1. Details of the parameters and their values are given in *Chapter 6* of the *Technical Reference* manual. The FRPO command should be used only when it is necessary to change a parameter permanently. They are not effective as temporary changes in the middle of a print job.

The FRPO command effects values for interface dependent and non-interface dependent parameters as they appear on the status page (See *STAT* on page 336.). Changes to the interface dependent parameters will effect the current interface only.

Changes made with FRPO commands can be reset to the factory-set values by the FRPO INIT (FRPO-INITialize) command. Note that the RES command does not reset those changes made with the FRPO command.

File

```
!R!
FRPO P1, 1; CMNT To power up in the IBM emulation;
FRPO P3, 2; CMNT To change the linefeed action to CR+LF;
FRPO U6, 9; CMNT To select the US Legal character set;
EXIT;
US Legal characters:
\^{|}~

!R! PAGE; EXIT;
```

Printout

US Legal characters:
®©§¶™

Related Command

FRPO INIT

FRPO INIT — FRPO-INITialize

Format

```
FRPO INIT;
```

Parameter

None

Function

The FRPO INIT command resets all FRPO parameters to their original, factory default values. This reset applies to the interface dependent parameters for the current interface as well as all non-interface dependent parameters. For a list of the factory default values, see *Chapter 6* of the *Technical Reference* manual. The FRPO INIT command causes a page break.

To activate the changes made with the FRPO INIT command, turn the printer power off then on again or reset the printer using the control panel keys.

File

```
!R! FRPO INIT; EXIT;
```

Printout

Omitted

FSET — change current Font SETting by characteristic

Format

FSET *spacing-pitch-height-style-weight-typeface* ;

Parameters

spacing:

fixed or proportional spacing, followed by a terminator P:

0 = fixed font

1 = proportional font

pitch:

pitch in cpi, from 0.44 to 99.99 cpi (ignored in a proportional font), followed by a terminator H

height:

the line-to-line spacing in points from 0.25 to 999.25; followed by a terminator V

style:

upright or italic, followed by a terminator S:

0 = upright

1 = italic

4 = condensed

5 = condensed italic

weight:

degree of bolding, -7 to +7; followed by a terminator B:

-7 = Ultra thin

-6 = Extra thin

-5 = Thin

-4 = Extra light

-3 = Light

-2 = Demi light

-1 = Semi light

0 = Medium

1 = Semi bold

2 = Demi bold

3 = Bold

4 = Extra bold

5 = Black

6 = Extra black

7 = Ultra black

typeface:

typeface as represented by one of the following numbers, followed by a terminator T:


- 0 = LinePrinterBM8.5-Roman
- 3 = CourierBM
- 4 = Swiss721BM
- 5 = Dutch801BM
- 6 = LetterGothicBM
- 8 = PrestigeEliteBM
- 4099 = Courier
- 4101 = CG Times
- 4102 = LetterGothic
- 4113 = CGOmega
- 4116 = Coronet
- 4140 = Clarendon-Cd
- 4148 = Univers
- 4168 = AntiqueOlive
- 4197 = GaramondAntiqua
- 4297 = Marigold
- 4362 = Albertus
- 16602 = Arial
- 16686 = Symbol
- 16901 = TimesNewRoman
- 24579 = CourierPS
- 24591 = Palatino
- 24580 = Helvetica
- 24607 = ITCAvantGardeGothic-Bk
- 24703 = NewCenturySchoolbook
- 24623 = ITCBookman
- 25093 = Times
- 31402 = Wingdings
- 45099 = ITCZapfChancery-MdIt
- 45101 = ITCZapfDingbats
- 45358 = SymbolPS

The typeface values for option fonts, if installed, can be obtained by printing a font list. To print a font list, use the FLST command or the control panel keys.

Function

The FSET command can be used to select either bitmap fonts or scalable fonts. It is also convenient for changing one or several characteristics of the current font, *e.g.* changing the upright style to italic. The example at the end of this section demonstrates this function.

Font selection is made based on the characteristics of the font, the font's location in the printer, and the font's orientation. In the list below, the characteristics of font are listed in order of priority, from highest to lowest:



In selecting a font, the printer engages in a process of elimination. It first examines all fonts to match based on symbol set (the highest priority). All matches are next examined based on spacing. When the printer has reduced the matches to one font, it selects that font. Bitmap fonts have priority over scalable fonts.

FSET enables the user to select a general class of font first, then specify characteristics (e.g. size and appearance) later. It is still preferable to specify all font parameters to ensure correct selection.

The *pitch* and *height* parameters are measured in cpi and points respectively, regardless of the current unit setting (UNIT).

Terminators following each parameter (P, H, V, S, B, and T) can use either lowercase or uppercase.

The example below shows an FSET command sequence that selects the fixed-pitch (0p), 12-cpi (12h), 12-point (12v), upright-style (0s), normal stroke weight (0b), Letter Gothic (6T) font. This command sequence may be followed by a CSET command which selects the desired symbol set.

Fixed or
proportional

Height

Weight

FSET 0p12h12v0s0b6T;

Pitch

Style

Typeface

File

```
!R! UNIT P; SFNT 'Swiss742SWC-Roman', 10; SLS 12; EXIT;
```

First, this line is printed in a 10 point Swiss742SWC-Roman font.

```
!R! FSET 14V; SLS 16; EXIT, E;
```

Let's see if the height parameter is changed to 14 points by FSET 14V;.

```
!R! FSET 1s3B; EXIT,E;
```

Now this line demonstrates what happens with FSET 1s3B;. Only the style and stroke weight are changed.

```
!R! FSET 4101T; SLS 14; EXIT,E;
```

The FSET 4101T; command switches the typeface to CGTimes. All other characteristics remain unchanged.

```
!R! FSET 0P; FTMD 15; EXIT,E;
```

Now the FSET 0P; command requests a fixed pitch font. The method for selecting this font is based on the change from the previous pitch characteristic.

```
!R! PAGE; EXIT;
```

Printout

First, this line is printed in a 10 point Swiss742SWC-Roman font.

Let's see if the height parameter is changed to 14 points by FSET 14V;.

Now this line demonstrates what happens with FSET 1s3B;. Only the style and stroke weight are changed.

The FSET 4101T; command switches the typeface to CGTimes. All other characteristics remain unchanged.

Now the FSET 0P; command requests a fixed pitch font. The method for selecting this font is based on the change from the previous pitch characteristic.

FTMD — bitmap Font MoDe

Format

FTMD *value* ;

Parameter

value:

an integer from 0 to 15

Function

The FTMD command determines which attributes will be changed automatically when a bitmap font (simulated based on scalable fonts) is selected by the FONT, ALTF, or RPF command. The attributes and values appear as follows:

Horizontal motion index (character spacing)	1
Vertical motion index (line spacing)	2
Fixed or proportional spacing	3
Portrait or landscape orientation	4

The values 1, 2, 4, and 8 change a single font attribute using the FTMD command. When the FTMD command is executed, the printer adjusts the designated attributes of the current font. Later, if the current font is changed by a FONT, ALTF, or RPF command, the printer automatically changes the designated attributes to the values of the new font. The default value of the FTMD command depends on the emulation mode.

The printer does not automatically change any attributes not designated by the FTMD command. If necessary, these attributes must be adjusted by other commands: SCS or SCPI for the character spacing; SLS or SLPI for the line spacing; and SPO for the page orientation. The FTMD command is ignored when the current font is a scalable font.

Font Mode Values

Font mode	Automatic changes			
	Character spacing	Line spacing	Fixed or proportional	Portrait or landscape
0				
1	Yes			
2		Yes		
3	Yes	Yes		
4			Yes	
5	Yes		Yes	
6		Yes	Yes	
7	Yes	Yes	Yes	
8				Yes
9	Yes			Yes
10		Yes		Yes

Font mode	Automatic changes			
	Character spacing	Line spacing	Fixed or proportional	Portrait or landscape
11	Yes	Yes		Yes
12			Yes	Yes
13	Yes		Yes	Yes
14		Yes	Yes	Yes
15	Yes	Yes	Yes	Yes

Default Font Modes

Emulation mode	Font mode	Printer's action when font is changed
0 (Line Printer)	15	Adjusts page orientation, character spacing, line spacing.
1 (IBM X-24E) or 2 (Diablo 630)	0	Does not adjust anything.
5 (Epson LQ-850)		
6 (HP LaserJet)	13	Adjusts page orientation and character spacing. Does not adjust line spacing.
8 (KC-GL)		

File

```

!R! RES; SLM 1; STM 2;
FTMD 5;
CMNT Character spacing is automatically adjusted at each
font change, but line spacing and page orientation are left
alone;
UNIT C; SLS 1;
BOX .5, -.5; MRP 1, 0; FONT 47;
TEXT 'Font 47, correctly spaced at 16.6 cpi', N;
BOX .5, -.5; MRP 1, 0; FONT 37;
TEXT 'Font 37, correctly spaced at 10 cpi', N;
BOX .5, -.5; MRP 1, 0; FONT 52;
TEXT 'Font 52, proportionally spaced', N;
BOX .5, -.5; MRP 1.1, -.1; FONT 26;
CMNT Landscape font in portrait orientation;
TEXT ' = <><><><><><><><><><><><><><><><><><><><><><><> = ';
RES;
EXIT;
    
```

Printout

- Font 47, correctly spaced at 16.6 cpi*
- Font 37, correctly spaced at 10 cpi*
- Font 52, proportionally spaced***
- ||VAVAVAVAVAVAVAVAVAVAVAVAVAVAVAVAV||

GPAT — set Gray PATtern

Format

```
GPAT gray-level;
```

Parameter

gray-level:

any number from 0 (black) to 1 (white)

Function


The GPAT command specifies the shade of gray for painting a block in both the standard mode and path mode graphics. Numbers with up to four decimal places are accepted. The default value is 0 (solid black).

The printer recognizes a dot in the pattern as having the size of 1/300 inch in the 300 dpi resolution, and 1/600 inch in the 600 dpi resolution, and 1/1200 inch in the 1200 dpi resolution. For printer models that support 600/1200 dpi resolution printing, the default resolution generated by the GPAT command is 300 dpi; however, 600/1200 dpi can be selected using the FRPO B8 command. See Section 6.2. of the *Technical Reference* manual (FRPO B8).

File

```
!R! RES; UNIT C;
MCRO GP;
NEWP;
GPAT %1;
PDRP 2, 0, 0, 2, -2, 0, 0, -2;
FILL 1;
PMRP .5, .5;
ENDM;
CALL GP, 0;
CALL GP, .125;
CALL GP, .25;
CALL GP, .375;
CALL GP, .5;
CALL GP, .625;
CALL GP, .75;
CALL GP, .875;
PAGE;
EXIT;
```

Printout



Related Commands

FPAT, PAT

GRAY — represent GRAY

COLOR

Format

GRAY *gray-mode* ;

Parameter

gray-mode:

- 0 = represents gray in K (black, default)
- 1 = represents gray in CMY

Function

The GRAY command determines whether gray in graphic images is represented using black or combination of cyan, magenta, and yellow.

If each value of the RGB signals is the same, the corresponding area will be achromatic (grayed). For example, if $R = G = B = 0$, the corresponding area will be black. If $R = G = B = 100$, the corresponding area will be gray. If $R = G = B = 255$, the corresponding area will be white, that is, there will be no color components. (White and black are not colors. They are states in which all the light is reflected and absorbed respectively.)

If each value of RGB is the same when processing images, the corresponding area can be represented by using either the same quantity of toner or only K toner for each of CMY. (If $C = M = Y$, the color is black in theory.) However, the black represented by $C = M = Y$ is not the same as the black represented by K alone.

When representing an image, for example, if it contains an area represented by $R = G = B$, the look and impression of that image will be completely different when it is represented by CMY and when it is represented by K alone. (Which look is better is dependent on the viewer's taste, the contents to be printed, and so on.)

The graying process defaults to representation using K toner alone. However, a toggle switch is provided since, as mentioned earlier, this process cannot be determined objectively.



Note

This command is irrelevant to models FS-8000C and KM-C830.

GRRD — GRaphic data ReaD

COLOR

Format

```
GRRD format, width, height; <image>;ENDG;
```

Parameters

format:

image format:

B = Bitmap

J = JPEG

T = TIFF

width:

number from 0 to 65535

height:

number from 0 to 65535

image:

file name based on a specified format

Function

The GRRD command allows the image data of a specified format to be drawn in a specified size at a specified location. This command is terminated with the ENDG command.

The color space for each format is compatible only with RGB. In the case of TIFF format, the command is compatible with non-compression, LZW, and PackBit. Upon receipt of incompatible format and/or color-space data, this command is ignored.

Width and height specifications are dependent on the current unit of measurement (UNIT). If the output image has a specified width of 0, the image will be printed in a width varying with the height of the output image, while retaining the vertical-horizontal ratio of the original image data as it is. If the output image has a specified height of 0, the image will be printed in a height varying with the width of the output image, while retaining the vertical-horizontal ratio of the original image data as it is. When the output image has both a specified width and height of 0, the output image pixels will be printed according to the raster resolution specified by the STR command.

Since the command ends with ;ENDG; do not insert space between the preceding ';' and ENDG. If space is inserted, the command is ignored.

When monochrome mode is specified, the output is printed as a grayscale image.

File

The following example uses DOS to transfer both the command sequences and the graphics data to the printing system's parallel interface.

```
c:¥> echo !R! STR 300; MZP 1,1; GRRD J,0,0; > lpt1  
c:¥> copy /b fluit.jpg lpt1  
c:¥> echo ;ENDG; PAGE; EXIT; > lpt1
```

Printout



HUE — adjust HUE

COLOR

Format

```
HUE color, angle[ , type ] ;
```

Parameters

color:

number representing the color for hue adjustment:

0 = all colors
 1 = R
 2 = Y
 3 = G
 4 = C
 5 = B
 6 = M

angle:

angle of hue rotation in degrees:

color = 0, -180 to 180 (default = 0)

color = 1 - 6, -10 to 10 (default = 0)


type:

object type


Function

The HUE command shifts all the pixels around the ‘color wheel’ to a different point for adjusting the tint of a color. For example, shifting the red pixels to green turns the green pixels to blue and yellow pixels to cyan. When *color* is 0, the HUE command shifts all hues at the specified angle.

As shown in the figure below, shifting hues can be likened to rotating the color wheel. When the wheel is rotated by 60 degrees clockwise, red shifts to magenta and yellow to red.



When *color* is other than 0, a particular hue can be shifted. The example below shows that the yellow hue has been shifted in the direction of green indicating that yellow images will be slightly reddish and the image in the vicinity of yellowish green will be yellow. Note that hues cannot be set to move beyond their next hues.



If this command is issued to rotate a particular hue when all hues are rotated, the hues are controlled after they are reset to their default state. The HUE command is enabled for all color spaces, all color modes and all monitor simulation modes. If ink simulation is set to other than NONE, the HUE is disabled.

INTL — InterNaTional characters

Format

INTL *country-code*, *code-set*;

Parameters

country-code:

- 0 = US ASCII
- 1 = France
- 2 = Germany
- 3 = UK
- 4 = Denmark/Norway
- 5 = Sweden
- 6 = Italy
- 7 = Spain
- 8 = Japan
- 9 = US Legal
- 10 = IBM PC-850 (Multilingual)
- 11 = IBM PC-860 (Portuguese)
- 12 = IBM PC-863 (Canadian French)
- 13 = IBM PC-865 (Norwegian)
- 14 = Norway
- 15 = Denmark 2
- 16 = Spain 2
- 17 = Latin America

code-set:

- 0 = Epson code set
- 1 = HP LaserJet code set
- 2 = Daisywheel code set
- 4 = IBM/Line Printer code set

Function

The INTL command selects a symbol set appropriate for the alphabet of a particular country. A symbol set includes the specific alphabetic and numeric characters and other symbols of a particular written language, country, or software application. Charts of the available symbol sets are displayed in *Emulation* of the *Technical Reference* manual, emulation by emulation.

Depending on the current emulation, the printer selects one of the symbol sets of the emulated printer appropriate for the country as requested by the *country-code* value. For example, when the printer is set in the default emulation mode 6 (HP LaserJet), country code 0 (US ASCII) selects the ISO-6 ASCII symbol set of the HP LaserJet printer. If the current emulation mode is 1 (IBM), country code 0 selects the IBM US symbol set of the IBM Proprinter.

Differences in symbol sets can be seen in the upper half of the character code table, consisting of character codes 128 through 254 (hex 80 through FE). The availability of the symbol sets to the country-code values appear in the table below.

For example, the character assigned to character code 93 (hex 5D) varies according to the country code as follows:

US]

France §

Germany Û

Denmark Å

Italy é

Spain ¿

The *code-set* parameter adjusts for the different symbol sets. According to the *code-set* parameter specified, the printer selects a symbol set available to the current emulation.

The default *country-code* value appears as the U6 parameter and can be set using an FRPO command. The factory setting is 0: the standard US ASCII symbol set. The default *code-set* value appears as the U7 parameter. The factory setting is 0.

Symbol Sets Available with INTL

Country-code	Code-set (emulation)			
	0 (Epson)	1 (HP LaserJet) ^a	2 (Daisywheel)	4 (IBM/Line Printer)
0 ^b (US ASCII)	LQ US	ISO-6 ASCII	Diablo US	IBM US
1 (France)	LQ France	ISO-25 France	Diablo France	
2 (Germany)	LQ Germany	HP German	Diablo Germany	
3 (UK)	LQ U.K.	ISO-4 U.K.	Diablo U.K.	
4 (Denmark/Norway)	LQ Denmark	ISO-60 Norway	Diablo Denmark	IBM D/N
5 (Sweden)	LQ Sweden	ISO-11 Swedish	Diablo Sweden	
6 (Italy)	LQ Italy	ISO-15 Italian	Diablo Italy	
7 (Spain)	LQ Spain	HP Spanish	Diablo Spain	
8 (Japan)	LQ Japan	ISO-14 JIS ASCII	Diablo Japan	
9 (US Legal)	LQ US Legal	US Legal	US Legal	US Legal
10 (IBM PC-850)	IBM PC-850	IBM PC-850		IBM PC-850
11 (IBM PC-860)	IBM PC-860	IBM PC-860		IBM PC-860
12 (IBM PC-863)	IBM PC-863	IBM PC-863		IBM PC-863
13 (IBM PC-865)	IBM PC-865	IBM PC-865		IBM PC-865
14 (Norway)	LQ Norway			
15 (Denmark 2)	LQ Denmark 2			
16 (Spain 2)	LQ Spain 2			
17 (Latin America)	Latin America			

^a The INTL command is primarily intended to access foreign characters not found in a current symbol set. (Default is US ASCII). It does not fully support all of the HP LaserJet symbol sets. To use one of the HP symbol sets not listed above, see the CSET command page.

^b Factory-set (default) value

File

```
!R!  
CMNT This file will not print correctly in the IBM;  
CMNT emulation (mode 1);  
RES; INTL 2, 2; SLM 1; STM 2;  
SFNT "Helvetica", 14;  
EXIT;  
Diese [pfel sind eine ]berraschung!  
!R! PAGE; EXIT;
```

Printout

Diese Äpfel sind eine Überraschung!

JOB — JOB Deletion

Format

JOB [*mode* [, *job-id* or *mailbox-id* [, *password*]] ;

Parameters

mode:

mode of deletion:

1 = Deletes temporary jobs

2 = Deletes permanent jobs

8 = Deletes the jobs in the virtual mailbox specified by *mailbox-id*

9 = Deletes the virtual mailbox specified by *mailbox-id*

15 = Resets all virtual mailboxes to a clean state

job-id or *mailbox-id:*

an integer or character string that specifies the job ID or mailbox ID

0 = for modes 8, 9, and 15 only, affects all the mailboxes

1 to 255 = tray ID (specified by JOBS)

Character string = job ID or alias of virtual mailbox, must be enclosed by single or double quotation marks

password:

four-digit number specified by VMPW to secure the mailbox

Function

The JOB command deletes the jobs stored in the hard disk. This command may be used to make more free space in the hard disk by deleting unnecessary jobs.

Modes 1 and 2 delete the temporary and permanent jobs, respectively. By giving a job ID (specified by the JOBS command), it deletes only the job having the matching job ID.

Modes 8, 9, and 15 are used for deletion of the jobs stacked in the virtual mailbox: Mode 8 deletes all jobs stored in the mailbox which is specified by *mailbox-id*. It does not affect the alias and password for the mailbox, if any. (For details on aliasing and password-protection of the mailbox, see VMAL and VMPW, respectively.) Mode 9 deletes the mailbox and its contents at once. The alias and password for the mailbox is also deleted. Mode 15 resets all mailboxes to a clean state by deleting all the contents in the mailboxes and clearing the aliases and the passwords altogether (JOB 15 , 0 , 0 ;).

Files

To delete the temporary job with an ID of *28L016T50* from the hard disk:

```
!R! JOB 1, "28L016T50"; EXIT;
```

To delete all permanent jobs in the hard disk:

```
!R! JOB 2; EXIT;
```

To delete all temporary and permanent jobs in the hard disk:

```
!R! JOB; EXIT;
```

To delete the jobs stacked in virtual mailbox number 3, which has the password of 9999:

```
!R! JOB 8, 3, 9999; EXIT;
```

JOBL — print JOB List

Format

JOBL *mode* ;

Parameter

mode:


- 1 = prints a list of temporary jobs in the hard disk
- 2 = prints a list of permanent jobs in the hard disk
- 8 = prints a list of virtual mailboxes to represent their usage status


Function

The JOBL command prints a list of all permanent jobs in the hard disk, or all temporary jobs, for checking the stored contents in the hard disk and to selectively delete jobs. The jobs stored can be directly printed by reading barcodes on the job list with the barcode reader BC-1.

The JOBL 8; command outputs a list representing the usage status of virtual mailboxes.

Example of VMB virtual mailbox usage status list output



Regardless of whether mailboxes contain any data, information is displayed for mailboxes which have been used at least once. The information displayed for each mailbox is: the mailbox (Bin) number, the alias set for the mailbox, the number of jobs in the mailbox, the number of pages, the size of jobs held in that mailbox only, and the size of jobs entered in other mailboxes as well. A  is displayed over the mailbox number for virtual mailboxes which contain jobs. A *Full!!* is displayed next to the job size entry for virtual mailboxes containing jobs which caused an error during data is written to the mailbox. An error log indicating failure is also printed when jobs within such a mailbox are printed.

File

To print a permanent job list:

```
!R! JOBL 2; EXIT;
```

JOBO — JOB Output

Format

```
JOBO mode , 'job-id' or 'mailbox-id' [ , password ] ;
```

Parameters

mode:

0 = normal
8 = virtual mailbox

job-id (mode 0):

job ID or page ID

mailbox-id (mode 8):

mailbox ID from 1 to 255, or alias

password (mode 8 only):

password for the mailbox

Function

The JOBO command retrieves a temporary or permanent job or the jobs saved in the virtual mailbox and prints with the finishing arrangements previously made.

Mode 0 is used to retrieve and print a temporary or permanent job. The job to be retrieved may be specified by the job ID which is given internally by the printing system. It is also possible to specify the page ID if only a particular page is required to be printed. The page is automatically assigned and printed on a page when one of the ID barcode position parameters of 128 through 135 is specified for JOBS. A list of job names currently stored in the hard disk can be obtained by printing a job list (See *JOBL* on page 134).

Settings such as the number of pages to print, whether to print barcode ID on the printed pages, etc., for the job to be retrieved by JOBO can be made by issuing JOBP followed by a JOBO (See *JOBP* on page 138).

If the job name does not exist or the job is corrupted in hard disk, an error is messaged on the operator panel display as follows:

Hard Disk Error	Meaning
99	The job does not exist.
98	The job is corrupted.

Mode 8 retrieves jobs stacked on a virtual mailbox for printing. In mode 8, the ID number (or the alias, if the mailbox has one given) for the mailbox should be specified in place of *job-name*. The mailbox ID number should be 1 to 255. All jobs in the mailbox specified by its mailbox ID are printed at once. After printing is finished, all jobs in the mailbox are automatically deleted.

All jobs are delivered to the current output stack (all jobs to the same stack). Note however the jobs with stapling enabled will be forcibly sent to the stapler.

The master password, if set previously by VMPW, is valid for all virtual mailboxes for which a password is set. An error will be displayed on the operator panel if no password or an incorrect password is specified. An error will result also if no virtual mailbox corresponds to the alias given or if no job is present in the specified mailbox. It is not possible to select a virtual mailbox using the operator panel when no jobs are posted in the virtual mailbox.

If the CANCEL key is pressed during virtual mailbox job output, processing will halt and the printing system will return to Ready status. The job being output and jobs in the virtual mailbox can be printed.

Files

To output the job with job ID *4H90058I0* in the hard disk, with no barcode printed, in 7 copies:

```
!R! JOBP 0, 7; JOBP 1, 2; JOBO 0, '4H90058I0';  
EXIT;
```

To print the jobs stacked on virtual mailbox number 3, for which no password is set:

```
!R! JOBO 8, 3; EXIT;
```

To output the jobs stacked on the virtual mailbox with the alias *Jonathan*, for which the password 9999 is set:

```
!R! JOBO 8, 'Jonathan', 9999; EXIT;
```

JOBP — JOB, print with Print options

Format

JOBP *print-option* , *parameter* ;

Parameters

print-option:

an integer that specifies the print option from the ones below

parameter:

an integer, the letter m, or character string that specifies the parameter for the print option as indicated in the table below

print-option	Function	Parameter	Effective to:
0	Copy count	1 to 999	
1	Printing barcode	0 = first-page only 1 = whole page 2 = disabled	
2	Duplex printing	0 = off 1 = long edge binding 2 = short edge binding	JOBO
3	Destination stack	1 = face-down 2 = face-up 3 = finisher tray	JOBO
4	Stapling	0 = off 1 = bottom left, angled 2 = top left, automatic angled or upright 3 = two centered	JOBO
5	Barcode position	0 to 7 or 128 to 135 (printing page ID's)	
6	Staple count	M or m = staples the maximum number of sheets. For example, 20 sheets for models DF-30/31. 0 = does not staple jobs that exceed the maximum number of sheets. -1 = staple only the sheets that exceed the maximum number of sheets. 2 to 20 = staples the number of specified sheets.	
7	Center stitch	0 = off 1 = center stitch and stapling	
8	Punching	0 = off 1 = main position 2 = alternate position	
9	Transparency separator	0 = off 1 = blank separator 2 = duplicated separator	
10	Source for transparency separator	number representing the paper source (tray)	
11	Rotated collation	0 = off 1 = on	

print-option	Function	Parameter	Effective to:
13	Booklet	0 = off 1 = longitudinal feeding, left binding 2 = longitudinal feeding, right binding 3 = lateral feeding, left binding 4 = lateral feeding, right binding	

Function

The JOBP command implements printing a print job which was prologued by JOBS and epilogued by JOBT in the similar manner as the JOBO command (See *JOBO* on page 136.). Use of the JOBP command allows to use the different settings for copy count, duplex printing, stapling, etc., from those originally set for the print job. For example, it is possible to duplex-print the print job (in which duplex-printing was not originally implemented) by commanding JOBP 2, 1; or remove stapling from the print job (in which stapling was originally implemented) by commanding JOBP 4, 0;.

Print options 2, 3, 4, and 16 are valid only when the JOBP command follows the JOBO command to print the job. For example, the JOBP command sequence JOBP 2, 1; will not activate duplex printing if the preceding job to print was assigned by JOBS.

A JOBP command sequence is cancelled if the host interface of the printing system is switched from one to another after the timer has expired for the interface.

Files

To print the print job 'monthlyreport,' disabling the original barcode printing option, in the number of copies of 15:

```
!R! JOBP 0, 15; JOBP 1, 2; JOBO 'monthlyreport'; EXIT;
```

JOBS — JOB Start

Format

```
JOBS save-mode, copy, barcode, barcode-position[, share, 'job-name',  
'user-name', 'time' ];
```

Parameters

save-mode:

save mode:
0 = electronic sorting only
1 = temporary
2 = permanent
8 = (Print to) virtual mailbox

copy:

number of copies to print:
1 to 999

barcode (*starting-mailbox* when *save-mode* is 8):

a number representing the page(s) barcode is printed¹:
0 = On cover page only
1 = On whole pages
2 = None

barcode-position (*ending-mailbox* when *save-mode* is 8):

a number representing the position of barcode:
0 to 7 or 128 to 135 (See next page.)

share:

reserved: Must be 0

job-name:

job name, character string up to 80 bytes

user-name:

user name, character string up to 80 bytes

time:

time stamp, character string up to 80 bytes

¹ *Save-mode* must be permanent (2).

Function


The JOBS command defines start of a job which is processed using e-MPS. The subsequent document is spooled in the hard disk or RAM disk. (Only HP mopier compatible spooling is possible if the printing system uses RAM disk.) Electronic sorting becomes possible in emulations other than PDL and KPD. This command must be terminated by JOBT as exemplified in *Files* below on page 144.

Among the e-MPS finishing functions, not all of them require a hard disk. For example, using the staple function does not require a hard disk in the printing system.

If the size of the permanent jobs saved in the hard disk has reached the maximum limit while the first parameter specifies the permanent job saving mode, the printing system indicates 'harddisk error 97' meaning that no more permanent jobs are storable. The printing system forcibly uses save-mode 0.

With mode 0, 1, or 2, it is possible to print a barcode ID which is directly read by an option barcode scanner for reprinting the job at any later time. The position for the barcode ID can be one of the following positions depending according to the given barcode position parameter. Note that parameters 128 through 135 allow a human-readable barcode ID printed along the barcode ID.

Barcode positions



Spooling to Virtual Mailbox

Specifying 8 for *save-mode* enables virtual mailbox spool mode.

The *starting-mailbox* value must be greater than the *ending-mailbox* value. Otherwise, only the starting mailbox becomes valid.

When no virtual mailbox corresponds to the alias given, the message display shows 'hard disk error 85 (see the *Technical Reference* manual.)' and information regarding the job which was thrown away is displayed in the error log.

If JOBS-JOBT pairs are nested or if they are not paired properly, the job defined between the innermost JOBS-JOBT pair will be valid.

The command is invalid if any of the job name, user name or time stamp exceed 31 bytes. Note that the job which would have been placed in a virtual mailbox will be immediately printed.

Note that if parameters *starting-mailbox* and *ending-mailbox* do not specify a valid virtual mailbox number (1 to 255), or if the JOBP command does not include a valid

virtual mailbox specification (See *JOBP* on page 138.), the job in question will not be saved to any virtual mailbox and no error will be displayed.

While jobs are being stacked on a virtual mailbox, a message indicating that jobs are being stacked on virtual mailboxes (when the VMB-compatible printer driver is being used) will be displayed on the message display instead of *Processing*.

Files

To print 25 copies of the job named *Project-report1* with user name *T. Anderson*, and time stamp of *July 29, 2002*. The job is saved in HP mode.

```
!R! JOBS 0,25,0,0,0,  
      'Project-report1',  
      'T. Anderson',  
      '07/29/02';  
EXIT;
```

To print 3 copies of the job named *Beauty-parlor-list* with user name *Salt Lake* and time stamp of *January 12, 2002*. The job is saved in permanent mode.

```
!R!  
JOBS 2,3,2,2,0,'Beauty-parlor-list', 'Saltlake','01/12/02';  
TEXT 'Page 1...';  
PAGE;  
TEXT 'Page 2...';  
PAGE;  
TEXT 'Page 3...';  
PAGE;  
JOBT;  
JOBL;  
EXIT;
```

To print one copy of the job named *Inventory List 2002* with user name *Susan* and time stamp of *February 12, 2002*. The job is saved in temporary mode.

```
!R!  
JOBS 1,1,1,0,0,'Inventory List 2002', 'Susan', '02/12/02';  
TEXT 'Page 1...';  
PAGE;  
TEXT 'Page 2...';  
PAGE;  
TEXT 'Page 3...';  
PAGE;  
JOBT;  
EXIT;
```

To stack the jobs between *JOB*S and *JOB*T on the virtual mailbox named *Sales*.

```
!R! JOBS 8, 3, 'Sales', 0, 0, 'Invoice', 'Sales', '01/14/02  
16:30:25';  
UNIT C; MZP 5, 10; BOX 6, 6; PAGE; TEXT 'November Result';  
PAGE; TEXT 'To Mr. Johnson'; PAGE;  
JOBT; EXIT;
```

To stack the jobs between *JOB*S and *JOB*T on all virtual mailboxes having numbers 3 to 15.


```
!R! JOBS 8, 3, 3, 15, 0, 'Service', 'Repair', '01/14/02  
16:30:25';  
UNIT C; MZP 5, 10; BOX 6, 6; PAGE; TEXT 'ABCDE'; PAGE;  
TEXT 'Domestic'; PAGE;  
JOB; EXIT;
```

JOBT — JOB Terminate

Format

JOBT;

Parameter

None

Function

The JOBT command terminates defining a job which is started by JOBS. The JOB command is used in combination with JOBS to constitute a job definition. The printing system prints the job in the number of copies as specified by the repeat count commanded by JOBS or JOBP in the cyclic manner.

At execution, the JOB command starts printing the job including stapling, job separation, job offset (if the relevant option device is installed).

File

The file below sorts and prints the defined job in 40 copies. The pages between JOBS and JOB are output to virtual mailbox number 3. Pages after JOB are output to the current physical stack.

```
!R! JOBS 0,40,2,0;  
    TEXT "Circulation Slip";  
    PAGE;  
    TEXT "Response Slip";  
    PAGE;  
    JOB;  
EXIT;
```

JOG — JOG output stacks for separation

Format

JOG;

Parameter

None

Function

It is possible to offset separate jobs in the output stack by explicitly repeating the JOG command so that the user can clearly distinguish his/her print jobs from the others. This command is only enabled when an optional stacker is attached. The STAK command must be used to explicitly forward printed pages to the optional stacker.

The optional stacker has two modes for 'jogging:' to jog only the first page or all pages of the job to the left or right. Selection of jog mode can be made by the FRPO S8 command as follows:

All pages	FRPO S8 = 0
First page only	FRPO S8 = 1

Unlike the job offset provided by HP PJI for each individual job, the JOG command affects any subsequent print jobs until another JOG command is issued. This is unaffected even by the reception of a command such as RES, FRPO and SEM which results in a reset of the printer.

When JOG is issued with a job defined by JOBS and JOBT, jogging affects the individual job (in the similar way as PJI). Therefore, JOG placed between JOBS and JOBT is ignored (except for the first page).

If used together with STPL, STPL overrides JOG, regardless of the order the commands are issued.

File

The example below prints 3 copies of the job defined between JOBS and JOBT (job name: My Trial, user name: Halle Berry, date: January 12, 2002) with the electronic sorting function and deliver printed pages to the face-down tray of the optional stacker in offset mode or separation mode. (A barcode is output on the first page only. The barcode position is "0".)

```
!R!
STAK 3; CMNT Face-down;
JOG;
JOBS 0,3,0,0,0,"My Trial", "Halle Berry", "01/12/02";
TEXT "Page One: Testing how JOG works.";
PAGE;
TEXT "Page Two: Testing how JOG works.";
PAGE;
TEXT "Page Three: Testing how JOG works.";
PAGE;
JOBT;
EXIT;
```

LAPI — Load API Program

Format

LAPI *Program type*, *PRESCRIBE command*, *function name*;

Parameters

Program type:

Type of the downloading program
"PRE" PRESCRIBE command

PRESCRIBE command name:

Name of the PRESCRIBE command to be stored
Alphanumeric string of 2 to 4 characters
(Only in the case of "PRE")

Function name:

Name of the function that starts the user program
(Only in the case of "PRE")
Up to 31 characters

Function

The LAPI command downloads the API program.

As described in the example below, add the string "WRED" to the end of the downloading data as a terminator.

It is possible to add a string other than "WRED." It is because if the downloading program includes the string "WRED," this must be changed. To change the terminator, use the RWRF T command.

When storing a new PRESCRIBE command, if the specified command name has already been stored, the sending data will be invalid.

Data downloaded with the LAPI command will be deleted (invalid) upon the system restart.

Example:

```
!R! LAPI "PRE" , "XXXX" , "XXXXXXXX" ;[Program Data]WRED;  
EXIT;
```

After "WRED" is changed to "TEST"

```
!R! LAPI "PRE" , "XXXX" , "XXXXXXXX" ;[Program Data]TEST;  
EXIT;
```

LDFC — Load Font Character

Format

LDFC *font-number*, *code*, *height*, *width*, *y-offset*, *x-offset*,
cell-width, *center*, *rotation*[, *resolution*]; *bit-map*;

Parameters

font-number:

font number (integer from 1000 to 65535). Same for all characters in the font. Must not duplicate any resident bitmap font number.

code:

ASCII code of the character (0 to 255)

height:

height of the character bitmap, in dots (integer from 0 to 3999)

width:

width of the character bitmap, in dots (integer from 0 to 3999)

y-offset:

distance from the baseline to the top of the character bitmap, in micro dots [1/32 dot] (integer from -32768 to 32767); positive – up, negative – down

x-offset:

distance from the left edge of the character cell to the left edge of the character bitmap, in micro dots [1/32 dot] (integer from -32768 to 32767); positive – up, negative – down

cell-width:

total width of the character cell, in micro dots [1/32 dot] (integer from 0 to 65535). This parameter is used in proportional spacing.

center:

distance from the left edge of the character cell to center of the character, in micro dots [1/32 dot] (integer from 0 to 65535)

rotation:

0 = portrait
1 = upside-down landscape
2 = upside-down portrait
3 = landscape

resolution:

resolution of the character, 300 [dpi], 600 [dpi], or 1200 [dpi]

bit-map:

dot pattern of the character, coded as a character string

Function

The LDFC command defines one character of a new bitmap font. It can be used to create special characters such as the one shown below, which is defined as the '+' symbol (character code 43) of font 1000. Details and another example appear in Chapter 4 of the *Technical Reference* manual.

The *resolution* is valid only on a 600/1200 dpi printer model.

File

```
!R!  
LDFC 1000, 43, 39, 29, 1216, 0, 1216, 608, 0;  
@@2@@0@@2@@0@@2@@0@@2@@0@@7@@0@@7@@0@@7@@0@@?`@0  
@@?`@0@A?p@0@A?p@0@C?x@0@G?|@0@O?~@0@_?/@0@/?/'0  
A/?/p0G/?/x0O/?/~0//?//8O/?/~0G/?/x0A/?/p0@/?/'0  
@_?/@0@O?~@0@G?|@0@C?x@0@A?p@0@A?p@0@@?`@0@@?`@0  
@@7@@0@@7@@0@@7@@0@@2@@0@@2@@0@@2@@0@@2@@0;  
  
FONT 1000; SLM 1; SCPI 6; SLPI 6;  
TEXT '+ + + + + + + + + + + + +', N;  
TEXT ' + + + + + + + + + + + + +', N;  
TEXT '+ + + + + + + + + + + + +', N;  
TEXT ' + + + + + + + + + + + + +', N;  
TEXT '+ + + + + + + + + + + + +', N;  
EXIT;
```

Printout



LDFN C — generate bitmap character for Loading FoNT

Format

LDFN C, *font-number*, *code*, *height*, *width*, *y-offset*, *x-offset*, *cell-width*;
bit-map;

Parameters

font-number:

font number (integer from 1000 to 65535)

code:

ASCII code of the character code (0 to 255 for European; 0 to 65533 for Japanese) or in hexadecimal notation (\$xxxx)

height:

height of the character bitmap, in dots (integer from 1 to 65535)

width:

width of the character bitmap, in dots (integer from 1 to 65535)

y-offset:

bitmap Y offset, in 1/32 dots (integer from -32768 to 32767)

x-offset:

bitmap X offset, in 1/32 dots (integer from -32768 to 32767)

cell-width:

character width, in 1/32 dots (integer from 1 to 65535)

bit-map:

bitmap transfer data (BBN format)

Function


The LDFN C command defines bitmap characters for each character code for the downloaded font header already defined using LDFN F. Note that each character code defined must match the font number, symbol set (1- or 2-byte code), and other information of the font header already created. The bitmap data transferred at the end of this command is transferred under Kyocera's original BBN format. This is the transfer format used by the LDFC and LDKC commands.

This command is recommended for use with downloading bitmap fonts. In other words, this command is intended as an equivalent for the old LDFC and LDKC commands. Use this command when creating a new bitmap font. LDFC, LDKC and CRKF will be deleted in the future.

Units used in *height*, *width*, *y-offset*, *x-offset* and *cell-width* are dots of the resolution defined in the font header.

The coordinate system used when defining bitmap fonts with this command is defined as follows. Note the sign of the offset value.

Font Registration Coordinate System



A command error will result and the command has no effect if any of the parameters are outside the given range. This command is ignored if even one parameter error occurs.

LDFN F — create header for LoadIng FoNt

Format

LDFN F, *font-number*, *resolution*, *name*, *symbol-set*, *spacing*, *width*, *height*,
style, *stroke*, *type-face*, *type*;

Parameters

font-number:

font number: 1000 to 65535

resolution:

resolution:
0 = 300-dpi bitmap
1 = 600-dpi bitmap
3 = 1200-dpi bitmap (Fine mode)

name:

font name (ASCII text string up to 31 characters long). This string must be enclosed in either single or double quotation marks.

symbol-set:

symbol set:
numeric value = 1-byte code symbol set (1 to 65535 [619: Shift JIS code])
J = 2-byte JIS code

spacing:

spacing:
0 = fixed spacing
1 = proportional spacing

width:

font width (in dots set by *resolution*): 3 to 3000

height:

font height (in dots set by *resolution*): 4 to 4165

style:

font style: 0 to 32767

stroke:

stroke weight: -127 to +127

type-face:

type face: 0 to 32767

type:

font type:

0 = character codes from 32 to 127

1 = character codes from 32 to 127 and from 160 to 255

2 = character codes from 0 to 255

Function

The LDFN F command creates a font header for downloading a font to the printer. After the header is created, you must continue with downloading bitmap characters using the LDFN C command. This command only generates a font information header; it includes no character definition information. Since the font header defined by this command is capable of registering extensions to ANK codes (a 1-byte code), it is necessary to select a font after assigning fonts using the ASFN command when using 2-byte characters. Refer to the ASFN command page for details.

This command is recommended for use with downloading bitmap fonts. This command is intended as an equivalent for the old LDFC, LDKC and CRKF commands to create a new bitmap font. LDFC, LDKC and CRKF will be discontinued in the future.

Font Select Command	Description
FONT, SFNT	European character (1-byte code) emulation
KFNT, SKFT	Japanese character (2-byte code) emulation (linked using ASFN)

A command error will result and the command has no effect if any of the parameters are outside the given range. This command is ignored if even one parameter error occurs.

LDFN S — Load truetype FoNt

Format

LDFN S, *format*, *font-number*, *name*, *length* [, *symbol-set*, *style*, *stroke*, *type-face*] ; *binary-data* WRED ;

Parameters

format:

font format:
0 = TrueType font
1 = Japanese TrueType font
2 = Japanese TrueType Collection font

font-number:

font number: 1000 to 65535
Since there are two font headers for Japanese TrueType fonts, the printer automatically creates a font number increased by 1.

name:

font name: ASCII text string of up to 31 characters long. This string must be enclosed in either single or double quotation marks.

length:

font data length: number of bytes sent as *binary data*

symbol-set:

font symbol set value: 1 to 65535 (619 = Shift JIS code)

style:

font style: 0 to 32767

stroke:

stroke weight: -127 to +127

type-face:

type face: 0 to 32767

binary-data:

font data (in TrueType format)

Function

The LDFN S command allows outline fonts (TrueType fonts) to be downloaded to the printing system. A binary format is used to represent font data. The size of this data must match that given by *length* (font data length).

Always explicitly terminate font data using WRED ; . The five bytes represented by WRED ; are not counted as part of the font data length.

The maximum number of fonts that can be downloaded is up to the memory available or 256 fonts (512 when *format* is 2). Use FONT or SFNT command to select 1-byte fonts to be defined with this command.

TrueType fonts defined using this command cannot be completely deleted unless both DELF n; and DELF n+1; are specified.

A command error will result and the command has no effect if any of the parameters are outside the given range. This command is ignored if even one parameter error occurs.

Although *symbol-set*, *style*, *stroke* and *type-face* may be omitted, we recommend that all parameters be used to define font selection information.

LGHT — adjust LiGHTness

COLOR

Format

LGHT *mode*, *degree*;

Parameters

mode:

mode of lightness adjustment:

0 = gamma control

1 = contrast control

degree:

degree of saturation level:

number from -10 to 10 (Default is 0)

Function


The LGHT command controls lightness and darkness balance based on the specified parameters. The gamma and contrast can be set independently of each other.

In gamma control mode, the color bars below represent the lightness gamma variation using hue R as an example. The top color bar represents the default lightness variation ($\gamma = 1.0$) with its median indicated by the Δ symbol. The left side represents the lower lightness range and the right side the higher lightness range. The middle color bar is an example lightness balance made brighter (set to $\gamma > 1.0$). The bottom color bar is an example lightness balance made darker (set to $\gamma < 1.0$).



Varying the lightness gamma allows the lightness of the image to vary. (You can achieve lightness variation, such as making dark images brighter by reducing low lightness areas, for example.)

In contrast control mode, the color bars below represent lightness contrast variations using hue R as an example. The top color bar represents the default contrast variation with its median indicated by the Δ symbol. The left side represents the lower lightness range and the right side the higher lightness range. The middle color bar is an example application of strong contrast to the top bar. The bottom color bar is an example application of weak contrast to the top bar.



Varying the darkness contrast allows the difference between the light and dark shades of the image to be articulated and the intensity of control changes the difference between the light and dark shades of the image.

Lightness gamma and darkness contrast cannot be controlled hue by hue. However, since impressions vary with hue, you need to control with care.

The LGHT command is enabled for all color spaces, all color modes and all monitor simulation modes.

If ink simulation is set to other than none, this command is disabled. This mode is not held permanently in the printing system. For this reason, saturation is reset to its default state when the power is turned on.

MAP — Move to Absolute Position

Format

```
MAP x-coordinate , y-coordinate ;
```

Parameters

x-coordinate:

distance from the left margin

y-coordinate:

distance from the top margin

Function


The MAP command moves the cursor to a position specified relative to the top and left margins, measured in the units designated by the UNIT command. The coordinates may be positive or negative. Negative coordinates move the cursor outside the margin.

The MAP command is ignored in path mode graphics.

File

```
!R! RES; UNIT C; STM 2; SLM 2;
SFNT 'Helvetica-Nr', 9;
MAP 0, 0; BOX 40, 40; SPD .2;
MAP 0, 1.4; TEXT '< = = = Margins';
MAP 0, 0; CIR .1; TEXT, L; TEXT ' MAP 0, 0;';
MAP 4, 0; CIR .1; TEXT, L; TEXT ' MAP 4, 0;';
MAP 0, 2; CIR .1; TEXT, L; TEXT ' MAP 0, 2;';
MAP 4, 2; CIR .1; TEXT, L; TEXT ' MAP 4, 2;';
MAP -1, -1; CIR .1; TEXT, L; TEXT ' MAP -1, -1;';
PAGE;
EXIT;
```

Printout



Related Commands

MRP, MZP, MRPA

MCLR — Match CoLoR

COLOR

Format

MCLR *mode* ;

Parameter

mode:

mode of color matching:

0 = none (default)

1 = auto

2 = vivid color

3 = display

Function

The MCLR command selects the mode of color matching.

Color matching mode is changed according to the various conditions of emulation, paper type, and print quality. In monochrome mode, color matching is defaulted to 0 (none).

Mode 0 does not perform color matching. Mode 1 automatically selects color correction mode depending on the printing data. Text or vector graphics data are processed using Vivid Color; raster image data are processed using display mode. Mode 2 (Vivid color) performs correction on cyan color. Mode 3 (display) performs color correction suited to simulate the standard display monitor.

This command causes a form feed.

MCRO — define MaCRO

Format

```
MCRO name[ dummy-sign[ , comment ] ] ;
```

Parameters

name:

character string starting with a letter

dummy-sign:

any character (default: % preceded by a space)

comment:

any character string

Function

The MCRO command assigns a name to the sequence of PRESCRIBE commands that follows, up to the ENDM (end macro) command. Thereafter, the entire sequence of commands can be executed by specifying the assigned name in a single CALL or AMCR (automatic macro) command.

The *name* can be any length but only the first four characters count. The rest will be ignored. The distinction between upper and lowercase letters is also ignored. For example, the following macro names are all the same:

```
ABCD
abcd
ABCDE
Abcdxyz
```

The name must start with a letter, but the other characters can include digits and special symbols such as hyphens. For example, F-1 and GRY2 are valid macro names.

dummy-sign (normally the percent sign) indicates dummy parameters in the body of the macro. Dummy parameters are given by writing the dummy sign followed by a number: %1 for the first dummy parameter, %2 for the second dummy parameter, and so on, up to a maximum of 19 dummy parameters. The same dummy parameter can be used any number of times. Values are assigned to dummy parameters when the macro is invoked by the CALL command.

If a macro with the same name has already been defined, the new definition is ignored. To redefine a macro, you must first delete the old definition with the DELM (delete macro) or DAM (delete all macros) command, or by switching power off.

See the *Technical Reference* manual for information on the limitations about the usage of macros.

The PCL5E command language, available in the mode 6 emulation, enables users to record HPGL/2 macros inside of HPGL/2 commands.

The following table shows the source of the PCL macros, the location in the printer, and the property value for the resource — important if resource protection mode is used. See also *Resource Protection* in the *Technical Reference* manual.

Macro	Location	Property
Downloaded	RAM	Temporary/Permanent
Memory card	Memory card	Permanent
Internal	ROM	Permanent

The internal printer firmware manages the PCL macros, recognizing the source, location and property value of the macros.

If the printer encounters PCL macros with the same macro ID, it will prioritize them in the following order: macros downloaded to RAM, macros residing on a memory card, and macros stored in the EPROM.

Memory card and EPROM based PCL macros cannot be deleted. Downloaded PCL macros, available for all FS-series printers, may be removed using the PCL reset or delete commands: <ESC>&f6X or <ESC>&f8X. Downloaded macros may also be sent to specific host buffers of the printer interfaces. The memory card and ROM macros will be available on all interfaces.

File

```
!R! RES; FSET 1p12v0s3b24580T;
MCRO SIGN %, string, length + 0.3;
SPD .03;
BOX %2, .3; MRP .15, .2;
TEXT %1; MRP -.15, .35;
ENDM;
CALL SIGN, 'CHECK ALL SWITCHES AND INDICATORS ...', 3.9;
CALL SIGN, 'BE SAFE; NOT SORRY', 2.1;
PAGE;
EXIT;
```

Printout

CHECK ALL SWITCHES AND INDICATORS ...

BE SAFE; NOT SORRY

MDAT — set MeDia type ATtribute

Format

MDAT [*media-type*[, *paper-weight*, *fuser-mode*, *duplex*[, *density*]]];

Parameters

media-type

- 1 = Plain
- 2 = Transparency
- 3 = Preprinted
- 4 = Label
- 5 = Bond
- 6 = Recycled
- 7 = Vellum
- 8 = Rough
- 9 = Letterhead
- 10 = Color
- 11 = Prepunched
- 12 = Envelope
- 13 = Cardstock
- 14 = Coated
- 15 = Second side
- 16 = Thick
- 21 = Custom 1
- 22 = Custom 2
- 23 = Custom 3
- 24 = Custom 4
- 25 = Custom 5
- 26 = Custom 6
- 27 = Custom 7
- 28 = Custom 8

paper-weight

- 0 = Normal
- 1 = Heavy (Thick)
- 2 = Light (Thin)
- 3 = Extra thick
- 10 = Light
- 11 = Normal 1
- 12 = Normal 2
- 13 = Normal 3
- 14 = Heavy 1
- 15 = Heavy 2
- 16 = Heavy 3
- 17 = Extra heavy

fuser-mode

0 = High
1 = Middle
2 = Low
3 = Vellum

duplex

0 = Enable
1 = Disable

density

1 = Light
2 = Medium Light
3 = Normal
4 = Medium Heavy
5 = Heavy

Function

The MDAT command is used to customize the default attributes of a media type for optimization. The *paper-weight* attribute affects the printing system's transfer bias, *fuser-mode* affects the fuser temperature, and *duplex* determines whether the paper with that paper type is allowed for duplex printing or not.

If the parameters *paper-weight* and *fuser-mode* are omitted, this command resets the three attributes back to their defaults. If all the parameter are omitted (MDAT;), all the attributes for the all media types are reset to the default.

File

To set the transparency paper type to the thick paper weight and the high fuser temperature mode, disabling the duplex printing, command:

```
!R! MDAT 2, 1, 0, 1; EXIT;
```

Related Command

MTYP

MID — Multi-tray ID

Format

```
MID tray-group;
```

Parameter

tray-group:

number from 1 (top) to the maximum

Function

The MID command selects a group of trays in the sorter, which is assigned by the ASTK command and allows the subsequent SSTK (select Sorter STacKer) or CSTK (select Collator STacKer) command to deliver the output pages into the trays of that group for sorting or collating.

The MID command should be placed at the beginning of a file or program, or at the top of a new page. The selection of a tray group made by the MID command is effective until it is changed by another MID command (It is not reset by the RES command).

If the printer receives an ASTK command or one of the FRPO commands, the MID command is canceled (the output pages will be sorted in the normal sorter mode) and you must reselect the tray group by specifying the MID command.

The MID command is ignored while the multiple environment mode is interface-specific (FRPO S0, 2;) or off (FRPO S0, 0;). Refer to the sorter's *User's Manual*.



Note

This command is irrelevant to certain combination of photocopier models and option finisher models.

File

To enable the multiple user mode (FRPO S0, 1;), to create three groups, to select the second group, and to deliver outputs in the fourth tray in the (second) group, send the following command sequence:

```
!R! FRPO S0, 1; ASTK 2, 5; MID 2; SSTK 2; EXIT;
This page must be delivered to the 4th tray in the sorter.
!R! RES; EXIT;
```

Related Commands

ASTK, CSTK, MSTK, SSTK

MPSS — e-MPS Storage

Format

```
MPSS 'job-type' , 'limit' , size ;
```

Parameters

job-type:

CODE = temporary or permanent job
 VMB = virtual mailbox job
 JOBR = retention job

limit:

TEMPLIMITSIZE = Limit of the capacity for temporary stored jobs
 PERMLIMITSIZE = Limit of the capacity for permanent stored jobs
 VMBLIMITSIZE = Limit of the capacity for virtual mailbox jobs
 QUICKLIMITCOUNT = Limit of the number of Quick Copy jobs

size:

number from 0 to 9999 in unit of megabytes
 number from 1 to 50 for QUICKLIMITCOUNT

Function

The MPSS command is used to allocate the limit values for the storage capacities to store temporary and permanent jobs, virtual mailbox jobs, and the number of Quick Copy jobs in the hard disk.

Files

To set the maximum capacity for the permanent jobs to 600 MB:

```
!R! MPSS 'CODE' , 'PERMLIMITSIZE' , 600 ; EXIT ;
```

To set the maximum capacity of the virtual mailbox to 800 MB:

```
!R! MPSS 'VMB' , 'VMBLIMITSIZE' , 800 ; EXIT ;
```

To set the maximum number of Quick Copy jobs (Job Retention) to 50:

```
!R! MPSS 'JOBR' , 'QUICKLIMITCOUNT' , 50 ; EXIT ;
```

MPTE — MP Tray Empty

Format

```
MPTE mode ;
```

Parameter

mode:

0= disables the paper-empty message despite the MP tray is empty.
1= enables the paper-empty message when the MP tray is empty.

Function

When the MP tray is configured as the default paper source, the MPTE command enables or disables the paper-empty message which is shown when the MP tray becomes empty.

Mode 1 (default) enables the paper-empty message to display when the MP tray is empty. When Mode is 0, this message is disabled despite the MP tray is empty and the other paper source(s) is not empty. (showing Ready, for example).

It is not reset by the RES command, therefore, to change mode, the MPTE command should be used.

File

To configure the MP tray as the default paper source and disable the paper-empty message, command as follows:

```
!R! FRPO R4, 0; MPTE 0; EXIT;
```


MRP — Move to Relative Position

Format

```
MRP x-coordinate , y-coordinate ;
```

Parameters

x-coordinate/y-coordinate:

coordinates as measured from the current cursor position

Function

The MRP command moves the cursor from its current position to a specified relative position, measured in the units designated by the UNIT command. The coordinates may be positive or negative.

The MRP command works in the standard mode graphics only. To move to a relative position in a path construction, use the PMRP (in Path, Move to Relative Position) command.

File

```
!R! RES; UNIT C;  
SFNT 'Helvetica-Bd';  
CMNT MRP and baseball;  
MRP 8, 8; CIR .2;  
TEXT ' 1st';  
MRP -2, -2; CIR .2;  
TEXT ' 2nd';  
MRP -2, 2; CIR .2;  
TEXT ' 3rd';  
MRP 2, 2; CIR .2;  
TEXT ' Home';  
PAGE;  
EXIT;
```

Printout

- 2nd
- 3rd 1st
- Home

Related Commands

MAP, MZP, MRPA

MRPA — Move to Relative Position specified by Angle

Format

MRPA *distance* , *angle* ;

Parameters

distance:

distance to move

angle:

angle in degree, measured clockwise from the vertical

Function


The MRPA command moves the cursor from its current position across a specified distance using a specified angle. The distance is measured in the units designated by the UNIT command.

The MRPA command is ignored in the path mode graphics. To move to a relative position specifying angle in a path construction, use the PMRA (in Path, Move to Relative position by Angle) command.

File

```
!R! RES; UNIT C;
CMNT Numbers indicate cursor movement;
SFNT 'Helvetica-Bd';
MAP 6, 12; CIR 1; TEXT '1';
MRPA 2, 60; CIR 1; TEXT '2';
MRPA 2, 120; CIR 1; TEXT '3';
MRPA 2, 180; CIR 1; TEXT '4';
MRPA 2, 240; CIR 1; TEXT '5';
MRPA 2, 300; CIR 1; TEXT '6';
PAGE;
EXIT;
```

Printout



Related Commands

MAP, MRP, MZP

MSTK — select Mailbox STacker

Format

MSTK [*tray, tray, ...*]; or MSTK 0;

Parameter

tray:

number from 1 to the maximum

Function

The MSTK command is valid only when the option sorter is installed with the printing system. It instructs the printing system to deliver printed pages in the mailbox mode. In the mailbox mode, the printed pages are delivered to the trays in the reverse order of tray parameters. The printed pages are delivered first to the last specified tray until it becomes full; when the last specified tray becomes full, then the pages are delivered to the second to the last specified tray, and so on until the first specified tray becomes full. When the first specified tray becomes full, the bulk tray receives the overflowed pages until it also becomes full.

When the bulk tray becomes full, the printing system pauses and displays a message which urges the user to remove paper. By removing all the pages in the bulk tray and pressing the ON LINE key, printing commences.

If the tray parameters are omitted, all trays excluding the bulk tray are used.

Upon receiving the MSTK command, the printer automatically performs a page feed.

The MSTK 0; command is applicable only for the SO-6. This command cancels the mailbox mode and delivers all pages in the bulk output tray.

File

```
!R! RES; MSTK; COPY; EXIT;  
PAGE 1 (2 copies, bottom tray) !R! PAGE; EXIT;  
PAGE 2 (2 copies, bottom tray) !R! PAGE; EXIT;
```

MTYP — select Media TYPE

Format

MTYP *media-type*;

Parameter

media-type:

- 1 = Plain
- 2 = Transparency
- 3 = Preprinted
- 4 = Label
- 5 = Bond
- 6 = Recycled
- 7 = Vellum
- 8 = Rough
- 9 = Letterhead
- 10 = Color
- 11 = Prepunched
- 12 = Envelope
- 13 = Cardstock
- 14 = Coated
- 15 = Second side
- 16 = Thick
- 21 = Custom 1
- 22 = Custom 2
- 23 = Custom 3
- 24 = Custom 4
- 25 = Custom 5
- 26 = Custom 6
- 27 = Custom 7
- 28 = Custom 8
- 99 = Not specified

Function

The MTYP command automatically selects the paper source (tray or cassette) depending on the given *media-type*. When the MTYP command is issued, the printing system looks for the paper source which has the matching media type. To assign a paper type to the paper cassette, use the operator panel or the FRPO X0 – X7 commands.

For the MTYP command to work, the automatic cassette switching mode must be set to on in the printing system. To do this, command: !R! PSRC 100; EXIT;. For details see the PSRC (set paper source) command section in this manual.

To ensure that the paper having the correct size and type of media is always selected, it is recommended to place the SPSZ (set page size) command following the MTYP command. For example, to automatically print on the plain A4 paper in the cassette one (topmost cassette), first assign the plain paper type to the cassette 1 by FRPO X1, 1;, then command MTYP 1; SPSZ 8;.

This command is valid only for the print systems that support the automatic media type function.

File

To print on the A4 label loaded on the printer's MP tray, set the MP tray size to A4 and the MP tray type to Labels (See the user's manual for details.)

```
!R! RES;  
PSRC 100;  
    CMNT Sets the printer to automatic cassette mode;  
    CMNT Depending on the current media type;  
    SPSZ  8; MTYP 4;  
    CMNT Paper size = A4, media type = label;  
RES; EXIT;
```

Related Commands

PSRC, SPSZ

MZP — Move to Zero-relative Position

Format

MZP *x-coordinate* , *y-coordinate* ;

Parameters

x-coordinate:

distance from the left edge limit of paper

y-coordinate:

distance from the top edge limit of paper


Function

The MZP command moves the cursor to a position specified relative to the top and left edge limits of the paper, measured in the units designated by the UNIT command. The coordinates may be positive or negative. Negative coordinates, however, move the cursor off the page, which should be avoided.

The MZP command is ignored in the path mode graphics. To move to a zero-relative position in a path construction, use the PMZP (in Path, Move to Zero-relative Position) command.

File

```
!R! RES; UNIT C; STM 2; SLM 2;
SFNT 'Helvetica-Nr';
MZP 0, 0; BOX 40, 40; SPD .2;
MZP 0, 4; TEXT '< = = = Edge limits';
MZP 0, 0; CIR .1; TEXT, L; TEXT ' MZP 0, 0;';
MZP 6, 0; CIR .1; TEXT, L; TEXT ' MZP 6, 0;';
MZP 0, 2; CIR .1; TEXT, L; TEXT ' MZP 0, 2;';
MZP 6, 2; CIR .1; TEXT, L; TEXT ' MZP 6, 2;';
PAGE;
EXIT;
```


Printout*Related Commands*

MAP, MRP, MRPA

NEWP — start NEW Path

Format

```
NEWP;
```

Parameter

None

Function

The NEWP command empties the current path and starts a new path. A path construction ends with the CLSP (CLOSe Path) command or with any painting operator that paints the path or draws a line along it (STRK, FILL).

For a full description of path, refer to *Chapter 1* of the *Technical Reference* manual.

File

```
!R! RES; UNIT C;  
  
NEWP;  
PMZP 5, 5;  
PDRP 1, -2, 3, 0, -1, 2, -3, 0;  
STRK; CMNT Stroking the path;  
  
NEWP;  
PMZP 10, 5;  
PDRP 1, -2, 3, 0, -1, 2, -3, 0;  
GPAT .5; FILL 1; CMNT Filling the path;  
  
PAGE; EXIT;
```

Printout



Related Commands

CLSP, FILL, STRK

OTRY — select paper Output TRaY

Format

```
OTRY tray;
```

Parameter

tray:

- 1 = printing system's face-down output tray
- 2 = face-up output tray of the printing system or the option stacker
- 3 = face-down output tray of the option stacker
- 4 = face-up output tray of the option stacker (subtray)
- 5 = face-down output tray of job separator 1
- 6 = face-up output tray of the option stacker (subtray2)
- 7 = face-down output tray of option stacker (subtray 1)
- 9 = face-down output tray of option stacker (subtray 2)
- 11 = face-down output tray of the option mailbox 1
- 12 = face-up output tray of the option mailbox 1
- 15 = face-down output tray of job separator 2
- 21 = face-down output tray of the option mailbox 2
- 22 = face-up output tray of the option mailbox 2
- 31 = face-down output tray of the option mailbox 3
- 32 = face-up output tray of the option mailbox 3
- 41 = face-down output tray of the option mailbox 4
- 42 = face-up output tray of the option mailbox 4
- 51 = face-down output tray of the option mailbox 5
- 52 = face-up output tray of the option mailbox 5
- 61 = face-down output tray of the option mailbox 6
- 62 = face-up output tray of the option mailbox 6
- 71 = face-down output tray of the option mailbox 7
- 72 = face-up output tray of the option mailbox 7

Function

The OTRY command instructs the printing system to deliver pages to the corresponding output stack of the printing system or the option stacker. The selected tray remains in effect until changed by another OTRY command. The default value is 1: the printing system's face-down tray.

If an optional device such as a stacker, a document finisher, or a sorter is installed, the *tray* value 2 instructs the printing system to deliver printed pages to its face-up tray; the value 3 delivers the pages to its face-down tray. The *tray* values of 5, 7, 9, 11, 12, 13, 14, 15, 21, 22, 31, 32, 41, 42, 51, 52, 61, 62, 71, and 72 are valid for certain copiers installed with the applicable option device.

The OTRY command should be placed at the top of a file or program.

File

```
!R! OTRY 2; EXIT;
```

```
PAGE 1
```

On printers with only one paper tray this file will be delivered face-down as usual, page 1 first.

```
!R! PAGE; EXIT;
```

```
PAGE 2
```

PAGE 2 next, and so on, stacking in the correct order.

```
!R! PAGE; EXIT;
```

```
PAGE 3
```

But on printers with two paper trays this file will be delivered to the face-up tray with the pages in reverse order, page 3 on top.

```
!R! PAGE; EXIT;
```

PAGE — start new PAGE

Format

```
PAGE ;
```

Parameter

None

Function

The PAGE command causes a page break without resetting the printer. Subsequent data will be printed on a new page with no changes made to page formatting parameters such as spacing, margins, font selection.

If there is no printing on the current page, the PAGE command will have no effect. It does not feed out a blank page.

File

```
!R! RES;  
SLM 1; STM 1;  
EXIT;  
Page 1:  
Following this line, a page break will occur.  
!R! PAGE; EXIT;  
Page 2:
```

PANT — create PANTone color palette

COLOR

Format

PANT *color-number* or '*color-name*' ;

Parameters

color-number:

Pantone color number, integer from 100 to 5999

color-name:

Pantone color name

Function

The PANT command can be used to specify the PANTONE uncoated color by number or name. (See <http://www.pantone.com> for details on Pantone Colors.)

The following should be noted when specifying a color by its PANTONE number.

- In PANTONE numbers, a C or U symbol is added after the number to distinguish between coated and uncoated. These symbols are not necessary when a color is specified using the PRESCRIBE command. In fact, the command becomes invalid if one of these symbols is added to a number. If there is no color corresponding to the number specified in the valid range, then the color black is specified.

The following should be noted when specifying a color by its PANTONE name.

- Always enclose character strings in single quotation marks or double quotation marks. Character strings may be entered in either upper case or lower case.
- In names that contain a combination of words and numbers, such as *Red 032* and *Cool Black 2*, a space must be inserted between the words and number. If spaces are not inserted, the words and numbers will be recognized as a different name, in which case the color will not be specified correctly.
- If a nonexistent name is specified, the command is ignored.

When PANTONE is specified, a color space RGB palette is created, and a color corresponding to the specified color number is assigned to the palette and selected. The color is always assigned to index number 0.

Colors may be added to the palette or changed. However, when a color is assigned to index number 0, the colors created with the PANT command are discarded.

The palette remains valid unless a new one is generated. Therefore, colors created with the PANT command can be selected using SCOL 0;

If the PANT command is issued multiple times (either in succession or not in succession), the color which is valid is the color corresponding to the last entered command. The colors corresponding to previously entered commands are deleted.

The palette ID number remains valid (active) and does not change.

A total of 45 colors can be specified by name (see list below). These colors must be spelled as shown, although the letters may be in upper case or lower case.

Process Yellow	Process Magenta	Process Cyan	Process Black
Process Blue			
Red 032	Warm Red	Rubine Red	Rhodamine Red
Blue 072 Reflex	Blue	Orange 021	Yellow
Green	Purple	Violet	
Black	Black 2	Black 3	Black 4
Black 5	Black 6	Black 7	
Warm Gray 1	Warm Gray 2	Warm Gray 3	Warm Gray 4
Warm Gray 5	Warm Gray 6	Warm Gray 7	Warm Gray 8
Warm Gray 9	Warm Gray 10	Warm Gray 11	
Cool Gray 1	Cool Gray 2	Cool Gray 3	Cool Gray 4
Cool Gray 5	Cool Gray 6	Cool Gray 7	Cool Gray 8
Cool Gray 9	Cool Gray 10	Cool Gray 11	

The PANT command may not work with certain printer models and versions. Set the print quality to Quick mode when this command is used with model FS-5800C.

PANTONE colors cannot be used by the PANT command in KPDLE emulations.

File

```
!R! PANT 'Cool Gray 1'; EXIT;
```

PARC — in Path, draw ARC

Format

PARC *x-coordinate* , *y-coordinate* , *radius* , *start-angle* ,
end-angle ;

Parameters

x-coordinate , *y-coordinate*:

coordinates of the center of arc, measured from the left and top edge limits

radius:

radius of arc

start-angle:

starting angle, measured counterclockwise from the positive *x* axis

end-angle:

ending angle, measured counterclockwise from the positive *x* axis

Function

This command first of all adds a line extending from the current position to the beginning of the arc, then adds an arc to the current path. Coordinates and radius are measured in the units designated by the UNIT command; the starting and ending angles are measured counterclockwise from the positive *x* axis.


To draw a simple arc as specified by the parameters, the cursor must be moved to the beginning of the arc by using the PMRA command. (See the example below.)

The arc is drawn at the starting angle through the ending angle in counterclockwise direction only. After the path is stroked, the cursor remains at the ending angle of the arc.

File

```
!R! RES; UNIT C;  
NEWP; SPD .1;  
PMZP 5.0, 10.0;  
PARC 6.5, 12.5, 1.5, 180, 270;  
STRK;  
  
NEWP;  
CMNT This program draws a circle;  
PMRA 6.5, 18, 1.5, 0;  
CMNT Moves cursor 1.5 cm by 0 degree;  
PARC 6.5, 18, 1.5, 0, 360;  
STRK;  
PAGE; EXIT;
```


Printout



PAT — select fill PATtern

Format

PAT *pattern-number*[, *resolution*];

Parameters

pattern-number:

number from 1 to 60 or 100 to 105

resolution:

300 [dpi], 600 [dpi], or 1200 [dpi]

Function


The PAT command specifies the shading pattern used by the BLK and ARC commands and to fill a path.

The *pattern-number* should be any number between 1 through 60 if a resident pattern is selected. A full list of the printer's resident patterns appear on the following page.

Pattern numbers 100 through 105 are reserved for selecting the expanded patterns and can be generated by using the XPAT command. If you specify a pattern number between 100 and 105 when no expanded patterns are defined, the printer selects the No. 1 (solid black) resident pattern.

Variable resolution support is valid for 600 dpi and 1200 dpi models only.

Resident Fill Patterns (300 dpi)



PCRP — in Path, Curve to Relative Position

Format

```
PCRP  $x1$ ,  $y1$ ,  $x2$ ,  $y2$ ,  $x3$ ,  $y3$ ;
```

Parameters

$x1$, $y1$:

coordinates of the first control point as measured from the current cursor position

$x2$, $y2$:

coordinates of the second control point as measured from the current cursor position

$x3$, $y3$:


coordinates of the end point as measured from the current cursor position

Function

The PCRP command adds a Bézier curve segment to the path, starting from the current cursor point, referred to as current position as shown at right, through two control points of $(x1, y1)$ and $(x2, y2)$, and ending at $(x3, y3)$.

Example:

```
NEWP ;  
PMZP 5 , 10 ;  
PCRP 2.5 , -4 , 7 , -5 , 7 , 0 ;  
STRK ;
```




File

```

!R! RES; UNIT C;
NEWP;
SPD .1;
PMZP 3, 3;
PCRP 4, 2, 5, 1.5, 6, 1.8;
STRK;
PAGE;
EXIT;


```

Printout*File*

```

!R! RES; UNIT C;
NEWP;
SPD .1;
PMZP 3, 3;
PCRP 5.5, 1.5, 1.5, 5.5, 7, 7;
STRK;
PAGE;
EXIT;

```

Printout

PCZP — in Path, Curve to Zero-relative Position

Format

```
PCZP x1, y1, x2, y2, x3, y3;
```

Parameters

x1, y1:

coordinates of the first control point as measured from the left and top edge limits of paper

x2, y2:

coordinates of second control point as measured from the left and top edge limits of paper

x3, y3:


coordinates of end point as measured from the left and top edge limits of paper

Function

The PCZP command adds a Bézier curve segment to the path, starting from the current cursor position through two controls points of (*x1, y1*) and (*x2, y2*), and ending at (*x3, y3*), measured from the left and top edge limits of the paper.

Example:

```
NEWP ;  
PMZP 1.5, 3.5 ;  
PCZP 4.5, 1.5, 5.5, 5.5, 8, 3.5 ;  
STRK ;
```




File

```
!R! RES; UNIT C;  
PMZP 3, 6;  
PCZP 11, 3, 2, 2, 9, 6;  
STRK;  
PAGE;  
EXIT;
```

Printout

● (2, 2)

● (11, 3)



PDIR — set Print DIRection

Format

PDIR *angle* ;

Parameter

angle:

angle in degrees; in 90° increments in a counterclockwise direction

0 = portrait

90 = landscape

180 = reverse portrait

270 = reverse landscape

Function


The PDIR command rotates the coordinates of the logical page by the degree represented by the value of the *angle* which orients the page in one of four directions. It does not change the current page orientation.

When the print direction is changed by the PDIR command, the logical page coordinate system is transformed so the margins maintain the same printable area (the area enclosed by the margins) as shown on next page. The current position uses the same coordinate values as the previous print direction; line-spacing, and the character-spacing of the current font are effected by the new print direction as are the raster and pattern graphics.


Scalable fonts should be used for printed text in different page directions. The PDIR command rotates only the coordinates of the logical page on the current page orientation and does not affect the current page orientation. A bitmap font prints text according to the current page orientation (it adheres to the current font mode).

The PDIR command does not cause a page break, so you can print text in four different directions on the same page as shown in the sample file on the next page.


Portrait (PDIR 0;)




Landscape (PDIR 90;)



Reverse Landscape (PDIR 270;)



Reverse Portrait (PDIR 180;)



File

```


!R! RES;
UNIT P; SFNT 'Helvetica', 8;
UNIT C; MZP 10, 12; SPD .144; CIR .075;

PDIR 270; TEXT ' ENDLESS VARIETY AND VERSATILITY';
PDIR 180; TEXT ' ENDLESS VARIETY AND VERSATILITY';
PDIR 90; TEXT ' ENDLESS VARIETY AND VERSATILITY';
PDIR 0; TEXT ' ENDLESS VARIETY AND VERSATILITY';

PAGE;
EXIT;

```

Printout



PDRP — in Path, Draw to Relative Position

Format

PDRP *x-coordinate*, *y-coordinate*, ... ;

Parameters

x-coordinate, *y-coordinate*:

coordinates as measured from the current cursor position

Function


The PDRP command adds a line to the current path from the current cursor position to a specified relative position, measured in the units designated by the UNIT command. The cursor moves to the specified coordinates. The coordinates may be positive or negative. Negative coordinates may draw the line extending outside the current page. The thickness of the line is designated by the SPD command.

Multiple pairs of coordinates may be specified. The printer then draws a connected series of line segments, proceeding from one point to the next. For example, PDRP 1, 2, 3, 4; is equivalent to PDRP 1, 2; PDRP 3, 4;.

File

```
!R! UNIT C; SPD .3;
  CMNT Cube;
  NEWP;
  PMZP 5, 15;
  PDRP 4, 0, 2, -2, 0, -4, -4, 0, -2, 2, 0, 4;
  CLSP;
  PMRP 4, 0;
  PDRP 0, -4, -4, 0, 4, 0, 2, -2;
  STRK;
  PAGE;
EXIT;
```

Printout



Related Command

PDZP

PDZP — in Path, Draw to Zero-relative Position

Format

PDZP *x-coordinate*, *y-coordinate* ;

Parameters

x-coordinate, *y-coordinate*:

coordinates as measured from the left and top edge limits of the current page

Function


The PDZP command adds a line to the current path from the current cursor position to a position specified relative to the top and left edge limits of the current page, measured in the units designated by the UNIT command. The cursor moves to the specified coordinates.

The coordinates may be positive or negative. Negative coordinates may draw the line extending outside the current page. The thickness of the line is designated by the SPD command.

File

```
!R! RES; UNIT C;
      NEWP;
      PMZP 0, 0; PDZP 0, 5;
      PMZP 1, 0; PDZP 0, 4;
      PMZP 2, 0; PDZP 0, 3;
      PMZP 3, 0; PDZP 0, 2;
      PMZP 4, 0; PDZP 0, 1;
      PMZP 5, 0; PDZP 0, 0;
      STRK;
      PAGE;
EXIT;
```

Printout



Related Command

PDRP

PELP — in Path, draw ELLipse

Format

```
PELP x-diagonal, y-diagonal[, x-start, y-start, x-end, y-end[, ellipse-path]];
```

Parameters

x-diagonal:

X-coordinate of diagonal point of bounding box

y-diagonal:

Y-coordinate of diagonal point of bounding box

x-start:

X-coordinate of start control point of the ellipse

y-start:

Y-coordinate of start control point of the ellipse

x-end:

X-coordinate of end control point of the ellipse

y-end:

Y-coordinate of end control point of the ellipse

ellipse-path:

ellipse options:

N = define an ellipse path


C = define an ellipse-based arc path

P = define an ellipse-based pie shape path

Function

The PELP command adds an ellipse-based path to the current path. This command cannot paint figures by itself. In addition, STRK or FILL is required. A complete ellipse path will be defined if the parameters in brackets are omitted. Executing this command will close current sub-paths.

An error will result and the command has no effect if the bounding box is specified as a straight line or a point.



File

```

!R!
RES;
UNIT C;

MZP 0, 0;
BOX 2, 3;
NEWP;
PMZP 0, 0;
PELP 2, 3, 2, 0, 0, 0, N;
STRK;

MZP 5, 5;
BOX 2, -3;
NEWP;
PMZP 5, 5;
PELP 2, -3, 2, 0, 0, 0, N;
STRK;

MZP 10, 5;
BOX -2, -3;
NEWP;
PMZP 10, 5;
PELP -2, -3, -2, 0, 0, 0, N;
STRK;

MZP 15, 0;
BOX -2, 3;
NEWP;
PMZP 15, 0;
PELP -2, 3, -2, 0, 0, 0, N;
STRK;

RES;
EXIT;

```


PIE — draw PIE chart

Format

PIE *radius*, *starting-angle*, *slice-size*, ...;

Parameters

radius:

radius of the pie chart

starting-angle:

position of the first slice

slice-size:

relative sizes of the slices

Function


The PIE command is a standard mode graphics command which draws a pie chart centered at the current cursor position. The sizes of the slices may be specified using any convenient non-negative integers. (PIE commands are ignored if they contain negative slices, or slices with decimal fractions.) The sum of the slice sizes should not exceed 9999. The total length of the command must not exceed 255 characters.

The printer automatically converts the slice sizes to angles totaling 360 degrees. It draws the first slice starting at the position specified by the second parameter (0 = vertically up), then draws the other slices in order, proceeding in a clockwise movement. The line thickness is designated by the SPD command.

File

```
!R! RES; UNIT C;  
MZF 10, 10;  
PIE 3, 0, 10, 20, 30, 40;  
PAGE;  
EXIT;
```

Printout



PMRA — in Path, Move to Relative position specified by Angle

Format

PMRA *x-coordinate*, *y-coordinate*, *distance*, *angle*;

Parameters

x-coordinate, y-coordinate:

coordinates as measured from the left and top edge limits of the page

distance:

distance specified in increments governed by the unit value.

angle:


angle in degrees, measured counterclockwise from the positive *x* axis

Function

The PMRA command moves the cursor to a position in a path construction, specified relative to the left and top edge limits of the page, then across a distance specified relative to the *x*- and *y*-coordinate and at a specified angle.

It should be noted that the *x* and *y* coordinates are measured from the top and left edge limits of the page, and not from the current cursor position.

The coordinates and distance are measured in the units designated by the UNIT command. The angle is measured counterclockwise from the positive *x* axis.



File

```
!R! RES; UNIT C; SPD 0.2;
```

```
NEWP;
```

```
PMRA 10, 19, 5, 45;
```

```
PARC 10, 19, 5, 45, 135;
```

```
STRK;
```

```
PAGE; EXIT;
```

Printout



PMRP — in Path, Move to Relative Position

Format

PMRP *x-coordinate*, *y-coordinate* ;

Parameters

x-coordinate, *y-coordinate*:

coordinates as measured from the current cursor position

Function


The PMRP command moves the cursor from its current position to a specified relative position in a path construction, measured in the unit value designated by the UNIT command.

The coordinates may be positive or negative.

File

```
!R! RES; UNIT C;  
NEWP;  
PMZP 5, 5;  
PDRP 5, 0, 0, 5, -5, 0, 0, -5; GPAT .2; FILL 1;  
PMRP 4, 4;  
PDRP 5, 0, 0, 5, -5, 0, 0, -5; GPAT .8; FILL 1;  
PAGE;  
EXIT;
```

Printout



Related Command

PMZP

PMZP — in Path, Move to Zero-relative Position

Format

PMZP *x-coordinate*, *y-coordinate* ;

Parameters

x-coordinate, *y-coordinate*:

coordinates as measured from left edge limit of the current page

Function


The PMZP command moves the cursor to a position in a path construction, specified relative to the top and left edge limits of the current page, and measured in the units designated by the UNIT command.

The coordinates may be positive or negative. Negative coordinates move the cursor off the current page.

File

```
!R! RES; UNIT C; STM 2; SLM 2;
SFNT "Helvetica-Nr";
NEWP;
PMZP 0, 0; SPD .2;
PMZP 0, 4; FONT 13; TEXT '< = = Paper edge limits';
PMZP 0, 0; PARC 0, 0, .1, 0, 360;
PMRP 0, .333; TEXT ' PMZP 0, 0;';
PMZP 4, 0; PARC 4, 0, .1, 0, 360;
PMRP 0, .333; TEXT ' PMZP 4, 0;';
PMZP 3, 6; PARC 3, 6, .1, 0, 360;
PMRP 0, .333; TEXT ' PMZP 3, 6;';
PMZP 7, 7; PARC 7, 7, .1, 0, 360;
PMRP 0, .333; TEXT ' PMZP 7, 7;';
STRK;
PAGE;
EXIT;
```

Printout



Related Command

PMRP

PNCH — PuNCH

Format

PNCH[*mode*] ;

Parameter

mode:

number of punched holes:

1 = 2 (Europe) or 3 (U.S.)

2 = 4 (Europe) or 2 (U.S.)

51 = 2 (Europe) or 3 (U.S.), left

52 = 4 (Europe) or 2 (U.S.), left

61 = 2 (Europe) or 3 (U.S.), right

62 = 4 (Europe) or 2 (U.S.), right

71 = 2 (Europe) or 3 (U.S.), top

72 = 4 (Europe) or 2 (U.S.), top

Function

The PNCH command lets the option finisher perform punching through the printed page with the number of holes depending on the country. *mode* selects either the standard or additional number of holes for the country. For example, *mode 1* selects 2-hole punching for European countries and *mode 2* selects 4-hole punching for European countries.



The *mode* parameters of 51, 52, 61, 62, 71, and 72 are valid for some copier models only. These parameters allow to position the holes in either of the left, right, and top sides of the page by logically aligning the page in reference to the punched holes.

Note that the *mode* parameter is valid for a finisher with the dual mode punching system, such as the DF-71, only. It becomes invalid if it is specified in a finisher with the standard punch unit. With a finisher with the standard punch unit only (e.g. the DF-75), the number of punched holes is uniquely determined according to the country of destination and paper size.

Like stapling and other finishing operations, punching is done per one job at a time. Therefore, this command is invalid unless it is used in combination with JOBS. See Page 140.

The option *finisher* is not switched automatically by this command to perform punching. Therefore, it is necessary in advance to specify the option *finisher* as the output destination using the STAK 3; command.

File

The following example is for European countries.

```
!R!  
  STAK 3; CMNT Stack to the finisher;  
  PNCH 1;  
  JOBS 1, 1, 2, 0, 0;  
  TEXT 'This page is punched with two holes.';  
  JOBT;  
EXIT;
```

PRBX — in Path, draw Round BoX

Format

PRBX *width, height, radius;*

Parameters

width:

width of round box

height:

height of round box


radius:

radius of round box corners

Function

The PRBX command adds a round box path to the current path. This command close the sub-paths but does not stroke the path by itself. Therefore, STRK or FILL is required to follow this command.

The parameter value for the width, height and radius of the round box should be positive.



File

```
!R! RES; UNIT C;  
SPD 0.125; NEWP;  
PMZP 1, 1;  
PRBX 11.7, 1.7, 0.5; STRK;  
MZP 1.5, 2.15;  
SFNT "Helvetica-Bd", 24;  
TEXT "CAUTION: HOT SURFACE!";  
PAT 5; NEWP;  
PMZP 1, 4;  
PRBX 5, 2.4, 0.3; FILL;  
MZP 1.4, 4.7;  
SFNT "Helvetica-Nr", 12;  
TEXT "Barefoot Systems, Inc.", L;  
TEXT "4203 Charleston Avenue", L;  
TEXT "P.O. Box 150", L;  
TEXT "Edinburgh, CA 94039-1057";  
  
PAGE;  
EXIT;
```

Printout

CAUTION: HOT SURFACE!

Barefoot Systems, Inc.
4203 Charleston Avenue
P.O. Box 150
Edinburgh, CA 94039-1057

PRRC — in Path, at Relative position, draw aRC

Format

PRRC *x-coordinate*, *y-coordinate*, *radius*, *start-angle*, *end-angle*;

Parameters

x-coordinate/y-coordinate:

coordinates as measured from the current cursor position

radius:

radius of arc

start-angle:

starting angle, measured counterclockwise from the positive *x* axis

end-angle:


ending angle, measured counterclockwise from the positive *x* axis

Function

The PRRC command first of all adds a line extending from the current position to the beginning of the arc which is centered at the coordinates relative to the current position, then adds the line and the arc to the current path.


Coordinates and radius are measured in the units designated by the UNIT command; the starting and ending angles are measured counterclockwise from the positive *x* axis. For example, the following example draws an arc extending over the center of the arc:

```
PRRC 1, 1, 0, 180;
```



Whereas, the following command, having the reversed starting and ending angles, draws the arc under the center of the arc:

```
PRRC 1, 1, 180, 0;
```




Negative values are acceptable for angle. For example, an angle of 315 degrees is the same as -45 degrees.

File

```
!R! RES;  
SPD 0.05;  
MZP 4.5, 5;  
NEWP;  
PRRC -1, 0, 1, 90, 270;  
PRRC 1, -2, 1, 0, 180;  
PRRC 2, 1, 1, -90, 90;  
PRRC -1, 2, 1, 180, 0;  
STRK;  
PAGE;  
EXIT;
```

Printout



PSRC — select Paper SouRCe

Format

PSRC *mode* ;

Parameter

mode:

0 = multi-purpose tray
 1 = Cassette 1
 2 = Cassette 2
 3 = Cassette 3
 4 = Cassette 4
 5 = Cassette 5
 6 = Cassette 6
 99 = envelope feeder
 100 = automatic

Function

The PSRC command selects whether to feed paper from the cassette, multi-purpose tray, cassettes of the paper feeders, or the envelope feeder. Each cassette of the paper feeder is assigned a cassette number in the order from the top as 1 through 6.

Mode 100 of PSRC enables the automatic cassette switching depending on media type and size. The printing system automatically selects the paper cassette having the paper that matches the type and size defined by MTYP and SPSZ respectively.

The PSRC command causes a page break, so it should be placed at the beginning of a file or program, or at the top of a new page. The current paper source will remain selected if the specified cassette is not installed.

File

```
!R! RES; UNIT C;

SFNT "Helvetica-Nr", 24;
PSRC 1;
TEXT "This page is fed from the default cassette.";
PAGE;
PSRC 0;
TEXT "This page is fed from the multi-purpose tray."
PSRC 100; SPSZ 10;
TEXT "This page is printed on a A3 sheet that the print
system looks for automatically.";
PAGE;
EXIT;
```

PXPL — PiXel PLacement

Format

PXPL *mode* ;


Parameter

mode:

- pixel placement mode
- 0 = Grid intersection
- 1 = Grid centered

Function

The PXPL command determines how pixels are rendered in images. It selects whether pixels are centered at grid intersections or inside the boxes created by the grid. See the figure below.



Assuming a rectangle extending from coordinate (1, 1) to coordinate (3, 4), a different result is obtained depending on each pixel placement mode. Grid intersection is default to HP PCL; and grid centered mode is used by Microsoft Windows.

This command is ignored when used with DAP, DRP, DRPA, DZP, BOX, BLK, CIR, ARC, or PIE. This command has no effect for text or images.

The printing system is reset to grid intersection mode by the RES command.

!R! — PRESCRIBE start sequence

Format

!command-recognition-character!

Parameter

command-recognition-character:

capital R, unless changed by the SCRC or FRPO P9 command

Function

When the printing system receives !R!, it enters PRESCRIBE mode and treats the following data as PRESCRIBE commands. The command that follows the !R! command should be separated from the !R! by a space or placed on a new line.

The command recognition character can be changed from capital R to another character. Temporary changes can be made using the SCRC (Set Command Recognition Character) command described later. The RES command restores the command recognition character to its FRPO-set value. Permanent changes can be made with the FRPO command described in *Chapter 6* of the *Technical Reference* manual.


If !R! is used by mistake when the printer is already in the PRESCRIBE mode, the printer ignores the mistaken !R! and whatever follows, up to the next semicolon.

File

```
!R! SFNT 'Helvetica', 14; EXIT;
Following these five
lines, the printer
enters PRESCRIBE
mode to execute
commands.
!R! UNIT I;
BOX 2, 2;
BOX 1, 1, E;
CIR 1;
BOX 1, 1;
PAGE;
EXIT;
```

Printout

Following these five lines, the printer enters PRESCRIBE mode to execute commands.



Related Commands

EXIT, SCRC

RCLT — Rotated CoLLaTion

Format


RCLT;

Parameter

None

Function

The RCLT command instructs the printing system to stack the sets on the tray in alternating orientations. The RCLT command require a hard disk installed in the printing system or a RAM disk.



The RCLT command is available when printing using A4, B5 and letter size paper. During printing, the same size paper must be loaded into two separate cassettes. For example, Portrait (A4) and Landscape (A4-R) would be used when printing onto A4 paper. The MP tray can also be used as a paper source.

File

```
!R! RES; RCLT; STAK 1;
JOBS 0, 5, 0, 0;
TEXT "Page 1"; PAGE;
TEXT "Page 2"; PAGE;
TEXT "Page 3"; PAGE;
JOB; EXIT;
```

RDMP — Received-data DuMP

Format

RDMP ;

Parameter

None

Function

The RDMP command causes all subsequent data to be printed in hexadecimal form (two hexadecimal digits per character) until an ENDD (END Dump) command is received, or until the printing system runs out of data and times out.

Related Command

ENDD

Sample

```

1B 25 2D 31 32 33 34 35 58 40 50 4A 4C 20 45 4E      .%-12345X@PJL EN
54 45 52 20 4C 41 4E 47 55 41 47 45 3D 50 43 4C      TER LANGUAGE=PCL
0A 1B 45 1B 26 6C 31 47 1B 26 6C 31 58 1B 26 6C      ..E.&l1G.&l1X.&l
30 53 1B 26 6C 30 6F 32 61 34 64 31 65 34 32 46      OS.&l0o2a4dle42F
1B 26 6C 31 48 1B 2A 74 33 30 30 52 1B 2A 70 30      .&l1H.*t300R.*p0
59 1B 2A 70 30 58 1B 2A 70 32 33 32 58 1B 2A 70      Y.*p0X.*p232X.*p
2B 33 39 39 59 1B 2A 62 32 4D 1B 2A 72 31 41 1B      +399Y.*b2M.*r1A.
2A 62 31 36 57 00 01 F8 FF 00 E0 F8 00 01 1F F0      *b16W.....
F3 00 01 3F E0 1B 2A 72 42 1B 2A 70 2D 31 59 1B      ...?..*rB.*p-1Y.
2A 70 31 32 38 38 58 1B 2A 72 31 41 1B 2A 62 33      *p1288X.*r1A.*b3
57 01 FE 1F 1B 2A 72 42 1B 2A 70 2D 31 59 1B 2A      W....*rB.*p-1Y.*
70 31 35 39 32 58 1B 2A 72 31 41 1B 2A 62 31 36      p1592X.*r1A.*b16
57 00 03 FD FF 00 E0 E1 00 00 07 FD FF 02 E1 FF      W.....
E0 1B 2A 72 42 1B 2A 70 32 33 32 58 1B 2A 72 31      ..*rB.*p232X.*r1
41 1B 2A 62 32 33 57 00 01 F8 FF 04 E0 7F FF FF      A.*b23W.....
F0 FD 00 03 01 FF FF 80 F5 00 02 03 FF FE 1B 2A      .....*
72 42 1B 2A 70 2D 31 59 1B 2A 70 31 32 38 30 58      rB.*p-1Y.*p1280X
1B 2A 72 31 41 1B 2A 62 35 57 03 07 FC 7F 80 1B      .*r1A.*b5W.....
2A 72 42 1B 2A 70 2D 31 59 1B 2A 70 31 35 39 32      *rB.*p-1Y.*p1592
58 1B 2A 72 31 41 1B 2A 62 32 34 57 00 1F FD FF      X.*r1A.*b24W....
00 E0 EB 00 04 01 FF FF FF C0 FC 00 00 07 FD FF      .....
02 E1 FF FC 1B 2A 72 42 1B 2A 70 32 33 32 58 1B      .....*rB.*p232X.
2A 72 31 41 1B 2A 62 32 34 57 00 01 F8 FF 04 E0      *r1A.*b24W.....
7F FF FF F0 FD 00 03 03 FF FF C0 F5 00 03 0F FF      .....
FF 80 1B 2A 72 42 1B 2A 70 2D 31 59 1B 2A 70 31      ...*rB.*p-1Y.*p1
32 38 30 58 1B 2A 72 31 41 1B 2A 62 35 57 03 1F      280X.*r1A.*b5W..
FD FF E0 1B 2A 72 42 1B 2A 70 2D 31 59 1B 2A 70      ....*rB.*p-1Y.*p
31 35 38 34 58 1B 2A 72 31 41 1B 2A 62 32 34 57      1584X.*r1A.*b24W
00 03 FC FF 00 E0 EB 00 04 01 FF FF FF C0 FC 00      .....
00 07 FD FF 02 E1 FF FF 1B 2A 72 42 1B 2A 70 32      .....*rB.*p2
33 32 58 1B 2A 72 31 41 1B 2A 62 32 34 57 00 01      32X.*r1A.*b24W..
F8 FF 04 E0 7F FF FF F0 FD 00 03 07 FF FF F0 F5      .....
00 03 1F FF FF E0 1B 2A 72 42 1B 2A 70 2D 31 59      .....*rB.*p-1Y
1B 2A 70 31 32 38 30 58 1B 2A 72 31 41 1B 2A 62      .*p1280X.*r1A.*b
35 57 03 3F F9 FF E0 1B 2A 72 42 1B 2A 70 2D 31      5W.?....*rB.*p-1
59 1B 2A 70 31 35 38 34 58 1B 2A 72 31 41 1B 2A      Y.*p1584X.*r1A.*
62 32 35 57 00 0F FC FF 00 E0 EB 00 04 01 FF FF      b25W.....
FF C0 FC 00 00 07 FD FF 03 E1 FF FF C0 1B 2A 72      .....*r
42 1B 2A 70 32 33 32 58 1B 2A 72 31 41 1B 2A 62      B.*p232X.*r1A.*b
32 34 57 00 01 F8 FF 04 E0 7F FF FF F0 FD 00 03      24W.....
0F FF FF F8 F5 00 03 1F FF FF F0 1B 2A 72 42 1B      .....*rB.
2A 70 2D 31 59 1B 2A 70 31 32 38 30 58 1B 2A 72      *p-1Y.*p1280X.*r
31 41 1B 2A 62 35 57 03 FF F9 FF F0 1B 2A 72 42      1A.*b5W.....*rB
1B 2A 70 2D 31 59 1B 2A 70 31 35 38 34 58 1B 2A      .*p-1Y.*p1584X.*
72 31 41 1B 2A 62 32 37 57 06 3F FF FE FF FF FF      r1A.*b27W.?.....
E0 EB 00 04 01 FF FF FF C0 FC 00 00 07 FD FF 03      .....
E1 FF FF F0 1B 2A 72 42 1B 2A 70 32 33 32 58 1B      .....*rB.*p232X.

```

RES — RESet

Format

RES ;

Parameter

None

Function

The RES command causes a page break and resets parameters as indicated below.

Some reset values can also be changed by the FRPO command. For these, the factory reset value is given, followed by the FRPO parameters that can be used to change this value in parenthesis.

In HP-7475A emulation, the RES command resets the printer environment in the same manner as the KC-GL DF instruction. Refer to section *HP 7550A Emulation [KC-GL] (Mode 8)* of the *Technical Reference* manual.

The RES command does not:

- delete permanent macros or permanent fonts from memory
- delete positions or fonts stored by the SCP and SCF commands
- reset font attributes set by the SFA command
- change any values in permanent memory (Refer to the FRPO INIT command page.)

Parameter	Value after reset (Related FRPO parameter)			Related PRE- SCRIBE Com- mands
	LaserJet (PCL)	IBM/Epson	Diablo	
Command recognition character	R (P9)	←	←	SCRC
Emulation mode	6 (P1) ^a	←	←	SEM
Unit of measurement	Inches	←	←	UNIT
Number of copies per page	1 ^a	←	←	COPY
Cassette	Cassette 1 (main cassette) (R4) ^a		←	PSRC
Paper tray	Face-down (R0) ^a	←	←	STAK
Paper size	Cassette size (R2) ^a	←	←	SPSZ
Top margin	1/2 inch (150 dots)	0 (A1/2)	←	STM
Left margin	Left logical	0 (A3/4)	←	SLM
Right margin	Right logical	Max. size (A5/6)	←	SRM, SPW
Bottom margin	1/2 inch (150 dots)	Max. size (A7/8)	←	SBM, SPL, SLPP
Page orientation	Portrait (Orientation of font C5/2/3)	Portrait (Orientation of font C5/2/3)	Portrait (C1) [†]	SPO
Line spacing	6 lpi (U0/1)	←	←	SLS, SLPI
Character spacing	10 cpi (Spacing of font C5/2/3)	10 cpi (U2/3)	←	SCS, SCPI

Parameter	Value after reset (Related FRPO parameter)			Related PRE- SCRIBE Com- mands
	LaserJet (PCL)	IBM/Epson	Diablo	
Underline thickness and distance below baseline	4 dots, 7 dots	←	←	SULP
Current font	Font 1 (C5/2/3) or V3 (scalable font)†	Font 8 (IBM) Font 1 (Epson)	Font 1 (C5/2/3) or V3 (scalable font) ^a	FONT, FSET, SFNT
Alternate font assignment	None	See <i>Emulation of the Technical Reference manual</i> .	←	SETF
Country code	0 (U6)†	←	←	INTL
Symbol set	Roman-8 (U6/7)	See <i>Emulation of the Technical Reference manual</i> .†	US ASCII (U7)†	CSET, INTL
Current path	Empty	←	←	NEWP
Graphics pen diameter	3 dots	←	←	SPD
Line cap	Square	←	←	SCAP
Line join	Beveled join	←	←	SLJN
Miter limit	10	←	←	SMLT
Flatness	1	←	←	FLAT
Dash pattern	Solid line	←	←	DPAT, SDP
Clipping rectangle	Size of printable page		←	CLPR
Fill pattern	Solid black	←	←	PAT, FPAT, GPAT
Image model	Transparent [3]	←	←	SIMG
Graphics resolution	75 dpi	←	←	STR
CR action	CR	CR (P2)	←	–
LF action	LF	LF (P3)	←	–

^a These values can be set as permanent using the prin system's operator panel only when the FRPO C6 (panel keep) is active with a parameter of 1 (default).

RESL — select RESoLution

Format

RESL *resolution* ;

Parameter

resolution:

300 [dpi], 600 [dpi] or 1200 [dpi]

Function

The RESL command switches the resolution of the printing system between 300 dpi, 600 dpi, and 1200 dpi. When this command is received, the image data currently remaining is forcibly output on paper. This does not affect the current unit of measure, but the system's memory will be reconfigured, causing PCL resources (fonts, macros, patterns, and symbol sets) which have been downloaded into RAM to be purged.

The printing system's resolution setting enabled by this command is reset to default (FRPO N8), if a command which results in reset (RES, FRPO and SEM) is received. Actual resolution may be decreased to 300 dpi even though printing resolution is set to 600 dpi or 1200 dpi when there is insufficient memory because a duplex unit is installed and duplex mode has been set or because the video memory size has been increased for page protection.

Printing resolution cannot be switched from 300 dpi to higher resolutions for the following emulations: line printer, IBM ProPrinter, Diablo 630, EPSON LQ850.

RGBL — control RGB Level

COLOR

Format

RGBL *color, level;*

Parameters

color:

0 = red
1 = green
2 = blue


level:

number from -10 to 10 (Default = 0)

Function

The RGBL command controls balance for each level of RGB. It individually corrects the gamma for each color.

The color bars below represent color balance variations for each of RGB. The top bar represents the default saturation variation. The output result for an input value is maintained in linear form. The middle color bar is an example showing an increase in output for the input value. Indicates $\gamma > 1.0$ when being equivalent to the gamma curve. The bottom color bar is an example showing a decrease in output for the input value. Indicates $\gamma < 1.0$ when being equivalent to the gamma curve.



Varying color balance allows the output image results to vary. Issuing this command resets hue, saturation, and luminosity control values to their default settings.

This command is enabled for all color spaces, all color modes and all monitor simulation modes.

If ink simulation is set to other than none, this command is disabled. The RES command resets saturation to its default state.

This mode is not held in the printing system. For this reason, color balance is reset to its default state when the power is turned on.

RGST — offset ReGiSTration

Format

RGST *horizontal-offset* , *vertical-offset* ;

Parameters

horizontal-offset:

horizontal offset value from the absolute coordinate (0, 0) for the physical paper feed direction, in current unit. The coordinates may be positive or negative.

vertical-offset:

vertical offset value from the absolute coordinate (0, 0) for the physical paper feed direction, in current unit. The coordinates may be positive or negative.

Function

This command moves the coordinate origin by the offset (physical horizontal and vertical directions) as measured from the values specified. The intersection (shared area) of the current printable area and the printable area shifted by the new offset registration is defined as the valid printable area.

No page break is caused when this command is received. The offset values set using this command are valid even when the printing system is in HP emulation mode.

Since this command is PCL command-compatible (<Esc>&l #U , <Esc>&l #Z), the offset values are cleared to zero only by a software reset (<Esc>E). In other words, the offset is only cleared by PRESCRIBE (RES and FRPO INIT commands) when the emulation is HP.

An error will result and the command has no effect if a command format other than the above is used. An error will result and the value in question will be ignored if *horizontal-offset* or *vertical-offset* is specified as a value exceeding the physical dimensions of the current paper size.

RPCS — Return to Previous Code Set

Format

RPCS ;

Parameter

None

Function

The RPCS command resumes use of a code set saved by a previous SCCS (save current code set) command. The RPCS commands return to the code sets in reverse order in which they were nested.

File

See *SCCS* on page 268.

Related Commands

RPF, SCCS, SCF

RPF — Return to Previous Font

Format

```
RPF;
```

Parameter

None

Function

The RPF command resumes use of a font saved by a previous SCF (save current font) command. The RPF commands return to the fonts in the reverse order in which they were saved.

SCF and RPF are particularly useful in macros.

File

```
!R! RES; UNIT P;  
MCRO SMALLPRINT;  
SCF; FSET 10V; TEXT %1, E;  
RPF;  
ENDM;  
SFNT 'Helvetica', 12; SLS 14;  
EXIT;  
Computer literature is littered with acronyms such  
as !R! CALL SMALLPRINT, 'CPU, DOS, EEPROM';  
EXIT; and the like. Small capitals  
make these easier to read. Compare the effect of  
printing CPU, DOS, and EEPROM in regular capitals.  
!R! PAGE; EXIT;
```

Printout

Computer literature is littered with acronyms such as CPU, DOS, EEPROM and the like. Small capitals make these easier to read. Compare the effect of printing CPU, DOS, and EEPROM in regular capitals.

RPG — Return to Previous Graphics state

Format

RPG;

Parameter

None

Function

The RPG command restores the graphics state most recently saved by a SCG (Save Current Graphics) command. The following items of graphics state revert to the state saved previously.

- Path and cursor position (if defined)
- Pen diameter (line width)
- Line join type
- Line cap type
- Miter limit
- Dash pattern
- Flatness
- Fill pattern (16 × 16 dots)
- Clipping rectangle
- Raster resolution
- Image model


It is useful to save the graphics state then later restore it when you want to use a path for both stroking and filling. One such example is shown below.

The RPG command does not affect other printing system parameters such as page orientation, unit of measurement, etc. (See *RES* on page 220.)

File

```
!R! RES; UNIT C;
NEWP;
PMZP 4, 2;
PARC 3, 3, 1, 90, 270;
PARC 5, 3, 1, 270, 90;
CLSP;
SCG; CMNT Saves the graphics state;
STRK; CMNT Strokes the path;
RPG; CMNT Restores the graphics state;
GPAT .5;
FILL 1; CMNT Fills the path;
PAGE;
EXIT;
```

Printout



Related Command

SCG

RPP — Return to Previous Position

Format

RPP;

Parameter

None


Function

The RPP command moves the cursor to a position saved by a previous SCP (save current position) command. If the RPP command is executed by mistake when no previous position has been saved, the resulting cursor movement is unpredictable.

The RPP command can be used in both standard mode and path mode graphics.

File

```
!R! RES; UNIT P;
SFNT 'Helvetica-BdOb', 16;
UNIT C;
MAP 8, 8; CIR .2;
MRP -.1, .1;
CMNT Save this position seven times;
SCP; SCP; SCP; SCP; SCP; SCP; SCP;
MRPA 2, 0; TEXT 'N';
RPP; MRPA 2, 45; TEXT 'NE';
RPP; MRPA 2, 90; TEXT 'E';
RPP; MRPA 2, 135; TEXT 'SE';
RPP; MRPA 2, 180; TEXT 'S';
RPP; MRPA 2, 225; TEXT 'SW';
RPP; MRPA 2, 270; TEXT 'W';
RPP; MRPA 2, 315; TEXT 'NW';
PAGE;
EXIT;
```


Printout*Related Command*

SCP

RPPL — Return to Previous PaLette

COLOR

Format

RPPL;

Parameter

None

Function

The RPPL command restores the color palette most recently saved by a SCPL (save current palette) command. A color palette restored using the RPPL command immediately becomes active.

If the saved palette does not exist, this command is ignored.

File and Printout

See *SCPL* on page 278.

RPU — Return to Previous Unit

Format

RPU ;

Parameter

None

Function

The RPU command brings the unit of measurement (inches, centimeters, points, or dots; as specified by the UNIT command) back to the unit saved by the SCU (save current unit) command.

File/Printout

See *SCU* on page 283.

Related command

SCU

RTTX — RoTate TeXt

Format

```
RTTX angle, 'string' [, U-option];
```

Parameters

angle:

angle of rotation in degrees

string:

character string

U-option:

Underline text

Function

Taking the normal output angle for text as zero, the RTTX command rotates a character string around its starting point by the angle of rotation specified.

After this command is executed, text output returns to original orientation.

This command is ignored if the current font is a bitmap font.


File

```
!R! RES; PSRC 1;

SFNT "Helvetica", 10;
MZF 3, 3;
RTTX 15, " ----- Leo, 23 - August 22";
RTTX 45, " ----- Cancer, 22 - July 22";
RTTX 75, " ----- Gemini, 21 - June 21";
RTTX 105, " ----- Taurus, 20 - May 20";
RTTX 135, " ----- Aries, 21 - April 19";
RTTX 165, " ----- Pisces, 19 - Mar 20";
RTTX 195, " ----- Aquarius, 20 - February 18";
RTTX 225, " ----- Capricorn, 22 - January 19";
RTTX 255, " ----- Sagittarius, 22 - December 21";
RTTX 285, " ----- Scorpio, 24 - November 21";
RTTX 315, " ----- Libra, 23 - October 23";
RTTX 345, " ----- Virgo, 23 - September 22";

PAGE; EXIT;
```

Printout



RTXT — print Right-aligned TeXT

Format

```
RTXT[ 'string' ][ , option ][ , U-option ] ;
```

Parameters

string:

any character string

option:

B = Leave cursor at back end of string
E = Move cursor to front end of string
L = Move cursor down one line
N = Move cursor to beginning of next line

U-option:

Underline text

Function

The RTXT command prints the given string, ending at the cursor location, then moves the cursor as designated by the option. It accepts either a bitmap font or a scalable font to print the string.

The string must be enclosed in either single or double quotation marks. (If the string includes one kind of quotation mark, use the other kind to enclose it.) If the string is omitted, nothing is printed. If the option is omitted, the cursor is not moved (same as option B). If the U-option is specified, the string is underlined. The underline thickness and location can be set by the Sulp command.

The RTXT command can be used to position a line of text so that it ends at an exact location. The RTXT command should be executed between left and right margins.

The length of the string is unlimited, except that the command as a whole may not be longer than 255 characters. Carriage-return and linefeed codes in the string are ignored.

The RTXT command can be used both in the standard mode and path mode graphics.

File

```
!R! RES;
UNIT P;
SFNT 'TimesNewRoman';
UNIT I;
MCRO PF;
MRP .15, 0;
BOX .1, -.1; MRP .5, 0;
BOX .1, -.1; MRP -.65, 0;
ENDM;
MAP 3, 3; TEXT 'Pass      Fail', L, U;
CALL PF;  RTXT 'Unit test', L;
CALL PF;  RTXT 'Integration test', L;
PAGE;
EXIT;
```

Printout

	<u>Pass</u>	<u>Fail</u>
Unit test	<input type="checkbox"/>	<input type="checkbox"/>
Integration test	<input type="checkbox"/>	<input type="checkbox"/>

RVCD — Receive Compressed raster Data

Format

RVCD[*compression-mode*]; *length*, *data*, . . . ;

Parameters

compression-mode:

0 = uncompressed
 1 = run-length encoding
 2 = TIFF (Tagged Image File Format) [rev. 4.0]

length:

bytes of raster data

data:

binary raster data, 0H to FFH

Function

The RVCD command transfers and encodes the subsequent raster data by the amount of bytes represented by the *length* parameter.

The STR (SeT dot Resolution) command sets the dot resolution for raster graphics which starts at the current cursor position. The SRO (Set Raster Options) command specifies the orientation of the raster image in either physical (default) or logical page direction.

If the *compression-mode* value is zero [or omitted], the printing system simply transfers raster data in an uncompressed binary format. A single bit means a single dot. Seventh bit in the first byte refers to the starting dot within that raster row and bit 0 refers to the eighth dot and continues.

In the *run-length* encoding mode, the printing system treats raster data in pairs of bytes. The first byte of each pair is the repetition count for the data in second pattern byte. The repetition count accepts any value within a range of 0H to FFH.

compression-mode 2, TIFF rev. 4.0, translates raster data to be *packbits* in which a control byte precedes the raster data. A TIFF rev. 4.0 encoded data arrangement appears below.

control byte (1 byte) + data (1–128 bytes) + *control byte*
 (1 byte) + data (1–128 bytes) ...

A control byte with a negative value from -127 to -1 (81H to FFH) means the repetition count for the subsequent data. A zero to 127 (0H to 7FH) indicates the number of data bytes that follow. A control byte of -128 (80H) represents a non-operative control byte.

Related Commands

ENDR, STR

RVCL — ReceiVe CoLor raster data

COLOR

Format

RVCL [*compression-mode*,] *length*, *data*;

Parameters

compression-mode:

- 0 = uncompressed (Default)
- 1 = run-length encoded
- 2 = TIFF (Tagged Image File Format) [rev. 4.0]

length:

bytes of raster data

data:

binary raster data, 0H to FFH

Function

The RVCL command is the color version of the RVCD (ReceiVe Compressed raster Data) command. It transfers the color raster data. Color specification must be made to each pixel. An 8 bps color specification is used for the color data for each pixel. Also, this command is only valid for an RGB color space. Accordingly, 3 bytes of data must be sent for each pixel to be drawn when transferring color images.

Although this command handles monochrome or color data for encoding, since it relies on the active palette, the correct palette matching the print data must be used.

The end of raster data must be declared by the ENDR (END Raster) command.

When transferring color data using RVCL (ReceiVe CoLor raster data), always specify a palette using the CID (Configure color-Image Data) command. Data will not be properly processed if the palette is specified using the SIMP (SIMPlE color palette) or PANT (PANTone color palette) commands.

If the image data actually sent exceeds the width or height of the specified source data, the excessive portion will be truncated, resulting a clipped image.

The following examples produce invalid raster data lines:

Wrong:

```
0, ; Count cannot be 0.
1; Use 1, ; for a blank line.
2, 1, 1, 1; Too many data values.
2, -1, -1; Negative values are not allowed.
2, 1
, 1; Carriage return before comma.
```

An invalid raster data line is not printed, and invalidates all raster lines that follow.

Again, the following example demonstrates how to create the raster graphic on the previous page.

```
!R! STR 75; RVRD;
2;A|0;
2;GG0;
2;^S;
2;~{>>;
2;^S;
2;GG0;
2;A|0;
ENDR;
PAGE; EXIT;
```

The value 2 in each raster data line indicates the number of data which follows. The subsequent data can be obtained by following the same procedure as explained in *Defining Fill Patterns* in *Chapter 2* of the *Technical Reference* manual.

RWER D — [Read/Write External Resource] Delete data on external media

Format

RWER D, *medium*, *data-type*, '*partition-name*' ;

Parameters

medium:

0 = hard disk or RAM disk
A = memory card

data-type:

type of data to be deleted
1 = Font data
2 = Printable data
16 = Macro data
112 = File data
128 = API data (downloadable program)

partition-name:

Name of partition from which data is deleted

Function

The RWER D command deletes data on any external media whether it is a memory card or the hard disk. This command has no effect if the media has been write protected using KPDL or PJL.

The partition name may be a maximum of 15 printable ASCII characters (20H through 7EH); upper and lower case letters are regarded distinct. A partition name may not have leading spaces, and in the command, the partition name must be enclosed in single or double quotation marks.

A partition list is not automatically printed when deleting data on external media, even though the automatic printing of partition information was enabled (by the RWER I command). (See page 244.)

A command error results and the command has no effect if parameters other than the above are used.

This command causes a page break.

RWER F — [Read/Write External Resource] Format external media

Format

RWER F, *medium*;

Parameter

medium:

0 = hard disk
A = memory card

Function

The RWER F command formats any external media whether it is a memory card or the hard disk. The hard disk is physically formatted when it is specified. This command has no effect if the media has been write-protected using KPDL or PJJ.

A command error results and the command has no effect if parameters other than the above are used.

This command causes a page break and prints a format information page (if the medium is a compact flash card) when executed.

RWER I — [Read/Write External Resource] automatically print partition Information

Format

RWER I, *mode*;

Parameter

mode:

control mode:

E = Enable automatic printing of partition information

D = Disable automatic printing of partition information

Function

The RWER I command toggles the automatic printing of partition lists for any external media whether it is a memory card or the hard disk. With the automatic printing of partition lists enabled (RWER I, E;), the printing system prints partition lists whenever data are written to external media.

The setting made with this command returns to default when the power is turned off.

A command error results and the command has no effect if parameters other than the above are used.

RWER L — [Read/Write External Resource] print partition List

Format

RWER L[, *medium*] ;

Parameter

medium:

0 = hard disk or RAM disk
A = memory card
E = EPROM

Function

The RWER L command explicitly prints partition lists for any external media whether it is a memory card, the hard disk, or the RAM disk. The lists of partitions for all external media can be printed by omitting parameter *medium*.

Disk capacity, font cache capacity and data type are listed, with login file names for file data output in separate columns.

A command error results and the command has no effect if parameters other than the above are used.

RWER R — [Read/Write External Resource] Read data from external media

Format

RWER R, *medium*, *data-type*, '*partition-name*' ;

Parameters

medium:

0 = hard disk or RAM disk
 A = memory card
 E = EPROM

data-type:

2 = Printable data
 128 = API data (downloadable program)

partition-name:

Name of partition from which data is read, a maximum of 15 printable ASCII characters

Function

The RWER R command reads data from any external media whether it is a memory card, hard disk, EPROM, or RAM disk.

Only data types 2 and 128 are allowed. Note that reading data of other data types, such as fonts and macros, does not require this command since the printing system automatically reads them at power up.

If the same partition name exists on different external media, the priority with which data is read is as follows.

High	>	Low
Card	Hard disk (RAM disk)	EPROM

The partition name may be a maximum of 15 printable ASCII characters (20H through 7EH); upper and lowercase letters are regarded as distinct. A partition name may not have leading spaces, and in the command, the partition name must be enclosed in single or double quotation marks.

This command causes a page break.

Reading data from a memory card:

To read partition *DataS001* from the memory card:

```
!R! RWER R, A, 2, 'DataS001'; EXIT;
```


RWER S — [Read/Write External Resource] Store TrueType font

Format

RWER S, *medium*, *font-type*, *file-name*, *font-name*, *length* [, *value*, *style*, *weight*, *type-face*] ; *binary-data* WRED ;

Parameters

medium:

drive specification:
 0 = hard disk or RAM disk
 A = memory card

font-type:

font type:
 0 = TrueType font
 1 = Japanese TrueType font
 2 = Japanese TrueType Collection font

file-name:

file name: (string of up to 11 characters; the extension .INF is added internally.) This string must be enclosed in either single or double quotation marks.

font-name:

font name (ASCII text string up to 31 characters long; only up to 28 characters if *font-type* = 2). This string must be enclosed in either single or double quotation marks.

length:

font data length (number of bytes sent as *binary data*)

value:

font symbol set value (1 to 65535, 619: Shift JIS code)

style:

font style (0 to 32767)

weight:

stroke weight (-127 to +127)

type-face:

type face (0 to 32767)

binary-data:

font data (TrueType format)

Function

The `RWER S` command allows outline fonts (in TrueType format) to be downloaded to the printing system and to be stored on the specified media as font data. TrueType fonts particularly can be defined easily just by adding the TrueType font format as *binary-data*.

Always explicitly terminate the end of font data using `WRED;`. The five bytes represented by `WRED;` are not counted as part of the font data length.

Since *value*, *style*, *weight* and *type-face* represent font attributes they have an affect during font selection. Although they may be omitted, we recommend that all parameters be used to define font selection information.

In order to delete a Japanese TrueType font defined using this command, it is necessary to delete two separate file names using `RWER D, 'xxxxxxx'` and `RWER D, 'xxxxxxx.inf'`. Up to 512 fonts may be registered on a single memory card or up to 4096 on a single hard disk. Note carefully that registration beyond this limit will result in data being interpreted as unneeded.

Only 28 characters are valid for Japanese TrueType font names when *font-type* = 2 is specified because the system automatically appends (P) for proportional fonts and (F) for fixed pitch fonts.

A command error will result and the command has no effect if any of the parameters are outside the given range. This command is ignored if even one parameter error occurs.

The font data following the semicolon for this command must match the length given by *length*. An error will result and the font data will not be defined if the data length is shorter than specified.

RWER T — [Read/Write External Resource] set Terminate string

Format

```
RWER T[ , 'terminator' ] ;
```

Parameter

terminator:

ASCII text string up to 4 bytes, enclosed in either single or double quotation marks

Function

The RWER T command sets a terminator string for external media whether it is a memory card or the hard disk. This terminator is used when storing binary data to external media. See the RWER S command page (page 247) for example. WRED is set as the default terminator if parameter *terminator* is omitted.

A command error results and the command has no effect if parameters other than the above are used.

RWER W — [Read/Write External Resource] Write data to external media

Format

RWER W, *medium*, *data-type*[, '*partition-name*'];

Parameters

medium:

0 = hard disk or RAM disk
A = memory card

data-type:

1 = Font data
2 = Printable data
16 = Macro data
112 = File data
128 = API data (downloadable program)

partition-name:

Name of partition to which data is written, in a maximum of 15 printable ASCII characters. If this parameter is omitted the printing system will automatically determine the partition name.

Function

The RWER W command writes data to any external media whether it is a memory card or the hard disk.

Although all data types may be written up to the amount allowed by the storage size of the given media, note carefully that writing beyond this limit will result in data being interpreted as unneeded.

Even if data is written to a device using the same file name, that data will be written without overwriting the original file. A partition list will be printed as a warning when this happens.

A command error results and the command has no effect if parameters other than the above are used.

This command causes a page break and prints a write information page when executed.

The following example demonstrates how to send data from the computer to a memory card. Typically, partition names should be specified when writing to the card. If omitted, the partition names will be assigned automatically as follows:

DataS001, DataS002 ... DataSnnn.

Writing data to a memory card:

To write the contents of file *DATA123.DOC* to partition *DATA123* on the memory card:

1. Verify the printing system's message display indicates *Ready*.
2. Enter the following at the DOS prompt:

```
C:\>ECHO !R! RWER W, A, 2, 'DATA123'; >PRN
```

The message display will change from *Ready* to *Processing*, and then to *Waiting*. The printing system is now ready to receive data, so send the file from the computer to the printing system.

3. To do this, enter the following at the DOS prompt:

```
C:\>COPY DATA123.DOC PRN
```

4. The message display will change to *Processing* as the file is written to partition *DATA123* on the memory card. Next, check that the message display has changed to *Waiting*, and then send a *WRED* command, as follows:

```
C:\>ECHO WRED; EXIT, E; >PRN
```

The *WRED* command terminates writing to the memory card.

Note that in *W* mode the *ICCD* command is accompanied by a form feed.

If a partition name conflict occurs when writing a new partition to a memory card, the name of the new partition will be modified as follows:

- If a partition name of 11 characters or less conflicts with an existing partition name, '*Snnn*' (*nnn* represents a numeric value) is appended to the partition's name.
- If a partition name of 12 or more characters conflicts with an existing partition name, the 12th and following characters of the new partition's name will be replaced by '*Snnn*' (*nnn* represents a numeric value).

Related Command

WRED

RWRF D — [Read/Write Resource File] Delete data on external device

Format

RWRF D, *medium*, *data-type*, "*file-name*";

Parameters

medium:

0 = hard disk
A = memory card

data-type:

"API" API program
"FNT" Font data
"MCR" Macro data
"PRN" Printable data

file-name:

ASCII text string up to 15 bytes

Function

The RWRF D command deletes data that is written to an external device such as the hard disk.

This command has no effect if the media has been write protected using KPDL or PJJL.

This command causes a page break.

Any write information page is not printed when the file is deleted.

Example:

Delete macro file "TEST", which is registered in the hard disk.

```
!R! RWRF D,0, "MCR" , "TEST"; EXIT;
```

RWRF F — [Read/Write Resource File] Format external device

Format

RWRF F, *medium*;

Parameters

medium:

0 = hard disk
A = memory card

Function

The RWRF F command physically and logically formats an external device such as the hard disk.

This command has no effect if the target device has been write-protected using KPDL or PJL.

This command causes a page break.

Example:

Format the hard disk.

```
!R! RWRF F , 0; EXIT;
```

RWRF L — [Read/Write Resource File] print resource file List

Format

```
RWRF L [ , 'medium' ] ;
```

Parameters

medium:

0 = hard disk

A = memory card

Prints a resource file list for all devices if omitted.

Function

The RWRF L command prints a resource file list registered in an external device.

Example:

Print a resource file list in the hard disk.

```
!R! RWRF L , 0 ; EXIT ;
```


RWRF P — [Read/Write Resource File] set hidden file

Format

RWRF P, *medium*, *data-type*, "*file-name*";

Parameters

medium:

0 = hard disk
A = memory card

data-type:

"API" API program
"FNT" Font data
"MCR" Macro data

file-name:

ASCII text string up to 15 bytes.

Function

The RWRF P command sets the specified file as a hidden one. The specified file, which is hidden, cannot be accessed from any PC. This command causes a page break.

Example:

Set macro file "TEST" in the hard disk as a hidden file.

```
!R! RWRF P , 0 , "MCR" , "TEST"; EXIT;
```

Set macro file "TEST" in the memory card as a hidden file.

```
!R! RWRF P , A , "MCR" , "TEST"; EXIT;
```

RWRF R — [Read/Write Resource File] Read

Format

```
RWER R, medium, "PRN", 'partition-name' ;
```

Parameters

medium:

0 = hard disk
A = memory card

partition-name:

Name of partition from which data is read, a maximum of 15 printable ASCII characters

Function

The RWRF R command reads data specified by file name from any external media. The data must be a printable data. When the file name is coexistent in both the hard disk and memory card that are installed in the printing system, the printing system will read the memory card first, then the hard disk.

The partition name may be a maximum of 15 printable ASCII characters (20H through 7EH); upper and lowercase letters are regarded as distinct. A partition name may not have leading spaces, and in the command, the partition name must be enclosed in single or double quotation marks.

The printing system will automatically read or execute the data in the media if it is stored under the file name of '02autostart' and saved in the PRE folder of the media. The printing system will look for the file when it is powered up. The RWRF R command is not required.

This command causes a page break.

Example:

To read partition DataS001 from the hard disk:

```
!R! RWRF R, 0, "PRN", 'DataS001'; EXIT;
```

RWR F T — [Read/Write Resource File] set Terminator string

Format

```
RWR F T [ , 'terminator' ] ;
```

Parameters

terminator:

ASCII text string up to 4 bytes
Set to default "WRED" if omitted.

Function

The RWR F T command changes a terminator (default: WRED).
The terminator that was changed with this command is also available for the LAPI command to load the API program.
Turning the power on again will return the changed terminator to "WRED".

Example:

Change the terminator to "TEST".

```
!R! RWR F T , "TEST" ; EXIT ;
```

RWRF W — [Read/Write Resource File] Write data to external device (API program)

Format

```
RWRF W, medium, "API", "program-type", prescribe-command-name, API-program-start-function-name, "file-name";
```

Parameters

medium:

0 = hard disk
A = memory card

program-type:

"PRE" prescribe command

prescribe-command-name:

ASCII text string up to 4 bytes

API-program-start-function-name:

ASCII text string up to 31 bytes
Cannot be omitted.

file-name:

ASCII text string up to 15 bytes
Automatically set by the system if omitted.

Function

The RWRF W command writes data to an external device such as the hard disk. The number of writable files, which is not limited, depends on the capacity of the external media. If the device runs out of free space during saving, the remaining data will be discarded.

Duplicating a file name causes the system to rename the specified one and save data (without overwriting data).

Example:

Write and register the FILT prescribe command.

```
!R! RWRF W , 0 , "API" , "PRE" , "FILT" ,  
"apl_main";[Program Data]WRED; EXIT;
```

RWRF W — [Read/Write Resource File] Write data to external device (font)

Format

RWRF W, *medium*, "FNT", *font-type*, "*file-name*", "*font-name*", *font-data-length*, *symbol-set-value*, *font-style*, *stroke-weight*, *typeface*;

Parameters

medium:

0 = hard disk
A = memory card

font-type:

0 = TrueType Font
1 = Japanese TrueType Font
2 = Japanese TrueType Font Collection
RSF = RomSlotFormat

file-name:

ASCII text string up to 11 bytes
Up to 15 bytes for font type RSF

font-name:

ASCII text string up to 31 bytes
Up to 28 bytes for font type 2

font-data-length:

Number of bytes in font data to be downloaded

symbol-set-value:

1 to 65535 (619 = Shift JIS code)

font-style:

0 to 32767

stroke-weight:

-127 to +127

typeface:

0 to 32767

Function

The RWRW W command writes data to an external device such as the hard disk. The available number of fonts is max. 512 in the same compact flash and max. 4096 in the hard disk.

Duplicating a file name causes the system to rename the specified one and save data (without overwriting data).

For Japanese TrueType Font Collection, the print system automatically appends (P) for proportional fonts and (F) for fixed pitch fonts. If TrueType Font Collection contains three or more fonts, the print system automatically appends (1), (2), (3)...(9).

You can omit the parameters ranging from symbol-set-value to typeface; however, you should not omit those font attributes to exactly select a font.

The length of the font data to be written must match the specified one. Failure to do so will result in an error, and disables you to write data.

Specify WRED; command following <binary data>. "WRED;" (five bytes) are not included in the font data length.

To delete the registered Japanese TrueType font, use the RWRW D command.

To write font data in the external media, you need to have the owner grant the intellectual property right for fonts.

Example:

```
!R! RWRW W , 0 , "FNT" , "RSF" , "testfont"; [FontData]
WRED; EXIT;
!R! RWRW W , 0 , "FNT" , 0 , "testfile" , "testfont" , 619 ,
1 , 10 , 0 , 100; [FontData] WRED; EXIT;
```

RWRW — [Read/Write Resource File] Write data to external device (macro)

Format

RWRW W, *medium*, "MCR", *macro-type*, "*macro-ID or macro-name*", "*file-name*";

Parameters

medium:

0 = hard disk
A = memory card

macro-type:

"PRE" prescribe macro
"PCL" PCL macro

macro-ID or macro-name:

macro-ID for PCL macro (0 to 32767)
macro-name for prescribe macro (ASCII text string up to 4 bytes)
file-name

file-name:

ASCII text string up to 15 bytes
Automatically set by the system if omitted.

Function

The RWRW command writes data to an external device such as the hard disk. The number of writable files, which is not limited, depends on the capacity of the external media. If the device runs out of free space during saving, the remaining data will be discarded. Duplicating a file name causes the system to rename the specified one and save data (without overwriting data).

Example:

Write and register prescribe macro "TEST".

```
!R! RWRW W,0, "MCR", "PRE", "TEST", "test_macro"; %[Macro Data] WRED; EXIT;
```

Write and register the PCL macro (macro ID: 1111).

```
!R! RWRW W,0, "MCR", "PCL", 1111, "test_macro"; [Macro Data] WRED; EXIT;
```

RWRF W — [Read/Write Resource File] Write data to external device (printable data)

Format

```
RWRF W, medium, "PRN", "file name";
```

Parameters

medium:

0 = hard disk
A = memory card

file-name:

ASCII text string up to 15 bytes
Automatically set by the system if omitted.

Function

The RWRF W command writes data to an external device such as the hard disk. The number of writable files, which is not limited, depends on the capacity of the external media. If the device runs out of free space during saving, the remaining data will be discarded.

Duplicating a file name causes the system to rename the specified one and save data (without overwriting data).

As described in the example below, add the string "WRED" to the end of the file name.

Example:

Write and register printable data "TEST" to the hard disk.

```
!R! RWRF W, 0, "PRN", "TEST"; !R! TEXT "TEST"; PAGE; EXIT;  
WRED; EXIT;
```


SATU — adjust SATUration level

COLOR

Format

SATU *degree* ;

Parameter

degree:

degree of saturation level:


number from -10 to 10 (Default = 0)

Function

The SATU command adjusts saturation level based on the specified parameter.

The color bars below show saturation variations using hue “R” as an example. The top color bar represents the default saturation variation. The Δ symbol indicates the median.

The left side represents the lower saturation range and the right side the higher saturation range. The middle color bar is an example showing an increase in saturation. (The higher saturation range is increased.) The middle color bar is an example showing a decrease in saturation. (The lower saturation range is increased.)



Varying saturation allows images in the lower saturation range to be more saturated and sharp or, conversely, images in the higher saturation range to be faint and shaded off.

Saturation cannot be controlled hue by hue. However, since impressions vary with hue, you need to control with care.

This command is enabled for all color spaces, all color modes and all monitor simulation modes.

If ink simulation is set to other than None, this command is disabled.

This mode is not held in the printing system. For this reason, saturation is reset to its default state when the power is turned on.

SBM — Set Bottom Margin

Format

SBM *length*;

Parameter

length:

distance from top of paper to bottom margin

Function

The SBM command sets the bottom margin, as measured from the top edge limit of the paper in the units designated by the UNIT command.

If an attempt is made to print below the bottom margin, the characters will be printed on the next page.


A bottom margin setting in a document file should be placed at the beginning of the file or at a page break.

The SPL (set page length) and SLPP (set lines per page) commands also set the bottom margin, but these commands measure from the top margin position.

File

```
!R! RES; UNIT C;
STM 1; SLS 1; SBM 11;
SFNT 'Helvetica-Nr';
EXIT, E;
Top margin at 1 cm, bottom margin at 11 cm. (Line 1)
Line 2
Line 3
Line 4
Line 5
Line 6
Line 7
Line 8
Line 9
Line 10
Line 11
Line 12
!R! PAGE; EXIT;
```

Printout



SCAP — Set line CAP

Format

SCAP mode ;

Parameter

mode:

line cap mode:

1 = square caps


2 = butt caps

3 = round caps

Function

The SCAP command determines the shape of line ends (or a line cap) in a path where they are finally stroked. The default caps is *butt caps* (2).


Square caps have square ends which extend one-half of a line width beyond the endpoints of the line. *Butt* caps are similar to square caps but the caps are flush with the endpoints and do not extend beyond them. *Round* caps end the lines with semicircular caps with diameters identical to the line width.



File

```
!R! UNIT C; SPD .5;
NEWP; CMNT Round-capped on every end;
SCAP 3;
PMZP 6, 6;
SCP; SCP;
PDRP 1, -1; RPP;
PDRP 1, 1; RPP;
PDRP 5, 0;
STRK;
PAGE; EXIT;
```

Printout



SCCS — Save Current Code Set

Format

SCCS ;

Parameter

None

Function

The SCCS command instructs the printing system to remember the code set of the current font. This command may be used in conjunction with a SCF (save current font) command to prevent the font from losing the current code set after a command operation has been made, even if the command changes the code set (macro, etc.).

To return to the saved code set later, use the RPCS (return to previous code set) command. In the case of using the SCCS command together with a SCF command, it is recommended that the RPCS command be used with a RPF command.

The current code set can be saved up to eight times using a separate SCCS for each. RPCS commands retrieve the previously saved code set in reverse order.

File

```
!R! RES; SFNT 'TimesNewRoman'; SCF; SCCS;
SLPI 3;
TEXT 'The following line is printed using ISO 6: ASCII', L;
TEXT '@ABCDEFG', L;
TEXT 'Now changing to Ventura Math, using a different
font', L;
SFNT 'Univers-Md'; CSET 5M; TEXT '@ABCDEFG', L;
RPF; RPCS;
TEXT 'Now bringing the previous font and code set back by a
RPF and RPCS command.', L;
TEXT '@ABCDEFG';
PAGE;
EXIT;
```

Printout

The following line is printed using ISO 6: ASCII

@ABCDEFGF

Now changing to Ventura Math, using a different font

≅ ABXΔEΦΓ

Now bringing the previous font and code set back by a RPF and RPCS command.

@ABCDEFGF

Related Commands

RPCS, RPF, SCF

SCF — Save Current Font

Format

```
SCF ;
```

Parameter

None

Function

The SCF command saves the current font so that it can be retrieved later using the RPF (return to previous font) command. The font to be saved can be either a bitmap font or a scalable font.

Up to eight font numbers can be saved in the printing system's memory by SCF commands (using a separate SCF command for each).

The SCF command only instructs the printing system to remember the font. It does not move actual fonts in or out of memory.

File

```
!R! RES;  
SLM 1; STM 2;  
SFNT 'TimesNewRoman', 12;  
EXIT;  
Affixed to the door was a large sign that read:
```

```
!R! SCF; SFNT 'Univers-BdIt', 20;  
MRP 1, .423;  
TEXT 'KEEP OUT';  
RPF;  
EXIT;
```

Being one of those people who assume that such signs are put up to be ignored, she reached out

Printout

Affixed to the door was a large sign that read:

KEEP OUT

Being one of those people who assume that such signs are put up to be ignored, she reached out

Related Command

RPF

SCG — Save Current Graphics state

Format

SCG;

Parameter

None

Function

The SCG command saves the graphics state comprised of the following characteristics:

- Current path and cursor position
- Current pen diameter
- Current line join
- Current line cap
- Current miter limit
- Current dash pattern
- Current flatness
- Current current fill pattern (16 × 16 dots)
- Current current clipping path
- Current raster resolution
- Current image model

Once the current graphics state is saved, it can be retrieved later using the RPG (return to previous graphics state) command. Saving and retrieving the graphics state in this way simplifies recovery of paths which stroke a filled and outlined shape. Note that the FILL command clears the current path, leaving no strokable path.

The printing system holds as much graphics state information as the user memory area can accommodate; however, graphics state information larger than the memory area can cause a memory error.

File

```
!R! RES; UNIT C;  
NEWP;  
PMZP 5, 5;  
SFNT 'Helvetica-Bd', 72;  
CPTH 'Mary';  
SCG;  
STRK;  
RPG;  
GPAT .9;  
FILL 1;  
PAGE;  
EXIT;
```

Printout

Mary

Related Commands

FILL, RPG

SCOL — Select COlor

COLOR

Format

SCOL *number*;

Parameter

number:

number from 0 to 255 (Default is 0.)

Function

The SCOL command selects a color from the currently active palette.

If no color is assigned to the specified number, this command is ignored. If the specified number exceeds the valid range, this command is invalid.

Color specification using SCOL affects the following objects to be printed:

- Texts
- Vector graphics
- Monochrome patterns
- Monochrome raster images

And, it does not affect the following objects:

- Color patterns
- Color raster images

Once a color is specified by SCOL, the active color does not change until the next color specification is made even if a palette is created, changed or deleted.

File and Printout

See *ACLI* on page 12.

Related Commands

ACLI

SCP — Save Current Position

Format

SCP ;

Parameter

None

Function

The SCP command instructs the printing system to remember the current cursor position, so that it can be retrieved later using the RPP (return to previous position) command.

Up to eight positions can be saved in the printing system's memory (using a separate SCP command for each). RPP commands return to the positions in the reverse order in which they were saved.

File and Printout

See *RPP* on page 230.

Related Command

RPP

SCPI — Set Characters Per Inch

Format

SCPI *number*;

Parameter

number:

number of characters per inch

Function

The SCPI command sets the number of characters per inch (even if the UNIT command designates a non-inch unit).

When a proportionally-spaced font is used and the font mode (FTMD) is 0 to 3 or 8 to 11, SCPI 0 (zero) must be specified to obtain the innate spacing of each character. A non-zero specification results in unnatural spacing of proportional fonts.

SCPI 0 does not change the width of the space character.

The character spacing can also be controlled by the SCS (set character spacing) command.

File

```
!R! RES; SLM 1; STM 2; EXIT;
This line has 10 characters per inch.
!R! SCPI 8; EXIT;
This line has 8 characters per inch.
!R! SFNT 'Helvetica', 10; SCS .2; EXIT;
This proportional font
is printed with fixed
spacing.
!R! SCS 0; EXIT;
This is correct proportional spacing.
!R! PAGE; EXIT;
```

Printout

This line has 10 characters per inch.

This line has 8 characters per inch.

T h i s p r o p o r t i o n a l f o n t
i s p r i n t e d w i t h f i x e d
s p a c i n g .

This is correct proportional spacing.

Related Command

SCS

SCPL — Save Current PaLette

COLOR

Format

```
SCPL;
```

Parameter

None

Function

The SCPL command saves the currently active palette, so that it can be retrieved later using the RPPL (return to previous palette) command.

Palettes can be saved as many as there is enough memory. If there is insufficient memory, a memory overflow will result and the palette will not be saved.

This command does not change the active palette. RPPL commands return to the saved palettes in the reverse order in which they were saved.

File

```
!R! RES;
SEM 6;
UNIT C;
SFNT 'Incised901SWC-Black', 20;
SPAL 100; CID 0, 1; ACLI 1, 128, 64, 64; SCPL;
SPAL 101; CID 0, 1; ACLI 1, 128, 0, 128; SCPL;
MZP 1, 1; RPPL; SCOL 1;
TEXT 'This line is printed using palette 101.';
MZP 1, 2; RPPL; SCOL 1;
TEXT 'This line is printed using palette 100.';
PAGE;
EXIT;
```

Printout

This line is printed using palette 101.
This line is printed using palette 100.

SCRC — Set Command Recognition Character

Format

SCRC character;

Parameter

character:

new command recognition character, any character from ASCII code 33 (!) to ASCII code 126 (~) except the apostrophe ('), quotation mark (") or semicolon (;).

Function

The SCRC command changes the command recognition character between the exclamation points in the PRESCRIBE start sequence.

The change remains in effect until power is turned off or a RES (reset) command is executed.

Originally the command recognition character is the capital letter R within the PRESCRIBE start sequence of !R!.

One reason for changing the command recognition character is to print listings of files or programs containing PRESCRIBE commands. If the command recognition character is not changed, then instead of printing the PRESCRIBE commands, the printing system will execute them.

File


```
!R! RES; UNIT C;
GPAT .7; BLK 1, 1;      BOX 1, 1, E;
GPAT .5; BLK 1.3, 1.3; BOX 1.3, 1.3, E;
GPAT .3; BLK 1.6, 1.6; BOX 1.6, 1.6, E;

CMNT Now list the above commands;

SCRC L;
EXIT;

!R! RES; UNIT C;
GPAT .7; BLK 1, 1;      BOX 1, 1, E;
GPAT .5; BLK 1.3, 1.3; BOX 1.3, 1.3, E;
GPAT .3; BLK 1.6, 1.6; BOX 1.6, 1.6, E;
PAGE; EXIT;
!L! RES; EXIT;
```

Printout



```
!R! RES; UNIT C;  
GPAT .7; BLK 1, 1; BOX 1, 1, E;  
GPAT .5; BLK 1.3, 1.3; BOX 1.3, 1.3, E;  
GPAT .3; BLK 1.6, 1.6; BOX 1.6, 1.6, E;  
PAGE; EXIT;
```

SCS — Set Character Spacing

Format

SCS distance ;

Parameter

distance:

character-to-character spacing, or 0

Function

The SCS command sets the distance from the beginning of one character to the beginning of the next character.

When a proportionally-spaced font is used and the font mode (FTMD) is 0 to 3 or 8 to 11, SCS 0 (zero) must be specified to obtain the innate spacing of each character. A non-zero specification results in unnatural spacing of proportional fonts.

SCS 0 does not change the width of the space character.

The character spacing can be set using either the SCS or the SCPI command.

File

```
!R! RES; SLM 1; STM 2; EXIT;
This line has the default spacing.
!R! UNIT C; SCS .3; EXIT;
This line has a slightly wider spacing.
!R! SFNT 'TimesNewRoman', 10; SCS .2; EXIT;
This is a proportional font printed with fixed spacing.
!R! SCS 0; EXIT;
This is correct proportional spacing.
!R! PAGE; EXIT;
```

Printout

This line has the default spacing.

This line has a slightly wider spacing.

This is a proportional font printed with fixed spacing.

This is correct proportional spacing.

SCSZ — Set Custom paper Size

Format

SCSZ *width, height, unit;*

Parameters

width:

width of custom paper size

height:

height of custom paper size

unit:

M = millimeters

I = inches

Function

The SCSZ command sets the paper size when the user-defined size paper is used in the universal cassette. This command defines the width and height values for the custom paper size which is summoned by the SPSZ (set page size) 19; command.

The SCSZ command causes a page break, so it should be placed at the top of a file or program, or at the beginning of a new page. This command uses the unit set in the *unit* parameter of this command, regardless of the unit set in the UNIT command. The setting made with this command provides a non-printing area of 5 mm × 5 mm, independent of emulation mode.

File

```
!R! SCSZ 260, 300, M; EXIT;
```

SCU — Save Current Unit

Format

```
SCU;
```

Parameter

None

Function

The SCU command instructs the printing system to save the current unit of measurement. Saving the current unit can prove useful when a macro containing a UNIT command changes the unit of measurement. The saved unit can be retrieved later using the RPU (return to previous unit) command.

The current unit can be saved up to eight times using a separate SCU command for each. RPU commands retrieve the unit of measurement in reverse order.

File

```
!R! FRPO N8, 0; RES;
SENT "Univers-MdCd", 9;

UNIT C; SCU;
MZP 1.5, 1.5;
TEXT "THE DAYS ON THE ISLAND: PART I", U;


MCRO TITLE;
FSET 24V;
UNIT I;
MZP 4, 3;
CIR 2;
MRP -1.5, .125;
TEXT %1;
ENDM;

CALL TITLE, "CHAPTER 1: PROLOGUE";

RPU;
SLM 2;
STM 13;
FSET 12v;
EXIT;
My long anticipated vacation started at Mallorca
island which
```

Printout

THE DAYS ON THE ISLAND: PART I



CHAPTER 1: PROLOGUE

My long anticipated vacation started at Mallorca island which

Related commands

SCF, SCP, UNIT

SDP — Store Dash Pattern

Format

SDP *pattern-number*, *dash1*, *space1*, *dash2*, *space2*, . . . , *dash10*, *space10*;

Parameters

pattern-number:

number from 11 to 20

dash1 ... *10*:

length of dash


space1 ... *10*:

space between dashes

Function

The SDP command generates a new dash pattern used by the DPAT (dashed pattern) command for stroking a path. The *dash* and *space* values represent the lengths of alternating black and white segments for stroking the line segment in the path, using the unit value designated by the UNIT command.

For example, SDP 11, 2, 2, 5, 5; stores the following dash pattern in pattern number 11:




The maximum number of pairs of dash and space allowed for a SDP command is 10. If the dash pattern begins with a space, 0 must be specified for the first *dash* value (*dash1*). To end the pattern with a dash, the last *space* value may be omitted.

The user-defined dash patterns are valid until another SDP command changes the current definition or until the printing system is switched off.

File

```
!R! RES; UNIT P;  
SDP 11, 2, 2, 5, 5;  
NEWP;  
UNIT C;  
SPD .2; PMRA 10, 10, 3, 0;  
DPAT 11; PARC 10, 10, 3, 0, 360;  
STRK;  
  
PAGE;  
EXIT;
```

Printout



Related Command

DPAT

SEM — Set Emulation Mode

Format

SEM *mode* ;

Parameter

mode:

- 0 = Line Printer
- 1 = IBM Proprinter X-24E
- 2 = Diablo 630
- 5 = Epson LQ-850
- 6 = HP LaserJet/PCL
- 8 = KC-GL (Plotter)
- 9 = KPDL (an option on some models)

Function

The SEM command temporarily changes the printing system's emulation mode. The new mode remains in effect until another SEM or an RES (reset) command is issued.

The SEM command also resets the margins and various other parameters. It should therefore be placed at the beginning of a program or file.

The SEM command does not change the power-up emulation mode. When power is switched off, the printing system loses the mode set by the SEM command. Upon power up, it retains the default emulation, which is set by the P1 parameter of the FRPO command.

If you usually work in 'emulation A' but occasionally use 'emulation B,' it is convenient to set 'emulation A' with the FRPO command and use the SEM command to change to 'emulation B' when necessary. In this case you must avoid using the RES command in 'emulation B,' because the RES command resets the printing system to 'emulation A.'

File

```
!R! CMNT To work temporarily in the LQ-850 emulation;  
SEM 5;  
EXIT;
```

SETF — SET alternate Font

Format

```
SETF alternate-number, font-number;
```

Parameters

alternate-number:

a number from 0 to 31

font-number:

a number from 0 to 65535

Function

The SETF command assigns a font to an alternate number, so that it can be selected by the ALTF (alternate font) command. The font number can be one of the resident font numbers or the number of a nonresident font defined by LDFC (load font character) commands.

A scalable font can also be assigned to an alternate number when given a temporary font number in advance. See the SFNT (set font) command page.

In some emulation modes, default assignments are given to low alternate numbers so that SETF commands can be embedded for some word processing software. See *Emulation* of the *Technical Reference* manual.

File

```
!R! RES;
SFNT 'Helvetica', 10, 2000;
SFNT 'Helvetica-BdOb', 10, 2001;
SETF 1, 2000;
SETF 2, 2001;
MCRO BIBLIO;
ALTF 1; TEXT %1, E; TEXT " ", E;
ALTF 2; TEXT %2, E; TEXT " ", E;
ALTF 1; TEXT %3, N;
ENDM;
ALTF 1; TEXT 'BIBLIOGRAPHY',N,U;
TEXT,N;
CALL BIBLIO, "Engelberger, Joseph L., ",
"Robotics in Practice,", "Kogan Page";
CALL BIBLIO, "Cohen, J.,",
"Human Robots in Myth and Science,", "Allen and Irwin";
CALL BIBLIO, "McCorduck, Pamela,",
"Machines Who Think,", "W. H. Freeman and Company";
CALL BIBLIO, "Asimov, Isaac,",
"I, Robot,", "Doubleday & Co.";
EXIT;
```

Printout

BIBLIOGRAPHY

Engelberger, Joseph L., ***Robotics in Practice***, Kogan Page
Cohen, J., ***Human Robots in Myth and Science***, Allen and Irwin
McCorduck, Pamela, ***Machines Who Think***, W. H. Freeman and Company
Asimov, Isaac, ***I, Robot***, Doubleday & Co.

SFA — Set bitmap Font Attributes

Format

SFA hmi, vmi, F/P[, rotation, symbol-set, height, U/I, weight, type-face][, font-type];

Executable Parameters

hmi:

horizontal motion index; in a fixed font, the distance from the center of one character to the center of the next; in a proportionally spaced font, the width of the space character

vmi:

vertical motion index—the line-to-line spacing

F/P:

F = fixed spacing
P = proportional spacing

Non-executable Parameters

rotation:

0 = normal
1 = face-down
2 = upside down
3 = face-up (used in landscape mode)

symbol-set:

symbol set (number from 0 to 65535):
4 = ISO 60 Norway
5 = Roman Extension
9 = ISO 15 Italian
14 = ECMA-94 Latin 1
19 = ISO 11 Swedish
21 = ISO 6 ASCII
37 = ISO 4 U. K.
38 = ISO 69 France
39 = ISO 21 Germany
53 = US Legal
75 = ISO
83 = ISO 17 Spain
277 = HP Roman-8
305 = PC Extension
341 = IBM PC-8
373 = IBM PC-8 (D/N)
405 = IBM PC-850

height:

maximum height of the font

U/I:

U = upright

I = italic

weight:

degree of bolding, from -7 (light) to +7 (bold)

type-face:

0 = Line Printer

1 = Pica 2 = Elite

3 = Courier

4 = Swiss 721

5 = Dutch 801

6 = Gothic

7 = Script

8 = Pret. Elite

9 = Caslon

10 = Orator

font-type:

0 = 7-bit, 96 characters (character codes 32 to 127)

1 = 8-bit, 192 characters (character codes 32 to 127 and 160 to 255)

2 = 8-bit, 256 characters (All character codes. The printing system must be set to the HP transparency mode. See *Emulation of the Technical Reference manual.*)

Function

The SFA command specifies the attributes of the current bitmap font. It can be used to change the default attributes of the bitmap font and will be ignored when the current font is a scalable font. The font retains its new attributes until they are changed by another SFA command or power is switched off. Font attributes are not reset by the RES (reset) command.

The SFA command has two types of parameters. The executable parameters adjust the intrinsic character and line spacing of the font. The non-executable parameters do not change the appearance of the font at all, but they affect the way in which the font is selected by word-processing software in the HP LaserJet emulation.

Executable Parameters

The first three parameters are required in every SFA command. The *hmi* parameter sets the character spacing of a fixed font or the word spacing of a proportional font. The *vmi* parameter adjusts the line spacing. The *vmi* and *hmi* parameters both use the unit value designated by the UNIT command.

The third parameter designates the font as fixed or proportional. Normally you should specify F for a fixed font (example: font 1) and P for a proportional font (example: font 2).

If the font mode (FTMD) value is 15, the assigned attributes take effect immediately. If the font mode value is less than 15, some of the attributes, although assigned to the font,

will not take effect. If the font mode is 0 (the default value in the daisywheel emulation mode), none of the attributes will take effect.

Non-executable Parameters

The fourth and subsequent parameters of the SFA command are optional, although if one of these parameters is desired, then the preceding parameters must also be specified. The optional parameters are used to describe downloaded fonts for selection by embedded commands in the HP LaserJet emulation mode. The parameters can also redefine the attributes of resident fonts. However, they do not actually change the physical appearance of the font, only how the printing system interprets the font's attributes. The rotation parameter should be 0 for a portrait font or 3 for a landscape font. The values 1 and 2 are not normally used. You cannot change the orientation of a font with this parameter.

The symbol set parameter identifies the font's character set. Specifying this parameter does not actually change the character set. The height parameter is measured in the unit value designated by the UNIT command.

File

```
!R! RES; FTMD 15;
UNIT C; SLM 2; STM 4;
FONT 1; SFA .25, 1, F;
FONT 2; SFA .15, .5, P;
FONT 1;
EXIT;
```

At these settings, in font mode 15, font 1 is double spaced (1 cm),
!R! FONT 2; EXIT,E;
while font 2 is single spaced (0.5 cm) with 1.5 mm spaces between words.

```
!R! FONT 1; EXIT,E;
The printer remembers
the spacing assigned
to each font.
```

Printout

At these settings, in
font mode 15, font 1 is
double spaced (1 cm),
while font 2 is single
spaced (0.5 cm) with 1.5
mm spaces between words.

The printer remembers
the spacing assigned
to each font.

SFNT — Select current FoNT by typeface

Format

```
SFNT 'type-face' [ , height [ , font-number [ , symbol-set , compression , shear ] ] ] ;
```

Parameters

type-face:

Name of the typeface of the font to be selected.

height:

Height of the characters in points. The minimum value is 0.25 points and the maximum value is 999.75 points, in 0.25-point increments.

font-number:

Integer from 1000 to 65535

symbol-set:

Symbol set, number from 0 to 65535; Refer to LaserJet Symbol Sets in *Emulation of the Technical Reference* manual.

compression:

1 for normal character width. Smaller values for narrow characters; and larger values for wider characters. The minimum value is 0.3 and the maximum value is 3.

shear:

0 for normal (unitalicized) characters. Positive values give forward italicization; and negative values give backward italicization. The minimum value is -1 and the maximum value is 1.

Function

The SFNT command selects the font which matches the *type-face* parameters. It enables the user to size its height, assign a font number and vary some of the font attributes so that characters can appear compressed or sheared. All parameters except *type-face* are ignored if a bitmap typeface is specified.

By designating a font number for a scalable font, it can later be retrieved by using a FONT command. If you specify the symbol set, compression, and shear parameters together, the FONT command recalls the font with these effects intact.

type-face should be a name from the font table in Section 4.2. of the *Technical Reference* manual. Specify the full name exactly as it appears in the table, including case sensitivity. This command is ignored and the current font is not changed if the typeface name is not in the table or is not typed correctly.

height defines the size of the font, using the unit of points regardless of the current unit setting. The SFNT command is ignored if the *height* value is out of the range. If the

height parameter is omitted, the printing system sizes the font to the height of the current font. The following example displays a range of font heights.

54 RRRRRRRRRR
 39 RRRRRRRRRRRR
 30 RRRRRRRRRRRRRR
 24 RRRRRRRRRRRRRRRR
 20 RRRRRRRRRRRRRRRRRR
 16 RRRRRRRRRRRRRRRRRRRR
 14 RRRRRRRRRRRRRRRRRRRR
 13 RRRRRRRRRRRRRRRRRRRR
 12 RRRRRRRRRRRRRRRRRRRR
 11 RRRRRRRRRRRRRRRRRRRR
 10 RRRRRRRRRRRRRRRRRRRR
 9.5 RRRRRRRRRRRRRRRRRRRR
 9 RRRRRRRRRRRRRRRRRRRR


You can assign any number from 1000 to 65535 as a temporary *font-number*, as long as you do not duplicate the number of a bitmap font or a font number already in use. If the number matches one already assigned to another font, the printing system ignores the SFNT command.

symbol-set uses HP PCL matching code to change the font's symbol set. Unsupported symbol set values, if specified, are ignored.

compression squeezes or expands characters in the horizontal direction.



shear imparts a slant to the characters: forward if the angle value is positive; backward if it is negative. Zero gives upright characters.



Changes made with SFNT to a scalable font in its height, symbol set, compression, and shear remain effective until another font selection command is encountered. To retrieve this scalable font later, use the FONT command.

File

```
!R! RES; SPO P;
SFNT 'Helvetica-Nr', 10, 3000, 277, 2, .3;
EXIT;
```

This is 10-point Helvetica-Nr,
with Roman-8 symbol set, compression is 2, and
shear is .3. You can later recall this font by the
FONT 3000; command.
!R! PAGE; EXIT;

Printout

***This is 10-point Helvetica-Nr,
with Roman-8 symbol set, compression is 2, and
shear is .3. You can later recall this font by the
FONT 3000; command.***

SGPC — Set kcGL Pen Color

COLOR

Format

SGPC pen-number, color;

Parameters

pen-number:

number from 1 to 8 corresponding to FRPO G1 to G8 parameter

color:

0 = Black
1 = Red
2 = Green
3 = Yellow
4 = Blue
5 = Magenta
6 = Cyan
7 = White

Function

The SGPC command sets the color of a pen 1 through 8 for use with KCGL to a specified color. The specified value is permanently saved in the printing system.

SHMI — Set HMI

Format

```
SMHI hmi ;
```

Parameter

hmi:

horizontal motion index; in a fixed font, the distance from the center of one character to the center of the next; in a proportionally spaced font, the width of the space character

Function

The SHMI command sets the HMI. The parameter is measured in the unit value specified by the UNIT command.

File

```
!R! RES; SLM 1; STM 2; EXIT;  
This line has the default spacing.  
!R! UNIT C; SHMI .3; EXIT;  
This line has a slightly wider spacing.  
!R! SFNT 'TimesNewRoman', 10; SHMI .3; EXIT;  
Proportional fonts are not influenced.  
!R! PAGE; EXIT;
```

Printout

```
This line has the default spacing.  
  
This line has a slightly wider spacing.  
  
This is a proportional font printed with wider spacing.  
  
This is a proportional font printed with default spacing.
```

SIMG — Set IMaGe model

Format

`SIMG operation-mode ;`

Parameter

operation-mode:

	<i>Source image</i>	<i>Pattern image</i>
1 =	Transparent	–
2 =	Opaque	–
3 =	Transparent	Transparent
4 =	Transparent	Opaque
5 =	Opaque	Transparent
6 =	Opaque	Opaque


Function

The SIMG command sets the mode for controlling the transparency and opaqueness of an image and the current pattern when they are overlaid on each other.


In the table above, the *source image* can be a rectangular area, characters, or a raster graphics image preceded by an SIMG command. The fill patterns and gray shade rendered with the PAT (pattern), FPAT (fill pattern), and GPAT (gray pattern) commands can be used for *pattern images*.

With operation modes 1 and 2, the SIMG command takes care of the source image's transparency only. The default operation mode is 1. The white pixels of the source image do not overlay on the destination. For example, you cannot pattern a character. With operation mode 2, the SIMG command applies the white pixels of the source image onto the destination directly.

SIMG=1



SIMG=2



The figure on the next page illustrates the effects of operation mode 3 through 6. In this example, the letter A represents the source image overlaid with the pattern image, while the two rectangles are referred to as the destination image.

operation-mode = 3: Both the source image and pattern image are transparent. Since the source image is transparent, in the first example, the letter A is patterned and overlaid onto the destination image (two rectangles).


operation-mode = 4: The source image is *transparent* and the pattern image is now *opaque*. The pattern's white pixels overlay the destination image in the shape of the letter A.

operation-mode = 5: The source image is *opaque* and the pattern image is *transparent*. Being *opaque*, the source image (the letter A) as it is enclosed within a square comes overlaid onto the destination. The pattern image fills only the area on the destination which is covered by white pixels. Note that the letter A is patterned only with its opposing quarters.


operation mode = 6: Both source and pattern images are *opaque*, the entire letter A is patterned, and the full source image is overlaid onto the destination.

Operation Modes


Operation mode=3




Operation mode=4



Operation mode=5



Operation mode=6



File

```
!R! RES; UNIT C;  
NEWP;  
PMZP 5, 5;  
PDRP 2.5, 0, 0, 2.5, -5, 0, 0, 2.5, 2.5, 0;  
PAT 1; SIMG 4;  
FILL 1;  
MRP -1, 3.5; PAT 18;  
FSET 4158T3B0S1P108V; TEXT "A";  
FILL 1;  
PAGE; EXIT;
```

Printout

SIMP — create SIMPLE color palette

COLOR

Format

SIMP *palette* ;

Parameter

palette:

- 1 = single plane K (Black) palette (Default)
- 3 = 3-plane device RGB palette

Function

The SIMP command creates a palette already defined by the system from either a single plane (monochrome) or 3-plane (8-color) RGB palette. The default palette is a monochrome palette.

Colors and parameters specifiable by SCOL command for each palette are as follows:

palette	SCOL command parameter	Color
1 (single plane)	0	White
	1	Black
3 (3-plane)	0	Black
	1	Red
	2	Green
	3	Yellow
	4	Blue
	5	Magenta
	6	Cyan
	7	White

Color assignments (using the ACLI command) cannot be made to the pallette created using the SIMP command.

The currently acative palette ID does not change.

File

```
!R! RES; UNIT C; SIMP 3;  
MCRO SIMP;  
NEWP;  
SCOL %2;  
PDRP 2, 0, 0, 2, -2, 0, 0, -2;  
FILL 1;  
PMRP .5, .5;  
ENDM;  
CALL SIMP,0,0;  
CALL SIMP,0,1;  
CALL SIMP,0,2;  
CALL SIMP,0,3;  
CALL SIMP,0,4;  
CALL SIMP,0,5;  
CALL SIMP,0,6;  
CALL SIMP,0,7;  
PAGE;  
EXIT;
```

Printout

SIR — Set Image Refinement level

Format

SIR *level*;

Parameter

level:

0 = Off
2 = On (Medium)

Function

The SIR command accesses different levels of Kyocera Image Refinement (*KIR*). The image refinement level remains in effect until the printing system receives a RES command or a different SIR command. The default value is On.

Refer to the *User's Manual* for details about how to obtain the optimum level of the image refinement.

File

```
!R! SIR 0; CMNT KIR is off; EXIT;
```

SLJN — Set Line Join

Format

SLJN *mode* ;

Parameter

mode:

line join mode:
1 = Beveled join
2 = Mitered join
3 = Round join
4 = Notched join

Function

The SLJN command determines how two connected lines in a path are joined. The default join is *beveled* (1).

Beveled join ends the lines with butt caps (Refer to the SCAP command) and fills the larger angle between the lines with a triangle. *Mitered* join extends the edges of the lines until they touch. The current *miter limit* value affects this join type (Refer to SMLT command on page 313). *Round* join connects lines with a circular join noted by a diameter equal to the width of the lines. *Notched* join enables the lines to overlap.

Beveled join



Mitered join



Round join



Notched join



File

```
!R! RES; UNIT C;
NEWP;

SPD .5;
SCAP 3;
CMNT Round-capped at the upper-left corner;
SLJN 3;
CMNT Round-joined at other corners;
PMZP 5, 16.5;
PDRP 3, 0, 0, 3, -3, 0, 0, -3;
STRK;

NEWP;
SCAP 1;
CMNT Square-capped at the upper-left corner;
SLJN 2;
CMNT Mitered-joined at other corners;
PMZP 11, 16.5;
PDRP 3, 0, 0, 3, -3, 0, 0, -3;
STRK;

PAGE;
EXIT;
```

Printout



SLM — Set Left Margin

Format

SLM distance ;

Parameter

distance:

size of left margin

Function

The SLM command sets the left margin, as measured from the left edge limit of the paper in the unit value designated by the UNIT command.

If the margin is moved to the right of the current cursor position, the cursor moves to the new margin.

When the left margin is moved to the right, the right margin does not change. But when the left margin is moved toward the left, the right margin moves with it, keeping a constant page width. To move the left margin to the left but leave the right margin in its original position, you must add an SRM command after the SLM command to restore the original right margin.

File


```
!R! RES; UNIT C; FONT 2;  
SFNT "TimesNewRoman";  
SLM 1;  
EXIT;  
Location of 1-cm left margin
```

```
!R! SLM 2; EXIT;  
Location of 2-cm left margin
```

```
!R! SLM 3; EXIT;  
Location of 3-cm left margin
```

```
!R! PAGE; EXIT;
```

Printout



SLPI — Set Lines Per Inch

Format

SLPI *number*;

Parameter

number:

Number of lines per inch

Function

The SLPI command sets the number of lines per inch (even if the UNIT command designates a non-inch unit).

File

```
!R! RES; STM 2;  
SFNT "Helvetica";  
EXIT;  
These lines have the  
default spacing of  
six lines per inch.  
!R! SLPI 2; EXIT;  
These lines are  
spaced at two lines  
per inch.  
!R! PAGE; EXIT;
```

Printout

These lines have the
default spacing of
six lines per inch.

These lines are

spaced at two lines

per inch.

Related Command

SLS

SLPP — Set Lines Per Page

Format

SLPP number ;

Parameter

number:

number of lines per page

Function

The SLPP command sets the bottom margin for an exact number of lines per page based on the current line spacing. As long as you do not change the line spacing or use PRESCRIBE commands that move the cursor, you will get the specified number of lines per page.


To control the page size in terms of inches or centimeters instead of a number of lines, use the SBM (Set Bottom Margin) or SPL (Set Page Length) command. The advantage of the SLPP command is that it sets the bottom margin to accommodate an exact number of lines, so no excess is carried over to the next page.

A bottom margin setting in a document file should be placed at the beginning of the file or at a page break.

File

```
!R! RES; UNIT C;
SFNT "TimesNewRoman";
STM 1; SLS 1; SLPP 10;
EXIT, E;
Top margin at 1 cm, with 10 lines per page. (Line 1)
Line 2
Line 3
Line 4
Line 5
Line 6
Line 7
Line 8
Line 9
Line 10
Line 11
Line 12
!R! PAGE; EXIT;
```


Printout



SLS — Set Line Spacing

Format

SLS distance ;

Parameter

distance:

Line-to-line spacing

Function

The SLS command sets the distance from the base of one line to the base of the next line. The distance is measured in the unit value designated by the UNIT command.

The line spacing can be set by either the SLS or the SLPI (set lines per inch) command. SLPI sets the number of lines per inch. SLS sets the number of inches (or other units) per line.

File

```
!R! RES; UNIT C; STM 5;
SFNT "TimesNewRoman";
EXIT;
These lines have the
default line spacing
of 0.423 cm (6 lpi).
!R! SLS .6; EXIT;
These lines are
spaced at intervals
of .6 centimeters.
!R! PAGE; EXIT;
```

Printout

These lines have the
default line spacing
of 0.423 cm (6 lpi).

These lines are
spaced at intervals
of .6 centimeters.

Related Command

SLPI

SMLT — Set Miter Limit

Format


SMLT *limit-value* ;

Parameter

limit-value:

Ratio of the length through a join to the width of line, expressed by the following formula (Default is 10. Minimum value is 1).

$$\text{limit-value} = \frac{\text{miter length}}{\text{line width}} = \frac{1}{\sin(a/2)}$$



Function

The *miter limit* value set by the SMLT command is the maximum ratio of the miter length through a join to the width of the lines connected to each other.

The default miter limit value is 10. In smaller miter-limit values, say 3, the miter join automatically switches to the beveled join. This switching prevents the problem of a spike extending beyond the intersection of the path because a small angle is used for the miter-join.




File

```
!R! RES; UNIT C;  
NEWP;  
PMZP 5, 15;  
SPD .5;  
SLJN 2; CMNT Mitered join;  
PDRP 1.5, -2, 1.5, 2, 1.5, -2,  
      1.5, 2, 1.5, -2, 1.5, 2;  
SMLT 10; CMNT Miter limit is 10;  
STRK;
```

```
NEWP;  
PMZP 5, 19;  
SPD .5;  
SLJN 2; CMNT Mitered join;  
PDRP 1.5, -2, 1.5, 2, 1.5, -2,  
      1.5, 2, 1.5, -2, 1.5, 2;  
SMLT 1; CMNT Miter limit is 1;  
STRK;
```

```
PAGE;  
EXIT;
```

Printout



Related Command

SLJN

SMNT — Set MoNiTor simulation

COLOR

Format

SMNT *monitor-type* [,*temperature*, *red-x*, *red-y*, *green-x*, *green-y*, *blue-x*, *blue-y*, *gamma*] ;

Parameters

monitor-type:

0 = None
 1 = SMPTE 240M
 2 = HDTV (sRGB) [default]
 3 = Sony Trinitron
 4 = Apple RGB
 5 = NTSC
 6 = KC-RGB
 7 = Custom

temperature:

color temperature
 3500, 5000, 5500, 6500 (default), 6740, 7500, 9300

red-x/red-y/...

chromaticity coordinates:
 red-x: number from 0.0 to 3.0 (default = 0.64)
 red-y: number from 0.0 to 3.0 (default = 0.33)
 green-x: number from 0.0 to 3.0 (default = 0.30)
 green-y: number from 0.0 to 3.0 (default = 0.60)
 blue-x: number from 0.0 to 3.0 (default = 0.15)
 blue-y: number from 0.0 to 3.0 (default = 0.06)

gamma

value of gamma correction:
 number from 1.0 to 3.0 (default = 2.2)

Function

The SMNT command simulates a particular monitor or a monitor customized by specifying the color temperature, etc. Parameters except monitor-name are enabled only when monitor-name value is 7 (custom).

Resetting the printing system or switching emulation modes resets all custom parameters to the default states.

The simulation provided by this command is enabled only when the device-dependent color space is in use.

This command is irrelevant to some color printing system models. At the time of release of this manual, the command is irrelevant to model FS-5800C; and monitor-name values of 0 and 2 are valid for model FS-8000C.

Resetting the printing system and switching emulation mode reset all parameters to the default states.

SPAL — Select PALETTE

COLOR

Format

SPAL *id*;

Parameter

id:

palette ID number: 0 to 32767 (default is 0)

Function

The SPAL command selects the palette to activate from among the palettes that are currently available. If no palette is associated with the specified number, this command is ignored.

The palette activated by SPAL is deleted when a new palette is created by the CID command. The active palette number does not change. All palettes are deleted by the reset command, leaving only the default palette.

File and Printout

See *ACLI* on page 12.

SPD — Set Pen Diameter

Format

SPD *diameter*;

Parameter

diameter:


line thickness

Function

The SPD command sets the thickness of the line used in the graphics commands (DRP, PDRP, etc.). The thickness is specified in the unit value designated by the UNIT command. The maximum thickness is 4200 dots (14 inches or 35.56 centimeters) in path graphics mode, and 127 dots (0.42 inches or 1.08 centimeters) in standard graphics mode. The minimum thickness is 1 dot (about 0.01 centimeters or 0.0033 inches). The printing system rounds off all decimal values to the nearest integer value.

The SPD command affects graphics segments created by DZP, DAP, DRP, DRPA, BOX, CIR, and PIE commands. It also affects the width of the line and curve segments stroked as paths.


A slant line drawn by the standard graphics commands becomes thicker than specified by the SPD command. The effect may be most noticeable with a 45° slant. In the path mode graphics, however, the thickness of any slant lines is automatically adjusted so that the slant lines are maintained in constant thickness as specified by SPD. (See below.)



File

```
!R! RES; UNIT C; STM 2; SLM 2; SLS .6;  
SFNT "Helvetica-Nr";  
SPD 0.01; DRP 5, 0; TEXT ' 0.01 cm', N;  
SPD 0.02; DRP 5, 0; TEXT ' 0.02 cm', N;  
SPD 0.05; DRP 5, 0; TEXT ' 0.05 cm', N;  
SPD 0.1; DRP 5, 0; TEXT ' 0.1 cm', N;  
SPD 0.2; DRP 5, 0; TEXT ' 0.2 cm', N;  
PAGE; EXIT;
```

Printout



SPL — Set Page Length

Format

SPL length ;

Parameter

length:

distance from top margin to bottom margin

Function

The SPL command sets the bottom margin, as measured from the top margin position in the unit value designated by the UNIT command.

If an attempt is made to print below the bottom margin, the characters are printed on the next page.

A page length setting in a document file should be placed at the beginning of the file or at a page break.

The SBM (Set Bottom Margin) and SLPP (Set Lines Per Page) commands also set the bottom margin.

File

```
!R! RES; UNIT C;
SFNT "TimesNewRoman";
STM 1; SLS 1; SPL 10;
EXIT, E;
Top margin at 1 cm, with 10 cm page length. (Line 1)
Line 2
Line 3
Line 4
Line 5
Line 6
Line 7
Line 8
Line 9
Line 10
Line 11
Line 12
!R! PAGE; EXIT;
```

Printout

Top margin at 1 cm, with 10 cm page length. (Line 1)

Line 2

Line 3

Line 4

Line 5

Line 6

Line 7

Line 8

Line 9

Line 10

Line 11

Line 12

SPO — Set Page Orientation

Format

SPO orientation :

Parameter


orientation:

- P = portrait (upright)
- L = landscape (lengthwise)

Function

The SPO command instructs the printing system how to orient the page, laying text and graphics on the page in a vertical direction (portrait orientation) or a horizontal direction (landscape orientation).

When the orientation is changed, the margins are adjusted so that they remain the same distance from the four edges of the paper (provided cassette paper is being used). See the illustration.



If the printing system cannot adjust the margins as described above, it sets them to the edge limits. For example, if you set a top portrait margin greater than the paper width, then switch to landscape orientation, the printing system resets the top margin to zero.



The printing system uses separate bitmap fonts for portrait and landscape orientations. When the font mode (FTMD) is 8 or higher, as it is in the IBM, Epson, and LaserJet III/4/4 Plus emulations, the printing system automatically sets the correct page orientation each time a new font is selected. Outside of the FTMD range, the printing system will not change orientation with a change in fonts. After using the SPO command, remember to select a font with the new orientation.

The orientation can be changed without starting a new page, as in the example below. It should be noted, however, the printable areas in portrait and landscape orientations are not symmetrical because the edge limits in each orientation are not identical. See the *Technical Reference* manual for details.

File

```
!R! RES;  
SFNT 'TimesNewRoman-It', 10;  
CMNT This file is for an A4 cassette.;  
SPSZ 8;  
PAT 16, 300;  
UNIT C;  
MZP 4, 24; BOX 6, -6;  
MZP 7, 24; BOX 3, -3; BLK 3, -3;  
MZP 4, 24;  
TEXT, L;  
TEXT 'Portrait orientation';  
SPO L;  
MZP 12, 10; BOX 6, -6;  
MZP 15, 10; BOX 3, -3; BLK 3, -3;  
MZP 12, 10;  
TEXT, L;  
TEXT 'Landscape orientation';  
PAGE;  
EXIT;
```

Printout



SPSZ — Set Paper Size

Format

SPSZ *size-code* [, *ratio*] ;

Parameters

size-code:

- 0 = Same as the size of the paper cassette
- 1 = Monarch (3-7/8 × 7-1/2 inches)¹
- 2 = Business (4-1/8 × 9-1/2 inches)²
- 3 = International DL (11 × 22 cm)[†]
- 4 = International C5 (16.2 × 22.9 cm)[†]
- 5 = Executive (7-1/4 × 10-1/2 inches)[†]
- 6 = US Letter (8-1/2 × 11 inches)
- 7 = US Legal (8-1/2 × 14 inches)
- 8 = A4 (21.0 × 29.7 cm)
- 9 = JIS B5 (18.2 × 25.7 cm)
- 10 = A3 (29.7 × 42 cm)
- 11 = B4 (25.7 × 36.4 cm)
- 12 = US Ledger (11 × 17 inches)
- 13 = A5 (14.8 × 21 cm)
- 14 = A6 (10.5 × 14.8 cm)[†]
- 15 = JIS B6 (12.8 × 18.2 cm)
- 16 = Commercial #9 (3-7/8 × 8-7/8 inches)[†]
- 17 = Commercial #6 (3-5/8 × 6-1/2 inches)[†]
- 18 = ISO B5 (17.6 × 25 cm)
- 19 = Custom (11.7 × 17.7 inches)^{††}
- 30 = C4 (22.9 × 32.4 cm)^{††}
- 31 = Hagaki (10 × 14.8 cm)^{††}
- 32 = Ofuku-Hagaki (14.8 × 20 cm)^{††}
- 33 = Officio II
- 36 = A3 Wide^{††}
- 37 = Ledger Wide^{††}
- 39 = 8K
- 40 = 16K
- 50 = Statement^{††}
- 51 = Folio^{††}
- 52 = Youkei 2^{††}
- 53 = Youkei 4^{††}

ratio:

- Enlargement/reduction ratio (for print systems handling A3/ledger):
- 0 = 100% [Cassette size to cassette size]
 - 5 = 70% [A3 to A4, B4 to JIS B5, A4 to A5]
 - 6 = 81% [B4 to A4, B5 to A5]
 - 7 = 86% [A4 to JIS B5]
 - 8 = 94% [A4 to letter]
 - 9 = 98% [Cassette size to cassette size × 98%]

¹ Inoperable if *ratio* is specified.

² Not supported by some print systems

Function

The SPSZ command sets the paper edge limits to the dimensions of a standard paper size, allowing for a 5 mm border of all sides. The margins are set to the edge limits. This command is particularly useful for printing on envelopes (*size-code* 1 through 4, 16, and 17). If *size-code* is 0, the printing system adjusts the edge limits to match the cassette size. *size-code* defaults to 0 at power-up and after a reset.

Size codes 10 through 12, 19, and 30 are supported only by the printing system model that handles the A3 size paper. Support for other paper sizes and reduction ratios vary depending on models. For details, see the user's manual supplied with the printing system.

If your printing system model supports the A3 size paper, the *enlargement/reduction ratio* parameter controls reduction and enlargement of printing a page sized by the *size-code* parameter. Although the *enlargement/reduction ratio* parameter is optional, an enlargement/reduction ratio setting remains effective until an SPSZ command with new enlargement/reduction ratio specified or a reset command is encountered.

Note that the *enlargement/reduction ratio* parameter is ignored unless the source paper size and the cassette paper size does not match.

The SPSZ command causes a page break, so it should be placed at the top of a file or program, or at the beginning of a new page.

Files

```
!R! CMNT Size the page as Monarch; SPSZ 1;  
EXIT;
```

```
!R! CMNT Size the page as A3 and reduce it to A4;  
    SPSZ 10, 5;  
EXIT;
```

```
!R! CMNT Size the page as letter and reduce it to  
    98%; SPSZ 6, 9;  
EXIT;
```

SPW — Set Page Width

Format

SPW *width*;

Parameter

width:

width of page, from left margin to right margin

Function

The SPW command sets the right margin, as measured from the left margin position in the unit value designated by the UNIT command.

If an attempt is made to print beyond the right margin, the characters are printed on the next line in the IBM and Epson emulation modes and discarded in the other emulations.

The SRM (Set Right Margin) command also sets the right margin, but it measures from the left edge limit of the paper.

File

```
!R! RES; UNIT C; SLM 1;
CMNT Emulation mode is 1 (IBM);
SFNT "Univers-MdCd", 18; SLPI 4;
SPW 10;
EXIT;
Line with 1-cm left margin and 10-cm width
!R! SPW 8; EXIT;
Line with 1-cm left margin and 8-cm width
!R! SPW 6; EXIT;
Line with 1-cm left margin and 6-cm width
!R! PAGE; EXIT;
```


Printout

Line with 1-cm left margin and 10-cm width

**Line with 1-cm left margin and 8-cm
width**

**Line with 1-cm left margin
and 6-cm width**

Related Command

SRM

SRM — Set Right Margin

Format

SRM *width*;

Parameter

width:

size of right margin

Function

The SRM command sets the right margin, as measured from the left edge limit of the paper in the unit value designated by the UNIT command.

If an attempt is made to print beyond the right margin, the characters are printed on the next line in the IBM and Epson emulation modes and discarded in the other emulations.

The SPW (Set Page Width) command also sets the right margin, but it measures from the left margin position.

File

```
!R! RES; UNIT C; SLM 1;
CMNT Emulation mode is 1 (IBM);
SFNT "Univers-MdCd", 18; SLPI 4;
SRM 11;
EXIT;
Line with 1-cm left margin and 11-cm right margin
!R! SRM 9; EXIT;
Line with 1-cm left margin and 9-cm right margin
!R! SRM 7; EXIT;
Line with 1-cm left margin and 7-cm right margin
!R! PAGE; EXIT;
```

Printout

Line with 1-cm left margin and 11-cm right margin

Line with 1-cm left margin and 9-cm right margin

Line with 1-cm left margin and 7-cm right margin

Related Command

SPW

SRO — Set Raster Options

Format

SRO *presentation-mode* [, *width* , *height*] ;

Parameters

presentation-mode:

0 = physical raster direction
1 = logical page direction

width:

width of raster area in dot; min. 1 to max. 4200 dots

height:

height of raster area in dot; min. 1 to max. 4200 dots

Function

The SRO command specifies the presentation mode which defines the orientation of the raster image. If the *presentation-mode* is 0 (default), the printing system creates the raster image along the width of the physical page regardless of the current page orientation. With the *presentation-mode* 1, however, the raster graphics prints in the orientation of the logical page. Presentation modes 0 and 1 are depicted in the next page.


The *width* and *height* options define the size of the raster area which has the current position as its left-top corner of the area. These options are measured in the dot unit regardless of the current designation of the UNIT command.

Raster data extending beyond the border along the width of the raster area is clipped in the increment of dot.


After a RVRD command sequence finishes, the current position is moved down by 1/300 inch.

presentation-mode = 0

The width and height of the raster area are defined based on the physical page regardless of the current page orientation. The current page orientation affects the position of the current point, however.

**presentation-mode = 1**

The orientation of the logical page determines the width and height of the raster image. The current position also depends on the current page orientation.

**File**

```
!R! SPO L; SRO 0; STR 75;
  RVRD;
  2;A|0; 2;GG0; 2;^S<; 2;~{>;
  2;^S<; 2;GG0; 2;A|0; ENDR;
  PAGE;
EXIT;
```

SROP — Set Raster Operation

Format

```
SROP value ;
```

Parameter

value:

logical operation value: 0 to 255

Function

The SROP command applies logical functions (AND, OR, XOR, NOT) to source, texture, and destination. Possible logical operations provided by this command is 255. These logical operations directory map to their raster operations (ROP3).

Logical operation value is eset to the default of 252 by the RES command. An error results and this command has no effect if the ROP3 value is outside the range 0 to 255.

File


```
!R!
DAM;
RES;

MCRO DBOX;
NEWP;
PMRP 0,0;
PDRP 0.5,0;
PDRP 0,0.5;
PDRP -0.5,0;
CLSP;
SROP 252;
GPAT 0;
FILL;
NEWP;
PMRP 0.2,0.2;
PDRP 0.5,0;
PDRP 0,0.5;
PDRP -0.5,0;
CLSP;
SROP %1;
PAT 25;
FILL;
MRP 0.8, -0.2;
ENDM;

MCRO TROP;
MZPO,2;
CALL DBOX,0;
CALL DBOX,1;
```

```
CALL DBOX, 2;  
CALL DBOX, 3;  
RES; EXIT;
```

Printout



SSTK — select Sorter STacker

Format

SSTK [*tray, tray, ...*]; or SSTK 0;

Parameter

tray:

number from 1 to the maximum (bottom)

Function

The SSTK command is effective only when the optional sorter is installed to the printing system. It instructs the printing system to deliver printed pages in the sorter mode. Each specified tray receives a complete set of pages, one copy of each.

If the parameters are omitted, the top *n* trays are used, where *n* is the copy count (equal or less than the number of trays). If you specify the copy count which is greater than the number of the trays being selected, the bulk tray receives the overflowed pages. If the tray parameters are specified, the COPY command is unnecessary as the printing system automatically makes the copy count to be the same as the number of the tray parameters specified.

Upon receiving the SSTK command, the printing system automatically performs a page feed.

If the parameter is 0 for SSTK (SSTK 0;) this command disables sorting and delivers all pages to the bulk output tray.



Note

This command is irrelevant to certain combination of photocopier models and option finisher models.

File

```
!R! RES; SSTK 1, 3, 5; EXIT;
Three copies of this file will be printed.
The output goes to sorter trays 1, 3, and 5.
!R! PAGE; EXIT;
If the tray parameters were omitted (SSTK; COPY3;), sorter
trays 1, 2, and 3 would be used.
!R! PAGE; EXIT;
```


STAK — select paper STAcKer

Format

STAK *stacker*;

Parameter

stacker:

- 1 = printing system's face-down output tray
- 2 = face-up output tray of the printing system or the option stacker
- 3 = face-down output tray of the option stacker
- 51 = face-up output tray of the option mailbox
- 52 = face-down output tray of the option mailbox
- 61 = face-down output tray of the option job separator (or the DF-610 subtray)
- 62 = face-up output tray of the DF-610 subtray

Function

The STAK command instructs the printing system to deliver pages to the corresponding output stack of the printing system or the option stacker. The selected tray remains in effect until changed by another STAK command. The default value is 1: the printing system's face-down tray.

If an optional device such as a stacker, a document finisher, or a sorter is installed, the *stacker* value 2 instructs the printing system to deliver printed pages to its face-up tray; the value 3 delivers the pages to its face-down tray. The *stacker* values of 51, 52, and 61 are valid for certain copiers installed with the applicable option device.

The STAK command should be placed at the top of a file or program.

File

```
!R! STAK 2; EXIT;
PAGE 1
On printers with only one paper tray this file will be
delivered face-down as usual, page 1 first.
!R! PAGE; EXIT;
PAGE 2
PAGE 2 next, and so on, stacking in the correct order.
!R! PAGE; EXIT;
PAGE 3
But on printers with two paper trays this file will be
delivered to the face-up tray with the pages in reverse
order, page 3 on top.
!R! PAGE; EXIT;
```

STAT — STATus

Format

```
STAT [mode];
```

Parameter

mode:

- 0 = prints a user status page (default)
- 1 = prints a service person status page

Function

The STAT command instructs the printing system to print a page of status information, starting on a new page. The status page lists the software version and release date as well as the following information:

- The current unit (inches, centimeters, points, or dots; not available in KPDL mode)
- The current page orientation, font, and country code (not available in KPDL mode)
- Current margins, line spacing, character spacing, and graphics line width, as measured in the current unit (not available in KPDL mode)
- Current memory allocations

The above status settings are temporary conditions set by PRESCRIBE commands. They revert to their default values when power is switched off. Other status page information includes:

- Miscellaneous status: a list of options stored in permanent memory. See *Chapter 6* of the *Technical Reference* manual for the definitions of the option values. These values do not change when power is switched off. Only the FRPO command can alter them.
- Service information to inform a service person.
- Nonresident fonts in memory: Fonts generated by the LDFC command, downloadable soft fonts from software or read from a memory card. Font size and numbers appear alongside the font name. These fonts are lost when power is switched off.




Note


Items and values on the status page may vary among the printing system's firmware versions.

Depending on the parameters, different information may be printed on the status page. (There are two types of status pages: one intended for the end user, and one intended for service personnel.) Also, the printing system can print the status page in different languages in accordance with the language setting of the printing system's control panel. The default is the English status page.

Sample Status Page

User status page




FS-1800+ Page Printer

STATUS PAGE

[0A00/A002][C1][30.00 MX1][00]
Firmware version: 82.00-01*
Released: 07/Aug/2001

Hardware Information

MP Tray A4 Plain
Cassette 1 A4 Plain

Ecoprint mode OFF
Buzzer control ON
Host buffer size 200 KB
Sleep timer 15 Minutes
FF time out 30 Seconds

Memory

Standard memory 8192 KB
Option memory
Slot 1 Empty
Slot 2 Empty
Total 8192 KB
User available 3618 KB
RAM disk OFF
Write protect
Total 0 KB
User available 0 KB

Page Information

Resolution 300 DPI
Copies 1
Total page 2138

Installed Options


Hard-disk drive NO
Option ROM NO
Memory card NO

Emulation

Line printer
IBM Proprinter
Diablo 630
Epson LQ-850
PCL 6
KPDL2

Gauge

Service status page


PAGE PRINTER STATUS PAGE FS-1800+

Firmware version: 82.00-01*
Released: 07/Aug/2001

Standard setting		Emulation		Paper		Margin		Memory	
Interfaces	Unit	Line printer		Feeder	Size	Top		Total	
Serial	*Inch	*IBM Proprinter		MP Tray	A4 01	0.000		8192 KB	
*Parallel	Centimeter	Diablo 630		*Cassette 1	A4 01	Bottom	11.300	User available	3618 KB
	Dots	Epson LQ-850				Left	0.000	Font	
	Points	PCL 6				Right	7.876	Ibyte code	8
Copies	Page orientation	KPDL2							
1	*Portrait Landscape								

Interface independent parameters		Interface dependent parameters			
Environment	Parameters	Environment	Parameters	Serial	Parallel
Reserved	B0 00	User top margin	A1+A2/100	00.00	00.00
Reserved	B7 01	User left margin	A3+A4/100	00.00	00.00
Reserved	B8 00	User page length	A5+A6/100	13.61	13.61
Default pattern switch	D0 00	User page width	A7+A8/100	13.61	13.61
		Page orientation	C1	00	00
		Default font number	C2.C3.C5 *1	00000	00000
		Font switch	C8	00	00
		Timeout	J2	06	06
		Parallel I/O	o0/o1/o2	70/01/02	
			P1	06	01
				01/01	01/02
				00/06	00/06

STM — Set Top Margin

Format

STM *distance* ;

Parameter

distance:

size of top margin

Function

The STM command sets the top margin, in the unit value designated by the UNIT command.

A top margin setting in a document file should be placed at the beginning of the file or at a page break.

If the top margin is moved below the current cursor position, the cursor moves to the new margin. But if the top margin is moved above the current cursor position, the cursor does not move.


File

```
!R! RES; UNIT C;  
SFNT "Helvetica-Nr", 18;  
STM 2;  
EXIT, E;  
First page with 2-cm top margin
```

```
!R! RES; UNIT C;  
SFNT "Helvetica-Nr", 18;  
STM 4;  
EXIT, E;  
Second page with 4-cm top margin
```

```
!R! PAGE; EXIT;
```

Printout



First page with 2-cm top margin

Second page with 4-cm top margin

STPC — set STaPle Counter

Format

STPC [*staple-count* or letter M (or m)] ;

Parameter

staple-count:

number of sheets to be stapled at a time (When this parameter is omitted, '0' is assumed):

2 to 50¹ = divides output pages into stacks of pages of this value and staples each of them. The odd stack is also stapled.

M = divides output pages into stacks of pages equal to the maximum allowable number of sheets of stapling and staples each of them. The odd stack is also stapled.

0 = same as M above but does not divide nor staple a job at all if the total number of pages of the job is more than the maximum allowable number of sheets of stapling. The odd stack is neither stapled.

-1 = same as 0 above but the odd stack is stapled.

Function

The STPC command specifies the number of sheets that the stapler (finisher) staples a job. This command does not execute stapling therefore should be used with the STPL (staple) command.

For more simplified explanation on the staple-count, here is an example.

To print a 50-page job using a DF-31 finisher (20 pages stapler):

STPC parameter	Stapling operation
M or m	20 sheets, 20 sheets, and 10 sheets are stapled.
15	15 sheets, 15 sheets, 15 sheets, and 5 sheets are stapled.
0	No stapling occurs.
-1	Only the last 10 sheets are stapled.

File

To staple every 5 sheets of a job at the center 2 staples, printing 15 copies of the job:

```
!R! STPL 3; STPC 5;
JOBS 2, 15, 1, 0, 1, "Interim Report 1", "Steve", "09/22/02
19:24";
TEXT "This is Interim Report 1 ..."; PAGE;
UNIT C; BOX 5,10; PAGE;
JOB; EXIT;
```

¹ This value depends on the stapler to be used.

STPL — STaPLe

Format

STPL [*position*];


Parameter

position:

stapling position:

- 1 = bottom left, one staple
- 2 = top left, one staple
- 3 = left, two staples (straight for Legal size)
- 50 = top left, one staple
- 51 = top right, one staple
- 52 = left, two staples
- 53 = right, two staples
- 54 = top, two staples

Staple parameters and positions



Function

The STPL command staples the pages of a job at the position specified by the *position* parameter (See above.) using a document finisher incorporating the stapler unit. With position 2 (top left), stapling is automatically straightened (vertical) when Legal size paper is used.

Parameter values 50 to 54 are valid for certain copier models only.

The number of sheets that the stapler can staple at a time depends on the STPC command setting.

The job to be stapled must be defined by the JOBS command. Stapling is valid only when the STPL command is issued preceding a JOBS command sequence. The STPL command automatically deactivates Jog and Job Separation if these functions are enabled in advance. The output stack is enforced to be the stapler output ignoring the

current output stack which may be specified by STAK. (For the document finisher such as the DF-31, the MSTK command may be used to guide stapled pages to one of the output trays as desired.)

STPL command causes a page break.

File

```
!R!  
STPL 3; CMNT Two staples on the left;  
JOBS 0,3,0,0;  
TEXT "Stapling Test, Page 1";  
PAGE;  
TEXT "Stapling Test, Page 2";  
PAGE;  
TEXT "Stapling Test, Page 3";  
PAGE;  
TEXT "Stapling Test, Page 4";  
PAGE;  
TEXT "Stapling Test, Page 5";  
JOBT;  
EXIT;
```

Related Commands

STPC

STR — SeT dot Resolution

Format

STR dot-resolution ;

Parameter

dot-resolution:


75, 100, 150, 200, 300, 600, or 1200

Function

The STR command sets the dot resolution for raster graphics to 75, 100, 150, 200, 300, 600, or 1200 dots per inch. The 200, 600, and 1200 dpi resolutions are available only for the printing system models that support the 600 or 1200 dpi printing resolution. The default value is 75 dots per inch.

The STR command affects only PRESCRIBE raster graphics. It does not change the resolution of vector graphics commands or HP LaserJet raster graphics.

The following diagrams show examples of printing a raster graphics data using different dot resolution.



STRK — STRoKe current path

Format

STRK ;

Parameter

None

Function

The STRK command strokes the constructed path onto the current page according to the following current graphics state:

- Pen diameter (SPD)
- Dash pattern (DPAT)
- Fill pattern (FPAT)
- Line cap (SCAP)
- Line join (SLJN)
- Flatness (FLAT)

A path itself will not place a line or graphic on a page because PRESCRIBE path-construction commands only define the path. For the path to be drawn or filled as a line, box, character, etc., on the page, the STRK command must be issued.

After stroking the current path, the STRK command clears the path and moves the cursor to the end of the path.


File

```
!R! RES; UNIT C;

NEWP;
PMZP 7, 17;
SPD .2;
PDRP -2.5, 0, 0, 2.5, 2.5, 0, 0, -2.5;
PDRP 2.1, 1.25, 1.25, -2.1, -2.1, -1.25, -1.25, 2.1;
PMRP 0, 2.5;
PDRP 2.1, -1.25, 1.25, 2.1, -2.1, 1.25, -1.25, -2.1;
STRK;

PAGE;
EXIT;
```

Printout



Related Commands

FILL, NEWP

SULP — Set UnderLine Parameters

Format

SULP *distance*, *thickness*;

Parameters

distance:

distance below the baseline

thickness:

thickness of the underline

Function

The SULP command sets the thickness of any subsequent underlining as well as the distance below the baseline, in the unit value designated by the UNIT command. The distance parameter is limited only by the size of the page. Negative values can be used to obtain overlines.

The minimum underline thickness is one dot. The maximum underline thickness is 127 dots. The default values include:

- Distance below baseline: 7 dots
- Thickness of underline: 4 dots
- One dot equals approximately 0.0083 cm or 0.0033 inches.

The SULP command does not affect the underline character (), which is an independent character with ASCII code 95 (hex 5F). If your word-processing software writes underlines by backing up and printing the underline character instead of using the printing system's automatic underline feature, the SULP command will control only underlines printed by the U-option in PRESCRIBE commands such as TEXT.

File

```
!R! RES; UNIT C; SLS 0.8;
SFNT 'Helvetica-Nr', 12;
TEXT 'Default paramaters', E, U; TEXT ' , ' , E;
SULP .2, .1;
TEXT 'heavier underlining', N, U;
TEXT 'and even', E;
SULP -.32, .04;
TEXT 'OVERLINING', E, U;
TEXT '!';
PAGE;
EXIT;
```

Printout

Default parameters, heavier underlining
and even OVERLINING!

TATR — apply Tray ATtributes

Format

```
TATR [source[, thickness, print-density]];
```

Parameters

source

paper source:
0 = MP tray
1 = main cassette
2 = second cassette

thickness

thickness of paper:
0 = normal (normal)
1 = thick
2 = thin

print-density

print density:
lighter (1) to darker (5), default is 3.

Function

The TATR command applies a paper source (MP tray or cassette) thickness of paper and print density. This enables the printing system to optimize the electrical properties (such as the fuser temperature and the transfer bias) so that the best print quality is achieved with different thicknesses of paper.

The attributes are reset for a tray when *thickness* and *print-density* parameters are omitted. All trays are reset when all three parameters are omitted.



Note

This command is relevant to the FS-1000 printer only.

File

To attribute the MP tray for thick paper and normal density:

```
!R! TATR 0, 1, 3; EXIT;
```

To reset the second paper cassette (option feeder) to the defaults:

```
!R! TATR 2; EXIT;
```

To reset all paper sources to the defaults:

```
!R! TATR; EXIT;
```

TEXT — print TEXT

Format

```
TEXT [ 'string' ] [ , option ] [ , U-option ] ;
```

Parameters

string:

any character string

option:

B = Leave cursor at beginning of string
E = Move cursor to end of string
L = Move cursor down one line
N = Move cursor to beginning of next line

U-option:

Underline text

Function

The TEXT command prints the given string at the cursor location, then moves the cursor as designated by the option. The string to be printed must be enclosed in either single or double quotation marks. (If the string includes one kind of quotation mark, use the other kind to enclose it.) If the string is omitted, nothing is printed. TEXT , N ; can be used to print a blank line, for example. If the option is omitted, the cursor is not moved (same as option B). If the U-option is specified, the string is underlined. The underline thickness and location can be set by the Sulp command.

The TEXT command can be used to position a line of text at an exact location.

The length of the string is unlimited, except that the command as a whole may not be longer than 255 characters. Carriage-return and linefeed codes in the string are ignored.

File

```
!R! RES; UNIT C; MAP 2, 2;
SFNT 'Univers-MdCd';
TEXT 'Check desired options:', L, U;
FONT 2;
BOX .1, -.1; TEXT 'Power steering', L;
BOX .1, -.1; TEXT 'CD auto changer', L;
BOX .1, -.1; TEXT 'Air conditioner', L;
PAGE;
EXIT;
```

Printout

Check desired options:

- Power steering
- CD auto changer
- Air conditioner

Related Commands

CTXT, RTXT

TPRS — Text PaRSing

Format

TPRS *mode* ;

Parameter

mode:

- 0 = ANK (alphabetic numeric and kana)
- 1 = shift JIS (2-byte code)

Function

The TPRS command allows character codes to be interpreted as 1-byte or 2-byte character code while printing using a text command.

To parse such characters, the current font must supports 2-byte characters and a 2-byte symbol set such as Windows 3.1J (19K). The parsing method will automatically be shift JIS encoding (2) if the default font and symbol set is of 2-byte.

This command is valid only in HP emulation.

TRSM — TRansparency Separate Mode

Format

```
TRSM mode[, tray];
```

Parameters

mode:

Transparency separate mode:

0 = Off

1 = Inserting a blank page

2 = Inserting a page which the transparency content is printed on

tray:

The number of the paper source which the inserting paper loads from: 1 or above. The number varies depending on the printing system and its configuration. When omitting the *tray* parameter, 1 is automatically set to the *tray* parameter.

Function

The TRSM command instructs printing system to insert blank or printed page between each transparency. The printing system loads transparency from MP tray and inserting paper from the paper source other than MP tray. When face-down printing, the printing system print the transparency page first. When face-up printing, the printing system print the blank or printed page first.

File

```
!R!  
MTYP 2;  
TRSM 1,1;  
JOBS 0,1,2,0;  
TEXT "page 1";  
PAGE;  
TEXT "page 2";  
PAGE;  
JOB;  
EXIT;
```

UNIT — set UNIT of measurement

Format

UNIT *code* ;

Parameter

code:

I = inches
 C = centimeters
 P = points
 D = dots

Function

The UNIT command sets the unit of measurement used in other PRESCRIBE commands.

The initial unit is inches. Other supported units of measurement can be calculated as follows:


- 1 inch = 2.54 centimeters
- 1 inch = 72 points
- 1 inch = 300 dots (300 dpi)/600 dots (600 dpi)/1200 dots (1200 dpi)


UNIT commands are not retroactive. They do not affect values set by previous commands.

File

```
!R! UNIT C; MRP 3, 3;
SFNT "Univers-MdCd", 12;
MRP 3, 3;
CIR .35;
TEXT '          7mm pistol';
UNIT I; MRP 0, 1;
CIR .225;
TEXT '          .45 caliber pistol';
PAGE;
EXIT;
```

Printout

 7mm pistol

 .45 caliber pistol

UOM — Unit Of Measurement per dots

Format

UOM *number*;

Parameter

number:

96, 100, 120, 144, 150, 160, 180, 200, 225, 240, 288, 300, 360, 400, 450, 480, 600, 720, 800, 900, 1200, 1440, 1800, 2400, 3600, 7200

Function

The UOM command applies only to the printing system models that support the 600 dpi resolution. It defines the unit of measurement in dots per inch for cursor movements. The command requires that the UNIT value be set to D.

The default unit of measurement is 300 (dpi). For example, the cursor movement of 50 dots equals 1/6 inches.

The command accepts only the above numbers for the unit of measurement. Any other numbers will be ignored when used with this command.

File

```
!R! FRPO N8, 1; RES;
UNIT C;
SFNT "Helvetica-Nr", 9;


MZP 5, 5;
TEXT "UOM 120; BOX 100, 100;", L;
UNIT D;
UOM 120;
BOX 100, 100;

UNIT C;
MZP 5, 9;
TEXT "UOM 720; BOX 100, 100;", L;
UNIT D;
UOM 720;
BOX 100, 100;

PAGE; EXIT;
```

Printout

UOM 120; BOX 100, 100;



UOM 720; BOX 100, 100;



Related Command

UNIT

VMAL — Virtual Mailbox ALias

Format

```
VMAL mailbox, alias[ , password] ;
```

Parameters

mailbox:

number from 1 to 255 or character string representing number or alias of virtual mailbox

alias:

Character string representing the alias being set for the virtual mailbox given by *mailbox* (up to 31 bytes)

password:

The password for the virtual mailbox given by *mailbox* if a password has been set for it.

Function

The VMAL command sets an alias for a virtual mailbox. When an alias is set for a virtual mailbox using this command, that alias may be used to identify the mailbox just like the virtual mailbox number.

If this command is used on a virtual mailbox for which an alias has already been set, the old alias will be overwritten. If the alias specified has already been set for another virtual mailbox, an error will be displayed on the operator panel and the specification will be invalid.

If an alias is to be set for a virtual mailbox for which a password is set, that virtual mailbox's password is required. An error will be displayed on the operator panel if no password or the incorrect password is specified. (See the *Technical Reference* manual.)

Note that if command syntax is incorrect, such as when an alias longer than 31 bytes is specified, the command will be invalid and no error will be displayed on the operator panel.

The master password is valid for all virtual mailboxes.

File

To set the alias *MyMailbox* for virtual mailbox number 3 for which the password 9999 has been set:

```
!R! VMAL 3, "MyMailbox", 9999; EXIT;
```

VMOB — Virtual Mailbox Output Bin

Format

VMOB *mailbox*;

Parameter

mailbox:

number from 1 to 255 or character string representing an alias

Function

The VMOB command specifies virtual mailboxes that receive print jobs. It is possible to specify multiple virtual mailboxes by issuing multiple instances of this command.

If no virtual mailbox corresponds to the alias given, an error will be displayed on the operator panel. (See the *Technical Reference* manual.)



This command is irrelevant to photocopy virtual mailbox function.

Files

To stack jobs on virtual mailbox number 3:

```
!R! VMOB 3; EXIT;
```

To stack jobs on the virtual mailbox having the alias *MyMailbox*:

```
!R! VMOB "MyMailbox"; EXIT;
```


VMPW — set Virtual Mailbox PassWord

Format

VMPW *mailbox*, *current-password*, *new-password*;

Parameters

mailbox:

0 = Sets the master password
 1 to 255 = Number of virtual mailbox for which the password is set
 Character string = Alias of virtual mailbox for which the password is set

current-password:

1 to 65535 = Current password for the virtual mailbox specified by *mailbox*
 0 = Indicates no password is set for the virtual mailbox specified by *mailbox*

new-password:

1 to 65535 = New password for the virtual mailbox specified by *mailbox*
 0 = Indicates no password is to be set for the virtual mailbox specified by *mailbox*

Function

The VMPW command sets the output password for a virtual mailbox.

If this command is used on a virtual mailbox for which a password is already set, that password must be specified by *current-password* or the command will be invalid and an error will be displayed on the operator panel. An error also will result if the password specification fails or no virtual mailbox corresponds to the alias given. For error displays, see the *Technical Reference* manual.

The master password is valid as the current password for all virtual mailboxes.

Virtual mailboxes for which a password has been set are not shown on the VMB LIST page (See page 134).



Note

This command is irrelevant to photocopy virtual mailbox function.

Files

To set 9999 as the password for virtual mailbox number 3 for which no password is set:

```
!R! VMPW 3, 0, 9999; EXIT;
```

To set no password for the virtual mailbox with the alias *MyMailbox* for which the password 9999 had been set:

```
!R! VMPW "MyMailbox", 9999, 0; EXIT;
```

To change the password for the virtual mailbox with the alias *Purchase Dept.* from 9999 to 3333:

```
!R! VMPW "Purchase Dept.", 9999, 3333; EXIT;
```

WIDE — set WIDE A4 mode

Format

WIDE *mode* ;

Parameter

mode:

0 = Off


1 = On

Function

The wide A4 function widens the A4 printing width in order to prevent unnatural line breaks or clipping of graphics in cases where a document created as a letter-size document is output in A4 size. The WIDE command sets and clears the wide A4 function. The setting made with this command is initialized by the RES and FRPO commands.

Printing area

- The width of the A4 printing area is set to that of Letter size.
- The width is increased symmetrically to the left and right.
- The width is always increased in the shorter direction (either width or height, whichever is shorter). Thus if the document is set to landscape orientation, the height is increased.
- The printing area for sizes other than A4 is not affected.



WRED — WRite EnD

Format

WRED ;

Parameter

None

Function

The WRED command terminates a sequence of writing data to the memory card or to the hard disk, using one of the commands listed below.

The WRED command can be renamed to a string of ASCII characters if necessary. For details, refer to the RWER T or RWRF T command (See page 249 or 257).

Related command

RWER S, RWER W, RWRF W

XBAR — print two dimensional barcode

Format

```
XBAR [ data-length ] ; data-string ; ENDB ;  
or  
XBAR buffer-name ; ENDB ;
```

Parameters

data-length:

length of data in bytes, terminated with a semicolon (;)

If this value is omitted, *data-string* is counted all characters after the first semicolon until the program sees ;ENDB; .

data-string:

data to be encoded, terminated with a semicolon (;) immediately followed by an ENDB command (256 international characters or binary data)

buffer-name:

name of buffer previously defined by XBUF command


Function

The XBAR command prints a two dimensional barcode (sometimes referred to as PDF417 symbol). The ENDB must follow as a terminator for the barcode data string. Note that carriage return, line feed, space and all other characters must be counted as part of data in the data string field. If *data-length* is omitted, 0, negative value, or non integer, then *data-string* is counted all characters until the program sees ;ENDB;. If *data-length* is 0, the semicolon to terminate the data cannot be omitted. (Command as XBAR;;ENDB;.) It is necessary to have no space between “;” and “ENDB;”.

Files

```
!R! XBAR; The Milky Way is a gravitationally bound
collection of roughly a hundred billion stars. Our Sun is
one of these stars and is located roughly 24,000 light years
(or 8000 parsecs) from the center of our the Milky Way.
The Galaxy has three major components. A thin disk
consisting of young and intermediate age stars - this disk
also contains gas and is actively forming new stars. Dust in
the disk makes it appear orange in the picture. Dust absorbs
blue light more than red light and thus makes stars appear
reddish. Our Galaxy has spiral arms in its disk - these
spiral arms are regions of active star formation.;ENDB;
```

Printout



XBCP 0 — select barcode type/reset all other XBCP parameters

Format

XBCP 0 [, *type*] ;

Parameter

type:

two dimensional barcode type:

1 = PDF417 (only type currently supported)

Function

The XBCP 0 command selects a barcode type and resets all parameters to the defaults.

File

See XBCP 1 (page 366) command page.

XBCP 1 — specify narrowest element width

Format

```
XBCP 1, element-width;
```

Parameter

element-width:

narrowest bar and space width:

More than or equal to 1 dot. If less than 1 dot, truncated to 1 dot. Default is 3 dots.

Function

PDF417 supports 8 different bar and space widths. The each width is a multiple of 1 to 8 times of the narrowest element width. The XBCP command set to mode 1 specifies the narrowest element width (narrowest bar width) in the current unit as designated by a UNIT command.

In the following example, the element width is set to .02 inches as the default unit is inches. Minimum width allowed by the printing system is 1 dot. Minimum readable width is probably higher. Minimum width meeting AIM recommendation for open systems is 3 dots. After the first example has been printed, the printing system moves cursor 1 inch down to print another example in which the minimum element width is reverted to its default (3 dots).

File

```
!R! RES;
MZF 0, 0;
XBCP 1, .02;
XBAR;Min. element width is .02 inches;ENDB;
MZF 0, 1;
XBCP 0;
XBAR;Min. element width is reset (to 3 dots);ENDB; EXIT;
```

Printout

Minimum element width is 0.02 inch.



Default minimum element width (3 dots)



XBCP 2 — specify error correction level by percentage

Format

```
XBCP 2, percentage;
```

Parameter

percentage:

value between 0 and 400

Function

PDF417 supports an error correction scheme. The XBCP 2 command specifies the error correction level as a percentage of the number of code word values for the data to be encoded. An illegal value defaults to 10.

Note that XBCP 3, as explained on the next page, sets the error correction level by predetermined values, and the execution of XBCP 3 disables the setting of XBCP 2, even if XBCP 2 is executed after XBCP 3, until RES (reset) or XBCP 0.

File

To set 50% error correction level:

```
!R! RES; XBCP 2, 50;  
XBAR;This is an error correction level test;ENDB;  
EXIT;
```

XBCP 3 — set error correction level

Format

```
XBCP 3 , level;
```

Parameter

level:

error correction level, integer from 0 to 8

Function

The XBCP 3 command sets the error correction level by predetermined values. An illegal value defaults to 0. Note that the execution of XBCP 3 overrides the setting of XBCP 2 and remains valid until RES (reset) or XBCP 0.

File

To sets the error correction level to level 4:

```
!R! RES;  
XBCP 3 , 4;  
XBAR;Error correction can also be set by level;ENDB;  
EXIT;
```

XBCP 4 — set data code word rows

Format

```
XBCP 4, number;
```

Parameter

number:

number of data code word rows, ranging from 3 to 90

Function

The XBCP 4 command sets the number of data code word rows in the PDF417 barcode. Illegal values make XBCP 4 ignored. XBCP 6 default provides default.

Note that this command is not effective if XBCP 8 is issued. Precedence of XBCP modes is: 4<8, 4>6, 4+8< 6.

File

To set 20 rows of barcode:

```
!R! RES;
XBCP 4, 20;
MZP 1.5, 1.5;
XBAR;Visit Sea World, the world famous San Diego Zoo &
Wild Animal Park, historic Old Town, Tijuana (Mexico),
Mission Bay, the downtown Gaslamp District and all
those great beaches. And, as the natives know, the end
of September is the best time of year.;ENDB;
EXIT;
```

Printout



Number of rows is set to 20.



Number of rows is normal.

XBCP 5 — set data code word columns

Format

```
XBCP 5, number;
```

Parameter

number:

number of data code word columns, ranges between 1 and 30 inclusive (Illegal values make XBCP 5 ignored.)

Function

The XBCP 5 command sets the number of data code word columns in the PDF417 barcode. There are also row indicator code words at each end of the row which are not counted as data code words. XBCP 6 default provides default.

Note that this command is not effective if XBCP 8 is issued. Precedence of XBCP modes is: 5<8, 5>6, 5+8<6.

File

To set 30 columns of barcode:

```
!R! RES;  
XBCP 5, 30;  
MZP 1.5, 1.5;  
XBAR;Visit Sea World, the world famous San Diego Zoo &  
Wild Animal Park, historic Old Town, Tijuana (Mexico),  
Mission Bay, the downtown Gaslamp District and all  
those great beaches. And, as the natives know, the end  
of September is the best time of year.;ENDB;  
EXIT;
```

Printout

Number of columns is set to 30.



Number of columns is normal.



XBCP 6 — determine aspect ratio of vertical height and horizontal width

Format

XBCP 6, *height*, *width*;

Parameters

height/width:

parts of the aspect ratio of vertical height and horizontal width, ranging between 10: 1 (XBCP 6, 10, 1) and 1: 10 (XBCP 6, 1, 10)

Function

The XBCP 6 command determines aspect ratio of vertical height and horizontal width of the barcode. Illegal values default to 1: 2. Default is 1: 2.


Note that this command is not effective if XBCP 4 and/or 5 is issued, however, this has the highest priority if both XBCP 8 and XBCP 4 and/or 5 are issued together. Precedence of XBCP modes is: 4/5<8, 4/5>6, 4/5+8<6.

File


To set aspect ratio 3:2 in vertical and horizontal direction:

```
!R! RES;
XBCP 6, 3, 2;
XBAR;Visit Sea World, the world famous San Diego Zoo & Wild
Animal Park, historic Old Town, Tijuana (Mexico), Mission
Bay, the downtown Gaslamp District and all those great
beaches. And, as the natives know, the end of September is
the best time of year.;ENDB;
EXIT;
```

Printout



Aspect ratio is 3:2 in vertical and horizontal directions.



Aspect ratio is normal.

XBCP 7 — determine bar height

Format

```
XBCP 7, mod-ratio;
```

Parameter

mod-ratio:

ratio of module height to element width, ranging from 1 to 10

Function

The XBCP 7 command determines the bar height by a ratio to the narrowest element width. An illegal value defaults to 3.

File

To set the module height twice as high as the element width:

```
!R! RES;
XBCP 7, 2;
XBAR;Visit Sea World, the world famous San Diego Zoo & Wild
Animal Park, historic Old Town, Tijuana (Mexico), Mission
Bay, the downtown Gaslamp District and all those great
beaches. And, as the natives know, the end of September is
the best time of year.;ENDB;
EXIT;
```

Printout



The bar height is twice as high as the element width. Since the default bar height is a triple of the element width, the barcode aspect ratio is now in ratio of 2:3. (See XBCP 6.)



Original (default) barcode

XBCP 8 — automatically set rows and columns

Format

XBCP 8 ;

Parameter

None

Function

The XBCP 8 command permits the system to automatically set rows and columns so that it satisfies the requirement that the multiplication of rows times columns is less than or equal to 928. Default is disabled.

Note that the XBCP 8 overrides XBCP 4 and/or 5 which set the number of rows and columns. However, this command is not effective if both XBCP 6 and XBCP 4 and/or 5 are issued together. Precedence of XBCP modes is: 4/5<8, 4/5+8<6.

XBCP 9 — enable truncation

Format

XBCP 9 ;

Parameter

None

Function

The XBCP 9 command can generate a truncated PDF417 symbol that omits the right row indicator and the stop pattern. Default is non-truncation.

XBCP 10 — enable file name

Format

```
XBCP 10 , 'file-name' ;
```

Parameter

file-name:

string containing a file name

Function

A Macro PDF417 symbol contains an optional file name field. The XBCP 10 command enables this field and defines its contents.

The string must be surrounded by single or double quote marks, be less than 200 characters long and only contain EXC mode compaction characters (20H~7FH). This command is ignored if the string does not satisfy these conditions. The execution of this command automatically sets Macro PDF417 symbol mode.

File

```
!R! RES ;  
XBCP 10 , 'A00254' ;  
MZP 1,0 ; XBCP 18,0 ;  
XBAR ; ...  
...  
EXIT ;
```

XBCP 11 — enable block count

Format

```
XBCP 11;
```

Parameter

None

Function

A Macro PDF417 symbol contains an optional block count field. The XBCP 11 command enables this function. Default is not enabled.

The execution of this command also automatically sets Macro PDF417 symbol mode.

File

```
!R! RES;  
XBCP 11;  
MZP 1,0; XBCP 18,0;  
XBAR; ...  
...  
EXIT;
```

XBCP 12 — enable time stamp

Format

```
XBCP 12, time;
```

Parameter

time:

total number of seconds since 00:00:00 AM January 1, 1970 Greenwich mean time

Function

A Macro PDF417 symbol contains an optional time stamp field. The XBCP 12 command enables this function. The execution of this command also automatically sets Macro PDF417 symbol mode.

File

To set time stamp at 06:15:30AM on Jan. 2, 1970:

```
!R! RES;  
  
XBCP 12, 108930;  
MZP 1,0; XBCP 18,0;  
XBAR; ...  
...  
EXIT;
```

XBCP 13 — enable sender ID

Format

```
XBCP 13, 'sender-id' ;
```

Parameter

sender-id:

string containing sender ID

Function

A Macro PDF417 symbol contains an optional Sender ID field. The XBCP 13 command enables this field and defines its contents.

The string must be surrounded by single or double quote marks, be less than 200 characters long and only contain EXC compaction mode characters (20H~7FH). This command is ignored if the string does not satisfy these conditions. The execution of this command automatically sets Macro PDF417 symbol mode.

File

```
!R! RES;  
XBCP 13, "BETTERDAYSINC"  
MZP 1,0; XBCP 18,0;  
XBAR; ...  
...  
EXIT;
```

XBCP 14 — enable addressee ID

Format

```
XBCP 14, 'addressee-id';
```

Parameter

addressee-id:

string containing addressee ID

Function

A Macro PDF417 symbol contains an optional Addressee ID field. The XBCP 14 command enables this field and defines its contents.

The string must be surrounded by single or double quote marks, be less than 200 characters long and only contain EXC compaction mode characters (20H~7FH). This command is ignored if the string does not satisfy these conditions. The execution of this command also automatically sets Macro PDF417 symbol mode.

File

```
!R! RES;  
XBCP 14, 'BAREFOOTINC';  
MZP 1,0; XBCP 18,0;  
XBAR; ...  
...  
EXIT;
```

XBCP 15 — enable file size

Format

```
XBCP 15;
```

Parameter

None

Function

A Macro PDF417 symbol contains an optional file size field. The XBCP 15 command enables this function. Note that the execution of this command also automatically sets Macro PDF417 symbol mode. This function is disabled in default condition.

File

```
!R! RES;  
XBCP 15;  
MZP 1,0; XBCP 18,0;  
XBAR; ...  
...  
EXIT;
```

XBCP 16 — enable checksum

Format

```
XBCP 16;
```

Parameter

None

Function

A Macro PDF417 symbol contains an optional checksum. The XBCP 16 command enables this function. Note that the execution of this command automatically sets Macro PDF symbol mode. This function is disabled in default condition.

File

```
!R! RES;  
XBCP 16;  
MZP 1, 0; XBCP 18, 0;  
  
XBAR; ...  
EXIT;
```


XBCP 17 — allow control of file ID

Format

```
XBCP 17, 'file-id' ;
```

Parameter

file-id:

string containing a file ID

Function

A Macro PDF417 symbol contains a required file ID. The XBCP 17 command allows control of this field. A random file ID is generated when XBCP 10 through 16 or XBCP 18 are executed without executing XBCP 17.

The string must be surrounded by single or double quotation marks, be less than 50 characters long and only contain EXC compaction mode characters (20H~7FH). This command is ignored if the string does not satisfy these conditions.




The execution of this command automatically sets Macro PDF417 symbol mode. Note that PDF barcode is positioned a half inch below the previous PDF barcode if there are no positioning parameters exist [See XBCP 19 (page 387).] or insufficient number of parameters exist. See the example in File below.

File

In Macro PDF mode, data such as large as 3000 bytes print 3 90-column-by-30-row PDF symbols which are a half inch apart vertically:

```
!R! RES ;
XBCP 17, "BookOne" ;
XBAR ; large data such as 3000 bytes ; ENDB ;
EXIT ;
```

Printout



XBCP 18 — enable Macro PDF417 symbol mode

Format

```
XBCP 18[ , block-index ] ;
```

Parameter

block-index:

0 to k-1, and k is the number of blocks

Function


Executing the XBCP 18 command enables the Macro PDF417 symbol mode. The Macro PDF417 symbol mode directs the reader to separate large data and allows Macro PDF417 two dimensional barcodes with a single file-id.

File

To bar-code data 'abcdXYZ12345' as a Macro PDF417 using block indices 0, 1, and 2:

```
!R!
UNIT C; CMNT Set unit to centimeters;
XBCP 17,"file-id"; CMNT Set file ID to file-id;
MZP 1,0; CMNT Move cursor to (1, 0) [centimeters];
XBCP 18,0; CMNT Set block index 0 to next data;
XBAR; abcd;ENDB; CMNT Encode and print data "abcd";
MZP 1,5; CMNT Move cursor to (1, 5) [centimeters];
XBCP 18,1; CMNT Set block index 1 to next data;
XBAR; XYZ;ENDB; CMNT Encode and print data "XYZ";
MZP 1,10; CMNT Move cursor to (1, 10) [centimeters];
XBCP 18,2; CMNT Set block index 2 to next data;
XBAR; 12345;ENDB; CMNT Encode and print data "12345";
EXIT;
```

Printout



XBCP 19 — position symbols at the specified locations

Format

```
XBCP 19 [ , x1 , y1 , x2 , y2 , x3 , y3 , ... ] ;
```

Parameter

x1, y1, x2, y2, x3, y3, ...

x and y coordinates that represent the positions of the divided barcodes measured from the current cursor position

Function

The multiple PDF417 symbol from one large data can be positioned at the specified locations using the XBCP 19 command. X_1 , Y_1 is the relative location of second PDF symbol to the first symbol, X_2 , Y_2 for third, and so on as measured from the current coordinate (like the MRP command described on the printing system's manual). The remaining pairs of positioning parameters are ignored if there are no more PDF symbol to print.

The relative location parameters, X_n and Y_n can not define the location beyond first page. The positioning command such as MZP and ENDB should be used to specify the macro PDF location if there are data beyond first page. PDF symbol will be positioned a half inch below the previous PDF symbol if there are no positioning parameter exist or insufficient number of positioning parameters exist.



The combination with multiple number of ENDB command allows to print multiple number of macro PDF on multiple pages. Note RES, XBCP0, and XBAR command will delete the remaining macro PDF data if all data has not been printed. Therefore, the page eject should not be done by RES command.

File 1

In Macro PDF mode, data such as large as 3000 bytes print 3 90-column-by-30-row PDF symbols at specified locations specified by XBCP 19. In the file below, the second line XBCP 17 activates Macro PDF mode and specifies *QualitySt* for *file-id*. The third line MZP command positions the first barcode at 1 inch each from the left and right paper edge limits. The fourth line XBCP 19 moves the cursor 1 inch from the left and 2 inches below the point where the first barcode started and prints the second barcode; then moves the cursor 1 inch from the left and 2 inches below the point where the second barcode started and prints the third barcode.:

```
!R! RES ;
XBCP 17 , "QualitySt" ;
MZP 1 , 1 ;
XBCP 19 , 1 , 2 , 1 , 2 ;
XBAR ; large data such as 3000 bytes ; ENDB ;
EXIT ;
```

Printout 1



File 2

To print 4 30-column-by-20-row PDF symbols at specified locations in Macro PDF mode. In example below, the first line XBCP command resets all XBCP parameters to the defaults and selects the barcode type for PDF417 (though it is the only type currently available); the second line XBCP sets the number of rows in the barcode to 20 (rows); the third line XBCP sets the number of data codeword in the barcode to 30 (codewords). Note that with the number of rows and codewords of 20 and 30 respectively, the data the XBAR encodes are split into 4 barcodes which can be positioned separately at different coordinates as specified by the seventh line XBCP command.:

```
!R! RES;  
XBCP 0,1;  
XBCP 4,20;  
XBCP 5,30;  
XBCP 17,"TrnOvr9";  
MZF 0.5, 0.5;  
XBCP 19,0.5,1,0.5,1,0.5,1,0.5,1;  
XBAR; large data such as 3000 bytes; ENDB;  
EXIT;
```


Printout 2*File 3*

To print multiple Macro PDF barcodes beyond the current page. In the example below, the fourth line XBCP 19 has no coordinating parameters specified. This allows the first segment barcode of the ensuing PDF data to be positioned at the current cursor position and the remaining data on the subsequent pages if there are data beyond the first page. The PAGE command on the sixth line starts the new page. The combinations of MZP and ENDB allow the second and third PDF barcodes printed at the specific coordinates on the following page.:


```
!R! RES;
XBCP 0,1;
XBCP 17,"TrnOvr0";
XBCP 19;
XBAR; large data such as 3000 bytes; ENDB;
PAGE;
MZP 1,2; ENDB;
MZP 4,4; ENDB;
EXIT;
```

Printout 3

First page



Second page



XBUF — define a BUfFer name

Format

```
XBUF buffer-name [ , [ data-length ] ] ; data-string ; ENDB ;
```

Parameters

buffer-name:

name of the buffer that contains XBAR data

data-length:

length of data in bytes, terminated with a semicolon

data-string:

data to be encoded, terminated with a semicolon immediately followed by an ENDB command (256 international characters or binary data)

Function

The XBUF command defines/undefines a buffer that contains actual data for the two dimensional barcodes.

The buffer name can be any length but only the first four characters count. The rest will be ignored. The distinction between upper and lowercase letters is also ignored. For example, the following names are all the same.

- ABCD
- abcd
- ABCDE
- Abcdxyz

The name must start with a letter, but the other characters can include digits and special symbols such as hyphens. For example, F-1 and GRY-2 are valid buffer names.

As the above Format instruction implies, the XBUF command sequence can have two other formats implementing the different functions as follows:

```
XBUF buffer-name , ; data-string ; ENDB ;
```

Note the comma (,) placed between the *buffer-name* and “;”. As is the case of XBAR, this format allows that *data-length* can be omitted, 0, negative value, or non integer, the comma after *buffer-name* must be left in place, however. Then *data-string* is counted all characters until the program encounters “; ENDB;”. It is necessary to have no space or linebreak between “;” and “ENDB;”.

The XBUF command sequence with no comma and *data-length* specified

```
XBUF buffer-name ;
```

is used to delete the buffer name from the printing system memory.

The XBUF command sequence with no comma, *buffer-name* and *data-length* specified

```
XBUF ;
```

is used to delete all buffer name from the printing system memory.

Files

...

```
XBCP 0,1;  
XBUF PDF 1, 25; This is a PDF417 example. ;ENDB;  
XBUF PDF 2, 15; This is a test. ;ENDB;
```

...

```
XBAR PDF 1;  
MRP 1,1;  
XBAR PDF 2;
```

The sixth line prints PDF417 barcode containing the data string buffered in the third line and the last line prints PDF417 barcode containing the data string buffered in the fourth line.

...

```
XBCP 0, 1;  
XBUF PDF 1, 25; This is a PDF417 example. ;ENDB;  
XBUF PDF 2, 15; This is a test. ;ENDB;
```

...

```
MCRO PDF;  
XBAR PDF 1;  
MRP 1, 1;  
XBAR PDF 2;  
ENDM;  
CALL PDF;
```

The last line prints the two PDF417 barcodes defined in the macro which is defined by MCRO in the sixth line.

XPAT — generate eXpanded fill PATtern

Format

```
XPAT pattern-number, bitmap;
```

Parameters

pattern-number:

number from 100 to 105

bitmap:

dot pattern of the pattern, coded as a character string

Function


The XPAT command generates a new, expanded shading pattern. The new pattern can be specified with the PAT command for use by the BLK and ARC commands and in a path.

Details for creation of new patterns appear in Chapter 4 of the Technical Reference manual.

File

```
!R! RES; UNIT C; MAP 5, 5;
XPAT 105;///?/?p@0p@0p@0p@0p@0p@0p@0p@0p@0p@0p@0p@0p@0p@0p@0p@0;
PAT 105; BOX 4, 1; BLK 4, 1;
PAGE;
EXIT;
```

Printout



Related Commands

PAT, BLK, ARC

Index

Numerics

2-dimensional barcode
See Two-dimensional barcode, 363

A

Absolute position, 73, 157
Alternate font
 assigning, 288
API, 242
Arc
 drawing an arc, 182
 standard graphics, 28

B

Barcode
 barcode positions diagrammed, 141
 command, 34
 flag, 35
Beveled line join, 305
Block, 40
 cursor moving option, 40
Booklet binding, 38
Box, 42
 cursor moving option, 42
Butt cap, 266

C

Carbon copy, 46
Character path, 64
Character spacing, 281
Circle, 52
Clipping
 rectangle, 55
Code set, 129, 226, 268
 returning to the saved set, 226
 saving current, 268
Collator, 68
Color, 159, 263, 274, 278
 adjusting lightness, 155
 color matching, 159
 commands, 7
 gamma control, 155
 monitor simulation, 315
 palette, 302
 saturation, 155, 263
 saving current palette, 278
 selecting in palette, 274
 vivid color, 159
Command recognition character, 215, 279
 changing, 279
Command start sequence, 215
 changing, 279
Comment, 59
Compression (SFNT), 294
 control, 63
Coordinates, 157

 moving to, 157
Country code, 129
Cursor option, 42
Cursor position, 275
Curve
 drawing to relative position, 186
 drawing to zero-relative, 188

D

Dash pattern, 78, 285
 a list as printout, 79
Dashed pattern, 79
Data dump, 218
 ending, 93
Destination image, 300
Downloading character, 147
Drawing rule (for fill), 101
Dummy sign (macro), 160
Dumping data, 218
Duplex
 inhibiting by media type, 162, 163

E

EcoPrint, 98
 level, 98
e-MPS, 165
 limiting values for storage, 165
e-MPS starting, 141
Emulation, 287
 changing, 287
End cap, 266
 butt, 266
 round, 266
 square, 266
Ending angle, 28, 182
EPROM, 246
Exit, 99
 exiting from PRESCRIBE, 99
External media
 assigning terminating string, 249
 deleting data, 242
 formatting, 243
 How to read data from, 246
 How to write data to, 250
 printing information, 244
 printing partitions, 245
 reading data from, 246
 storing TrueType fonts, 247
 terminator, 362
 writing data to, 250

F

Face-down output tray, 335
Face-up output tray, 335
Fill, 184
Fill pattern, 110
 8 by 8 bit, 110

a list of resident, 185
a list of resident fill patterns, 185
column value, 110
Filling path, 101
Firmware, 112
 initializing, 114
 Reprogramming, 112
Flatness of curves, 104
Font, 71, 75, 115, 270
 assigning attributes (bitmap), 290
 changing by characteristic, 115
 changing by number, 108
 characteristics, 116
 deleting a, 75
 deleting all nonresident, 71
 evaluation by priority, 116
 font mode, 108, 119
 saving current, 270
 selection by assigning typeface, 294
Font attributes, 291
Font list, 105
 printing a, 105
Font mode, 119
Fuser mode, 162, 163

G

Graphics state, 228, 272
Gray pattern, 122
Grouping, 27

H

Hard disk, 141, 246, 250, 256, 258
 error codes, 136

I

Image model, 299
International characters, 129

J

Job, 136, 138, 140
 printing a list of, 134
 printing jobs, 136
 printing with different finishing options, 138
 starting to define a ..., 140
 terminating to define a ..., 144
Job offset, 145
Job separation, 145
 choosing first or all pages, 145
 separating copies by command, 145

K

KCGL
 pen color, 297

KIR, 304
 refinement level, 304
KPDL, 287
Kyocera Image Refinement, 304

L

left, 307
Lightness, 155
Line, 73, 80, 82, 87
 drawing by angle, 82
 drawing to absolute, 73
 drawing to relative, 80
 drawing to zero-relative, 87
Line join, 305
Line end (caps), 266
Line join, 313
 beveled, 305
 miter limite, 313
 mitered, 305
 notched, 305
 round, 305
Line spacing, 312
Lines per inch, 309
Lines per page, 310
Logical page, 190
 diagrams, 191
 Print direction, 190

M

Macro, 72, 76
 automatic macro, 25
 calling a, 44
 commands, 8
 defining, 160
 deleting a, 76
 deleting all, 72
 dummy parameter, 160
 ending a, 94
 macro overlay, 25
 sources, locations, properties, 161
mailbox, 171
Margin, 264, 307
 bottom, 264
 right, 328
 top, 338
Margine
 left margin, 307
Media type, 162, 163, 172
 attributing media type, 162, 163
Memory card, 246, 250, 256, 258
 autostart, 256
Miter limit, 305, 313
Mitered line join, 305
Monitor simulation, 315
Move, 157, 167, 174
 moving to absolute, 157
 moving to relative, 167
 moving to zero-relative, 174
Moving specifying angle, 169

N

Notched line join, 305
Number of copies, 61

O

Operation mode, 299

P

Page break, 179
Page length, 320
Page orientation, 322
Page width, 326
Palette, 63, 278
 creating simple color, 302
 selecting, 317
Pantone, 180
 creating a palette, 180
 Pantone colors, 36, 181
Paper cassette
 grouping, 27
Paper size, 282, 324
 customize, 282
 enlargement/reduction, 324
Paper weight, 162, 163
Parsing text, 351
Path, 176, 182, 184, 193, 195, 197, 201, 203, 205
 ellipse path, 197
 filling a, 101
 Line, drawing to relative, 193
 Line, drawing to zero-relative, 195
 move, moving to relative, 203
 move, moving to relative by angle, 201
 move, moving to zero-relative, 205
 selecting a fill pattern, 184
 stroking a path, 344
Pen
 color, 297
 diameter, 318
 thickness, 318
Permanent job, 141
Pie chart, 199
Pitch, 276, 281, 309, 310
 characters-per-inch, 276
PRESCRIBE start sequence, 215
Presentation mode, 330
Previous position, 230
print density, 348
Printing right-aligned text, 236
Punch, 207

R

Radius, 28, 199
RAM disk, 246, 250
Raster
 compression, 238
 dot resolution, 343

logical operation, 332
presentation mode, 330
receiving, 240
receiving color raster data, 239
ROP3, 332
run-length encoding, 238
TIFF rev. 4.0, 238

Raster data
 ending, 96
Registration, 225
Relative position, 80, 82, 167
Resetting the printer
 by emulation, 220
Resolution, 222
RGB
 color balance level, 223
 gamma, 223
ROP3, 332
Round cap, 266
Round line join, 305
Run-length, 239

S

Saturation, 155, 263
saving, 278
Saving current, 275
Saving cursor position
 restoring, 230
Saving graphics state, 272
 restoring, 228
separate mode, 352
Shade of gray, 122
Shear, 294
Sorter, 164, 334
 addressing a tray group, 164
 selecting trays, 334
Source image, 299
Square cap, 266
Stacker, 335
 selecting tray, 335
stacker, 171
Staple, 141, 340
 changing number of pages, 340
 position, 341
Starting a new page, 179
Starting a new path, 176
Starting angle, 28, 182, 199
Status page, 336
 user mode and service mode, 336
Symbol set, 65, 129, 226, 268, 294
 as a bitmap attribute, 290
 ID by CSET, 65
 returning to the saved set, 226

T

Temporary job, 140
Test page, 336
Text, 69, 349
 centering a, 69

Text parsing, 351
Thickness, 318
TIFF 4.0, 239
Transparency, 299, 352
TrueType, 153, 247
 loading ..., 153
Two-dimensional barcode
 addressee ID, 380
 automatic rows and columns, 374
 bar height, 373
 block count, 377
 checksum, 382
 data code word columns, 370
 data code word rows, 369
 error correction level, 368
 error correction level by percentage, 367
 file ID control, 383
 file name, 376
 file size, 381
 macro PDF417, 385
 naming buffer, 391
 positioning at specified locations, 387
 printing a, 363
 ratio of height and width, 371
 selecting type, 365
 sender ID, 379
 specifying narrowest width, 366
 time stamp, 378
 truncation, 375
type of paper, 348
Typeface, 294

U

Underline, 346
 distance and thickness, 346
Unit, 283, 353
 saving current, 283
Unit of measurement, 283, 353, 355
 per dots, 355

V

Vector graphics
 commands, 5
Virtual mailbox
 spooling as a job, 141
Virtual mailbox, 357, 358
 aliasing a mailbox, 357
 assigning a mailbox to receive jobs, 358
 defining a job to store in a mailbox, 140
 password, 359
 printing a list of, 134
 printing using command, 136

W

Wide A4, 361

Z

Zero-relative position, 87, 174

